
0. FAQ

Downloading

You can download a snapshot of the repository here. 

Users may find more flexibility if they fork this repository. At https://github.com/NVlabs/moderngpu, click 
on the Fork button in the upper-right. This creates a copy of the repository in your own github account.

git clone git@github.com:yourname/moderngpu

From the command line you can clone your own fork of the project onto your local machine. You can make 
changes to the project and these will be updated in your own repository. Users forking MGPU are treated to 
Github's excellent suite of development tools. Use the Network Graph Visualizer to stay current with Modern 
GPU updates.

Compiling

The Modern GPU library is entirely defined in headers under the include directory, except three .cpp files 
that must be compiled and linked manually: src/format.cpp, src/random.cpp, and src/mgpucontext.cpp. You 
cannot compile for architectures below sm_20 (Fermi), as the sm_1x compiler doesn't support all the features 
used by MGPU.

All device and host functions are included from include/moderngpu.cuh; this is all you need to include to 
access everything. Additionally, all functions and types are defined inside the mgpu namespace.

To compile from the command line (from the moderngpu/tests directory):

nvcc -arch=sm_20 -I ../include/ -o demo ../src/format.cpp ../src/random.cpp ^
    ../src/mgpucontext.cpp demo.cu

To specifically target multiple device architectures (necessary if you are using LaunchBox to tune kernels), 
try something like this:

nvcc -gencode=arch=compute_20,code=\"sm_20,compute_20\" ^
    -gencode=arch=compute_35,code=\"sm_35,compute_35\" -I ../include -o demo ^
    ../src/format.cpp ../src/random.cpp ../src/mgpucontext.cpp demo.cu

If you are a Visual Studio user, MGPU includes a solution for VS2010 with projects for the demo and each 

https://github.com/NVlabs/moderngpu/tree/master/tests
https://github.com/NVlabs/moderngpu/blob/master/include/moderngpu.cuh
https://github.com/NVlabs/moderngpu/blob/master/src/mgpucontext.cpp
https://github.com/NVlabs/moderngpu/blob/master/src/random.cpp
https://github.com/NVlabs/moderngpu/blob/master/src/format.cpp
https://github.com/blog/39-say-hello-to-the-network-graph-visualizer
https://github.com/NVlabs/moderngpu
https://github.com/NVlabs/moderngpu/archive/master.zip


benchmark. To start a new project that uses CUDA and MGPU, create a new "Win32 Project" or "Win32 
Console Project." Right-click on the project in the Solution Explorer and choose "Build Customizations..." 
This lists configuration files for each CUDA Toolkit installed on your system. Check the newest one:

Right-click on the project again, select "Add->Existing Items..." and add format.cpp, random.cpp, and 
mgpucontext.cpp from the src directory of your Modern GPU directory.

Optional: If you want to use the same project settings as MGPU, in the menu bar select "View->Property 
Manager." Right click on your project in the Property Manager and choose "Add Existing Property Sheet..." 
Select vs.props from the base directory of your MGPU install.

To configure CUDA properties for the project, go back to the Solution Explorer, right click on the project, 
and choose "Properties."



Make sure to compile with compute_20,sm_20 and higher; compute_1x will not build. You'll need to set 
mgpu/include under "Additional Include Directories" in the C/C++->General property page. Additionally 
you'll need to link against cudart.lib in Linker->Input->Additional Dependencies.

Debugging

NVIDIA has offers Nsight, a rather impressive development and debugging suite for Visual Studio and 
Eclipse. I'm a bit of a luddite and mostly get by with two simple tools:

cuda-memcheck

cuda-memcheck is a post-mortem debugger for the command line. When your kernel makes an out-of-range 
load/store or something else forbidden, cuda-memcheck aborts the program and prints detailed information 
on the nature of the error.

1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16
17
18
19
20

#include <cuda.h>
 
__global__ void Foo(int* data_global) {
    __shared__ int s[128];
 
    int tid = threadIdx.x;
    s[tid + 1] = tid;           // out-of-range store!
    __syncthreads();
 
    data_global[tid] = s[tid];
}
 
int main(int argc, char** argv) {
    int* data;
    cudaMalloc((void**)&data, 128 * sizeof(int));
    Foo<<<1, 128>>>(data);
 
    cudaDeviceSynchronize();
    return 0;
}

cuda-memcheck tests.exe
========= CUDA-MEMCHECK
========= Invalid __shared__ write of size 4
=========     at 0x00000020 in c:/projects/mgpulib/tests/test.cu:7:Foo(int*)
=========     by thread (127,0,0) in block (0,0,0)
=========     Address 0x00000200 is out of bounds
=========     Saved host backtrace up to driver entry point at kernel launch time
=========     Host Frame:C:\Windows\system32\nvcuda.dll (cuLaunchKernel + 0x166) 
[0xc196]

cuda-memcheck reports the nature of the error (invalid __shared__ write of size 4) and the function it 
occurred in. If you compile with -lineinfo (or select the appropriate box in the Visual Studio CUDA C/C++ 
properties), cuda-memcheck might even give you the line number, as it did in this case.

https://developer.nvidia.com/cuda-memcheck
https://developer.nvidia.com/nsight-eclipse-edition
https://developer.nvidia.com/nvidia-nsight-visual-studio-edition


If you want more context, use cuobjdump to dump the disassembly of your kernel:

cuobjdump -sass tests.exe

Fatbin elf code:
================
arch = sm_20
code version = [1,6]
producer = cuda
host = windows
compile_size = 32bit
identifier = c:/projects/mgpulib/tests/test.cu

        code for sm_20
                Function : _Z3FooPi
        /*0000*/     /*0x00005de428004404*/     MOV R1, c [0x1] [0x100];
        /*0008*/     /*0x84001c042c000000*/     S2R R0, SR_Tid_X;
        /*0010*/     /*0xfc1fdc03207e0000*/     IMAD.U32.U32 RZ, R1, RZ, RZ;
        /*0018*/     /*0x08009c036000c000*/     SHL R2, R0, 0x2;
        /*0020*/     /*0x10201c85c9000000*/     STS [R2+0x4], R0;
        /*0028*/     /*0xffffdc0450ee0000*/     BAR.RED.POPC RZ, RZ;
        /*0030*/     /*0x00201c85c1000000*/     LDS R0, [R2];
        /*0038*/     /*0x80209c0348004000*/     IADD R2, R2, c [0x0] [0x20];
        /*0040*/     /*0x00201c8590000000*/     ST [R2], R0;
        /*0048*/     /*0x00001de780000000*/     EXIT;
                .........................

cuda-memcheck reported an "invalid __shared__ write of size 4" at address 0x00000020. The disassembly 
shows us the instruction at this address, and it is indeed an STS (store to shared 4 bytes).

printf

Device-side printf is available on architectures sm_20 and later. It is extremely helpful. However you don't 
want 100,000 threads all printing to the console at once. Try to narrow down your problem to a single 
offending CTA and print from that. Individual printf statements are treated atomically (the entire string will 
come out at once), however the order in which threads print is undefined. It is helpful practice to store 
arguments to shared memory, synchronize, and have thread 0 read out the elements in order and printf in a 
loop.

The results of a device printf are not displayed until the next synchronizing runtime call after the kernel 
launch. This could be a cudaDeviceSynchronize, cudaMalloc, or cudaMemcpy.

Although printf is among the most primitive of debugging tools, it is surprisingly effective with data-parallel 
languages. Active debugging is often too fine-grained to understand the activity across an entire CTA.

Getting started

How do I get started with CUDA?

The best place to get started with CUDA is the official Programming Guide. This is an up-to-date, correct, 
and concise overview of all of the device's capabilities and the APIs needed to use them.

There is a growing library of textbooks that paint a more detailed picture of GPU computing:

http://docs.nvidia.com/cuda/cuda-c-programming-guide/index.html


• The CUDA Handbook   - Nicholas Wilt

• CUDA Programming   - Shane Cook

• CUDA Application Design and Development   - Rob Farber

• CUDA by Example   - Jason Sanders

• Programming Massively Parallel Processors   - David Kirk and Wen-mei Hwu

Professor John Owens of UC Davis and Professor David Luebke, Graphics Research chief at NVIDIA, 
produced a video-rich CUDA course, available for free at Udacity, that covers hardware architecture, the 
CUDA toolkit, and parallel algorithms.

The CUDA Forums are the most trafficked pages for giving and receiving help. Stackoverflow also is very 
popular.

Contact

To contact me on email, use moderngpu@gmail.com. Follow @moderngpu for notifications of new content. 
I can often be found in #cuda on Freenode IRC.

License

The new Modern GPU library is provided under the 3-clause BSD license:

/******************************************************************************
 * Copyright (c) 2013, NVIDIA CORPORATION.  All rights reserved.
 * 
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 *     * Redistributions of source code must retain the above copyright
 *       notice, this list of conditions and the following disclaimer.
 *     * Redistributions in binary form must reproduce the above copyright
 *       notice, this list of conditions and the following disclaimer in the
 *       documentation and/or other materials provided with the distribution.
 *     * Neither the name of the NVIDIA CORPORATION nor the
 *       names of its contributors may be used to endorse or promote products
 *       derived from this software without specific prior written permission.
 * 
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" 
 * AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
 * IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE 
 * ARE DISCLAIMED. IN NO EVENT SHALL NVIDIA CORPORATION BE LIABLE FOR ANY
 * DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES
 * (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES;
 * LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND
 * ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT
 * (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS
 * SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.
 *
 ******************************************************************************/

mailto:moderngpu@gmail.com
http://opensource.org/licenses/BSD-3-Clause
http://freenode.net/
https://twitter.com/moderngpu
http://www.stackoverflow.com/questions/tagged/cuda
https://devtalk.nvidia.com/
https://www.udacity.com/course/cs344
https://twitter.com/davedotluebke
https://twitter.com/jowens
http://www.amazon.com/Programming-Massively-Parallel-Processors-Edition/dp/0124159923
http://www.amazon.com/CUDA-Example-Introduction-General-Purpose-Programming/dp/0131387685
http://www.amazon.com/CUDA-Application-Design-Development-Farber/dp/0123884268
http://www.amazon.com/CUDA-Programming-Developers-Computing-Applications/dp/0124159338
http://www.amazon.com/CUDA-Handbook-Comprehensive-Guide-Programming/dp/0321809467


1. Introduction
Parallel computing is important because it enables much higher throughput than conventional systems. The 
GPU is revolutionary because it does this affordably.

Libraries

Massive parallelism is the future of computing, but it comes with some challenges. Without good tools, 
targeting these systems complicates development and slows productivity—we'd need to see serious 
performance gains to justify the added effort. Fortunately, the CUDA software ecosystem has a number of 
quality libraries which ease development, improve productivity, and help users see super-charged 
performance. These packages solve many common problems, are reliable, and abstract away the complexities 
of the device. With them, realizing the high-performance of GPU computing is made easy for software 
professionals and made possible for the non-ninja domain expert. If you aren't already familiar with them, 
check out these pre-compiled libraries that ship with the CUDA Toolkit:

• CUBLAS   - The NVIDIA CUDA Basic Linear Algebra Subroutines library is a GPU-accelerated 
version of the complete standard BLAS library that delivers 6x to 17x faster performance than the 
latest MKL BLAS.

• CUSPARSE   - The NVIDIA CUDA Sparse Matrix library provides a collection of basic linear algebra 
subroutines used for sparse matrices that delivers up to 8x faster performance than the latest MKL.

• CURAND   - The NVIDIA CUDA Random Number Generation library delivers high performance 
GPU-accelerated random number generation (RNG). The cuRAND library delivers high quality 
random numbers 8x faster using hundreds of processor cores available in NVIDIA GPUs.

• CUFFT   - The NVIDIA CUDA Fast Fourier Transform library provides a simple interface for 
computing FFTs up to 10x faster. By using hundreds of processor cores inside NVIDIA GPUs, cuFFT 
delivers the floating point performance of a GPU without having to develop your own custom GPU ‐
FFT implementation.

• NPP   - The NVIDIA Performance Primitives library is a collection of GPU-accelerated image, video, 
and signal processing functions that deliver 5x to 10x faster performance than comparable CPU-only 
implementations. Using NPP, developers can take advantage of over 1900 image processing and 
approx 600 signal processing primitives to achieve significant improvements in application 
performance in a matter of hours.

Pull the latest revisions of these versatile template libraries developed by my colleagues at NVIDIA 
Research:

• Thrust   - Thrust is a parallel algorithms library which resembles the C++ Standard Template Library 
(STL). Thrust's high-level interface greatly enhances programmer productivity while enabling 
performance portability between GPUs and multicore CPUs.

• CUB   - CUB is a library of high-performance parallel primitives and other utilities for constructing 
CUDA kernel software. CUB enhances productivity, performance, and portability by providing an 
abstraction layer over complex block-level, warp-level, and thread-level operations.

http://nvlabs.github.io/cub/
http://thrust.github.io/
https://developer.nvidia.com/npp
https://developer.nvidia.com/cufft
https://developer.nvidia.com/curand
https://developer.nvidia.com/cusparse
https://developer.nvidia.com/cublas


Goals

Modern GPU was developed to help ambitious CUDA programmers push out the frontier. Every user's 
problem is unique; real applications require functionality that existing libraries don't provide. Programmers 
will have to solve original problems by getting their hands dirty and writing device code. This project covers 
algorithms, idioms, and strategies to help users craft the best kernels they can.

Modern GPU is code and commentary intended to promote new and productive ways of thinking about 
GPU computing. The library, moderngpu, was designed with four goals:

1. Utility

The great selling point of GPUs is their extraordinary floating-point throughput, and this has been 
well exploited with compute libraries like CUBLAS and visualization APIs like OpenGL. Users have 
been successful in writing kernels that chew through dense, regular, numerical problems. These are 
often domain-specific operations like particle-particle interactions for physics and chemistry or image 
filters for computer vision.

Difficulty occurs between the calls to these float-heavy routines. How do you gather and utilize your 
results? How do you prepare inputs for the next launch? I want to push the GPU further into this 
territory. The GPU has high memory bandwidth and an amazing latency-hiding architecture that is 
well suited for fine-grained manipulation of data. MGPU focuses on the most generic of problems: 
manipulation of arrays and fundamental CS algorithms. We look at sorting and searching, the bricks 
and mortar of programming. 

The biggest challenge impeding wide-spread adoption of this technology is the difficulty in finding 
and exposing parallelism in irregular problems. I introduce strategies for dealing with this, and 
demonstrate how to put these ideas into execution by building a high-performance, comprehensive 
array-processing library.

2. Novelty

Most of the ideas in Modern GPU are new and may seem unusual even to CUDA veterans. Although 
the code is very high performance, it isn't overtly concerned with the traditional nuts and bolts of 
GPU programming. MGPU de-emphasizes low-level programming and focuses on making CUDA 
more expressive. It emphasizes geometric reasoning and develops an idiom for solving problems by 
breaking solutions into two distinct phases:

1. Find a coarse-grained partitioning of the problem that exactly load-balances work over each 
thread. Scheduling—the exercise of mapping work items to CTAs (cooperative thread arrays) 
and threads on the GPU—is handled in this phase.

2. Execute simple, work-efficient, sequential logic that solves the problem. Because scheduling 
is part of the partitioning phase, this sequential phase runs embarrassingly parallel.

Partitioning involves a search over one or more input sequences. All the characteristic trickiness of 
parallel computing is isolated to this phase. Specific problem-solving code is run independently by 
each thread, and resembles code you'd have written for a CPU. By decoupling partitioning from the 
work logic, we improve modularity and make both phases easier to reason about.

https://www.github.com/NVlabs/moderngpu/


This two-phase idiom comprises a new style for programming GPUs. Programmed with this idiom in 
mind, a number of functions are elegantly expressed:

3. Segmented sort   is a work-efficient mergesort on multiple variable-length arrays.

4. Interval move   schedules multiple variable-length coarse-grained memcpys.

5. Load-balancing search   coordinates work-items with the objects that scheduled them, 
allowing perfect load-balancing for functions that expand and contract data.

6. Relational joins   compute the outer products of variable-length duplicate ranges from two 
sorted inputs. Although this functionality is very difficult to express using conventional 
CUDA strategies, it comes out effortlessly with these new primitives.

Performance tuning is simple and well-understood in this framework. Each thread processes VT 
(Values per Thread) items. Increasing VT assigns more work to each thread, amortizing the cost of 
partitioning, which is constant per thread. Although this increases work-efficiency, it decreases 
parallelism and the GPU's ability to hide latency. As described below, programmers who use this 
framework may benchmark to find an optimal VT parameter for a particular device architecture, input 
type, size, and distribution, without having to modify any device code. 

3. Clarity

The central content of MGPU is the source code. Accordingly I've made clarity and conciseness a 
point of pride. The code is organized in a very flat directory structure, in a single namespace, with no 
dependencies other than the CUDA Toolkit. Routine operations that offer little of interest to the 
reader have been factored out, resulting in code that is dense with algorithms. The user shouldn't have 
to chase a function's flow of execution through more than a couple of files to see its entire 
implementation.

For flexibility the functions are heavily templated, each parameterized over input and output types, to 
support Boost-style iterators, comparators, and predicate objects. Still, this library is not an exercise in 
meta-programming, and the use of generics is limited.

A call to each host-level function is demonstrated in a simple form at the top of its corresponding web 
page (drawn from tests/demo.cu), and in a more complex form in its benchmark code, which 
generates the performance charts. The articles are organized so that functions are rolled out with 
increasing complexity and build on one another. Readers with CUDA experience who take the pages 
in order should be should not be unprepared for anything presented. 

4. Hackability 

A quality of software that's been undervalued in recent years is hackability. Under schedule pressure, 
programmers grab and modify whatever works. The world runs on code spliced from Numerical 
Recipes and Sedgewick, O'Reilly books, Wikipedia samples, conference talks, powerpoints, and 
message boards.

Much emphasis has been put on the idea of composability in CUDA software. For example, Thrust 
combines simple functions with powerful iterators to increase functionality. While MGPU supports 
these same iterators and is largely interoperable with Thrust, my philosophy is that composability is 

http://en.wikipedia.org/wiki/Composability
https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
http://nvlabs.github.io/moderngpu/intro.html#performance
http://nvlabs.github.io/moderngpu/join.html
http://nvlabs.github.io/moderngpu/loadbalance.html
http://nvlabs.github.io/moderngpu/intervalmove.html
http://nvlabs.github.io/moderngpu/segsort.html


just one aspect of software reuse. This library is designed to be forked and modified. If a function 
does nearly what you desire, but not exactly, creating a derivative function will not be a burden.

Two-phase decomposition

A major challenge in parallel programming—and especially in the massively fine-grained parallelism that 
GPUs provide—is deciding which work to run on which processor, and when to run it. This is a problem of 
scheduling or decomposition. Attempts to implement a function and manage decomposition with a single 
strategy can be unwieldy or inefficient, because the user is trying to solve two problems at once.

CPU Merge implementation

template<typename T, typename Comp>
void CPUMerge(const T* a, int aCount, const T* b, int bCount, T* dest,
    Comp comp) {
 
    int count = aCount + bCount;
    int ai = 0, bi = 0;
    for(int i = 0; i < count; ++i) {
        bool p;
        if(bi >= bCount) p = true;
        else if(ai >= aCount) p = false;
        else p = !comp(b[bi], a[ai]);
 
        dest[i] = p ? a[ai++] : b[bi++];
    }
}

Consider this sequential merge implementation. It takes two sorted inputs and loops over each output. During 
each iteration, two inputs are compared and the smaller one is emitted. The implementation is simple and 
work-efficient. Because it's totally sequential, there's no real consideration given to scheduling or 
decomposition.

tests/parallelmerge.cu

https://github.com/NVlabs/moderngpu/blob/master/tests/parallelmerge.cu

template<int NT, typename InputIt1, typename InputIt2, typename OutputIt,
    typename Comp>
__global__ void ParallelMergeA(InputIt1 a_global, int aCount, InputIt2 
b_global,
    int bCount, OutputIt dest_global, Comp comp) {
 
    typedef typename std::iterator_traits<InputIt1>::value_type T;
 
    int gid = threadIdx.x + NT * blockIdx.x;
    if(gid < aCount) {
        T aKey = a_global[gid];
        int lb = BinarySearch<MgpuBoundsLower>(b_global, bCount, aKey, 
comp);
        dest_global[gid + lb] = aKey;
    }
}
 
template<int NT, typename InputIt1, typename InputIt2, typename OutputIt,
    typename Comp>
__global__ void ParallelMergeB(InputIt1 a_global, int aCount, InputIt2 
b_global,
    int bCount, OutputIt dest_global, Comp comp) {
 
    typedef typename std::iterator_traits<InputIt2>::value_type T;
 
    int gid = threadIdx.x + NT * blockIdx.x;
    if(gid < bCount) {
        T bKey = b_global[gid];
        int ub = BinarySearch<MgpuBoundsUpper>(a_global, aCount, bKey, 
comp);
        dest_global[gid + ub] = bKey;
    }
}

Now consider this first attempt at parallel merge. There are two kernels: ParallelMergeA, which assigns 
one thread to each element in A, binary searches for the lower-bound in B, and outputs A keys to the 
destination; and ParallelMergeB, which assigns one thread to each element in B, binary searches for the 
upper-bound in A, and outputs B keys to the destination.

Although this implementation is highly concurrent, it's also highly inefficient. The O(n)-efficiency sequential 
code is now O(n log n), as each output requires a binary search over the input. The code only runs quickly if 
the number of processors is large compared to the input size. Additionally, the new code looks nothing like 
the sequential version. Because we've baked scheduling logic into the problem-solving logic, we've written 
code that is difficult to optimize. Perhaps even worse, it's difficult to extend—for example, how would we 
support multiset operations with this PRAM-style scheduling?

include/device/ctasearch.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctasearch.cuh
http://nvlabs.github.io/moderngpu/sets.html

template<MgpuBounds Bounds, typename It1, typename It2, typename Comp>
MGPU_HOST_DEVICE int MergePath(It1 a, int aCount, It2 b, int bCount, 
int diag,
    Comp comp) {
 
    typedef typename std::iterator_traits<It1>::value_type T;
    int begin = max(0, diag - bCount);
    int end = min(diag, aCount);
 
    while(begin < end) {
        int mid = (begin + end)>> 1;
        T aKey = a[mid];
        T bKey = b[diag - 1 - mid];
        bool pred = (MgpuBoundsUpper == Bounds) ? 
            comp(aKey, bKey) : 
            !comp(bKey, aKey);
        if(pred) begin = mid + 1;
        else end = mid;
    }
    return begin;
}

MGPU focuses on a two-phase strategy for execution. In the first phase we address partitioning: this phase 
handles scheduling and decomposition. The goal is to map work onto each thread in a load-balanced and 
work-efficient manner. All the functions in this library are parameterized over a grain size (the parameter 
VT), which controls the amount of work scheduled per thread. Increasing the grain size amortizes 
partitioning costs, improving work-efficiency, while simultaneously reducing occupancy and potentially 
sacrificing execution efficiency. Because partitioning is not tied up with the problem-solving logic, we can 
reuse this code in many functions, and optimize it by searching the grain-size parameter space.

Examples of partitioning functions used in MGPU are Merge Path, Balanced Path, and load-balancing 
search. The MergePath implementation above decomposes merge-like problems, which includes 
mergesort, vectorized sorted search, and the load-balancing search scheduling function. This function binary 
searches both input arrays simultaneously, producing a decomposition that maps a uniform amount of work 
to each thread.

include/device/ctamerge.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctamerge.cuh
http://nvlabs.github.io/moderngpu/loadbalance.html
http://nvlabs.github.io/moderngpu/sortedsearch.html
http://nvlabs.github.io/moderngpu/mergesort.html
http://nvlabs.github.io/moderngpu/loadbalance.html#algorithm
http://nvlabs.github.io/moderngpu/loadbalance.html#algorithm
http://nvlabs.github.io/moderngpu/sets.html#balancedpath
http://nvlabs.github.io/moderngpu/bulkinsert.html#mergepath

template<int VT, bool RangeCheck, typename T, typename Comp>
MGPU_DEVICE void SerialMerge(const T* keys_shared, int aBegin, int aEnd,
    int bBegin, int bEnd, T* results, int* indices, Comp comp) { 
 
    T aKey = keys_shared[aBegin];
    T bKey = keys_shared[bBegin];
 
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        bool p;
        if(RangeCheck) 
            p = (bBegin >= bEnd) || ((aBegin < aEnd) && !comp(bKey, 
aKey));
        else
            p = !comp(bKey, aKey);
 
        results[i] = p ? aKey : bKey;
        indices[i] = p ? aBegin : bBegin;
 
        if(p) aKey = keys_shared[++aBegin];
        else bKey = keys_shared[++bBegin];
    }
    __syncthreads();
}

The second phase in the strategy is the actual problem-solving logic. For efficiency and clarity we prefer that 
this code resembles sequential functions. The CUDA device function SerialMerge resembles the 
sequential CPU functon CPUMerge. Because concurrency was addressed in the partitioning phase, this 
phase can implement GPU-specific optimizations without complicating the decomposition. In addition to its 
improved work efficiency (we process VT inputs for each binary search), SerialMerge realizes extra 
throughput by unrolling the serial merge loop and storing intermediates to register to conserve shared 
memory and improve occupancy.

Kernels written using the two-phase decomposition are more involved than direct solutions like the naive 
parallel merge, but are more efficient, easier to optimize, and more flexible; they easily accommodate 
algorithmic changes to solve related problems. MGPU's functions continually stress the same few 
decomposition strategies, turning them into boilerplate code that sets up the problem-specific second phase.



This figure benchmarks the two-phase implementation of Merge (square markers) against the naive parallel 
version (round markers). Due to the device's advantage in memory bandwidth, even the unoptimized GPU 
code beats STL by 10x (run on a Sandy Bridge i7 at 2.8ghz). The two-phase implementation beats the naive 
code by 5x for large inputs.

The two-phase implementation's throughput grows as the workload increases, better filling the device. The 
naive code hits its highest throughput for small problem sizes. While wider workloads run with better 
execution efficiency on the device (more concurrency means better latency hiding), this benefit is 
counteracted by the O(n log n) work-efficiency—the cost of the binary search grows with the log of the input 
size.

Two-phase design delivers consistently high throughput of merge-like functions, while promoting code reuse 
and readability.

From scan to load-balancing search

Most early work on GPU algorithms reduce to scan or scan-like patterns. Scan is a miracle of efficient 
parallel communication. Radix sort, perhaps the most successful general-purpose CS algorithm to build on 
GPU, is essentially a very intricate scan: because of the mechanical and regular nature of radix sort, scan 
manages to both evenly distribute work over threads and solve the key-ranking problem. 

Conventional wisdom is to lower every problem to scan, because we've proven that scan helps solve 
problems with cooperative parallelism. The flaw in this reasoning is that we don't need to actually solve most 
problems in parallel—it is simpler and more efficient to partition problems in parallel, then solve them 
sequentially. Modern GPU pushes out the frontier by experimenting with new idioms and only using scan 
when it is the right tool for the job.



We introduce a new pattern, load-balancing search, which can be thought of as a particular type of inverse 
of scan. Access to this operator makes certain problems trivial and generally helps reduce the 
circumlocutions of scan-centric parallel programming. The load-balancing search uses two-phase 
decomposition to make certain dependencies explicit and further ease scheduling burdens.

Expand

Consider a vectorized fill function. It replicates each input, in order, a variable number of times. We'll call it 
expand.

CPU Expand example.

template<typename T>
void Expand(int numOutput, const int* scan, int numInput, const T* 
values,
    T* output) {
 
    for(int i = 0; i < numInput; ++i) {
        int offset = scan[i];
        int end = (i + 1 < numInput) ? scan[i + 1] : numOutput;
        std::fill(output + offset, output + end, values[i]);
    }
}
 
int Scan(const int* counts, int numTerms, int* scan) {
    int x = 0;
    for(int i = 0; i < numTerms; ++i) {
        scan[i] = x; 
        x += counts[i];
    }
    return x;
}
 
int main(int argc, char** argv) {
    const char* Alphabet = "ABCDEFGHIJKLMNOPQRSTUVWXYZ";
    const int Counts[26] = { 
        3, 1, 0, 0, 7, 3, 2, 14, 4, 6, 0, 2, 1,
        5, 3, 0, 5, 1, 6, 2, 0, 0, 9, 3, 2, 1       
    };
     
    // Scan the counts
    int Offsets[26];
    int total = Scan(Counts, 26, Offsets);
 
    std::vector<char> results(total + 1);
    Expand(total, Offsets, 26, Alphabet, &results[0]);
     
    printf("%s\n", &results[0]);
    return 0;
}

We start with a set of input values (the alphabet) and a corresponding set of counts. Scan the counts to 
compute the offsets for each fill operation. Loop through the list of offsets and call std::fill to copy 
each input, values[i], Counts[i] times.

Thrust provides a set of primitives (transform, scan, gather, scatter, compact) that are composed with 
iterators, operators, and comparators to solve problems. The user typically calls transform, gather, and scatter 
to prepare intermediate values, scans or compacts them, and uses transform, gather, and scatter to complete 
the function. The difficulty is that there is no separation between two basically distinct challenges—
partitioning and work logic.

Consider this implementation of expand written with Thrust:

thrust/examples/expand.cu

// This example demonstrates how to expand an input sequence by 
// replicating each element a variable number of times. For example,
//
//   expand([2,2,2],[A,B,C]) -> [A,A,B,B,C,C]
//   expand([3,0,1],[A,B,C]) -> [A,A,A,C]
//   expand([1,3,2],[A,B,C]) -> [A,B,B,B,C,C]
//
// The element counts are assumed to be non-negative integers
 
template <typename InputIterator1,
          typename InputIterator2,
          typename OutputIterator>
OutputIterator expand(InputIterator1 first1,
                      InputIterator1 last1,
                      InputIterator2 first2,
                      OutputIterator output)
{
  typedef typename thrust::iterator_difference<InputIterator1>::type
      difference_type;
   
  difference_type input_size  = thrust::distance(first1, last1);
  difference_type output_size = thrust::reduce(first1, last1);
 
  // scan the counts to obtain output offsets for each input element
  thrust::device_vector<difference_type> output_offsets(input_size, 
0);
  thrust::exclusive_scan(first1, last1, output_offsets.begin()); 
 
  // scatter the nonzero counts into their corresponding output 
positions
  thrust::device_vector<difference_type> output_indices(output_size, 
0);
  thrust::scatter_if
    (thrust::counting_iterator<difference_type>(0),
     thrust::counting_iterator<difference_type>(input_size),

https://github.com/thrust/thrust/blob/master/examples/expand.cu

     output_offsets.begin(),
     first1,
     output_indices.begin());
 
  // compute max-scan over the output indices, filling in the holes
  thrust::inclusive_scan
    (output_indices.begin(),
     output_indices.end(),
     output_indices.begin(),
     thrust::maximum<difference_type>());
 
  // gather input values according to index array
  // (output = first2[output_indices])
  OutputIterator output_end = output; thrust::advance(output_end, 
output_size);
  thrust::gather(output_indices.begin(),
                 output_indices.end(),
                 first2,
                 output);
 
  // return output + output_size
  thrust::advance(output, output_size);
  return output;
}

Counts:
    0:     3    1    0    0    7    3    2   14    4    6
   10:     0    2    1    5    3    0    5    1    6    2
   20:     0    0    9    3    2    1
   
Result of exclusive_scan:
    0:     0    3    4    4    4   11   14   16   30   34
   10:    40   40   42   43   48   51   51   56   57   63
   20:    65   65   65   74   77   79
   
Result of scatter_if:
    0:     0    0    0    1    4    0    0    0    0    0
   10:     0    5    0    0    6    0    7    0    0    0
   20:     0    0    0    0    0    0    0    0    0    0
   30:     8    0    0    0    9    0    0    0    0    0
   40:    11    0   12   13    0    0    0    0   14    0
   50:     0   16    0    0    0    0   17   18    0    0
   60:     0    0    0   19    0   22    0    0    0    0
   70:     0    0    0    0   23    0    0   24    0   25
   
Result of inclusive_scan with thrust::maximum():
    0:     0    0    0    1    4    4    4    4    4    4
   10:     4    5    5    5    6    6    7    7    7    7
   20:     7    7    7    7    7    7    7    7    7    7
   30:     8    8    8    8    9    9    9    9    9    9
   40:    11   11   12   13   13   13   13   13   14   14
   50:    14   16   16   16   16   16   17   18   18   18
   60:    18   18   18   19   19   22   22   22   22   22
   70:    22   22   22   22   23   23   23   24   24   25

Result of gather:

As in the CPU code, an exclusive scan converts item counts to output indices. The meaning of the rest of the 
implementation is somewhat obscure; I had to print the intermediate arrays to understand it. Scan is not the 
most natural primitive to use here, but when all you have is a hammer...

Temporary space to hold one integer per output is allocated and zeroed. scatter_if outputs the index of 
each input value to the start of the corresponding output run (the exclusive scan of count) if and only if the 
count is non-zero. Because zero counts cause consecutive output indices to match, multiple threads would 
attempt to store to the same address. scatter_if avoids this race condition by giving priority to the input 
with the non-zero count. The call to inclusive_scan specialized over the maximum functor fills the 
zeros with the largest indices encountered to the left. A gather loads input values at these indices and and 
stores them to the output, completing the expand.

Expand is a trivial function, but the use of scatter_if and inclusive_scan on maximum is far from 
an obvious solution. Adopting scan as the primary cooperatively-parallel function is more puzzle-solving 
than problem-solving. Scan is highly composable and Thrust lets you solve problems without writing new 
kernels. However, because you're trying to satisfy the logic of the scan function instead of targeting your 
specific needs, it may require non-intuitive design.

Expand with load-balancing search

MGPU introduces the Load-Balancing Search, a pattern that helps developers write elegant implementations 
of functions like expand. Although this search is available as a host-callable function, it is best invoked from 
inside a kernel. The MGPU idiom is less composable than Thrust's: users will need to write their own 
kernels. The solutions are much more intuitive, however, because the parallel demands of the architecture 
(i.e. scheduling) are satisfied in the partitioning phase, and the problem-specific logic is executed in a simple, 
sequential fasion. The implementation of IntervalExpand is a more direct solution to the expand 
problem: it loads input elements just once from global into shared memory and cooperatively fills the output 
arrays. The function makes only a single pass over the data and requires no auxiliary storage.

As problems become less scan-like, a gather/scatter/scan solution becomes more difficult to understand and 
express, and the value of composability decreases.

include/kernels/intervalmove.cuh

template<typename Tuning, typename IndicesIt, typename ValuesIt,
    typename OutputIt>
MGPU_LAUNCH_BOUNDS void KernelIntervalExpand(int destCount, 
    IndicesIt indices_global, ValuesIt values_global, int sourceCount, 
    const int* mp_global, OutputIt output_global) {
 
    typedef MGPU_LAUNCH_PARAMS Tuning;
    const int NT = Tuning::NT;
    const int VT = Tuning::VT;
    typedef typename std::iterator_traits<ValuesIt>::value_type T;
 
    union Shared {
        int indices[NT * (VT + 1)];
        T values[NT * VT];
    };

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/intervalmove.cuh
http://nvlabs.github.io/moderngpu/loadbalance.html

    __shared__ Shared shared;
    int tid = threadIdx.x;
    int block = blockIdx.x;
 
    // Compute the input and output intervals this CTA processes.
    int4 range = CTALoadBalance<NT, VT>(destCount, indices_global, 
sourceCount,
        block, tid, mp_global, shared.indices, true);
 
    // The interval indices are in the left part of shared memory 
(moveCount).
    // The scan of interval counts are in the right part 
(intervalCount).
    destCount = range.y - range.x;
    sourceCount = range.w - range.z;
 
    // Copy the source indices into register.
    int sources[VT];
    DeviceSharedToReg<NT, VT>(NT * VT, shared.indices, tid, sources);
 
    // Load the source fill values into shared memory. Each value is 
fetched
    // only once to reduce latency and L2 traffic.
    DeviceMemToMemLoop<NT>(sourceCount, values_global + range.z, tid,
        shared.values);
     
    // Gather the values from shared memory into register. This uses a 
shared
    // memory broadcast - one instance of a value serves all the 
threads that
    // comprise its fill operation.
    T values[VT];
    DeviceGather<NT, VT>(destCount, shared.values - range.z, sources, 
tid,
        values, false);
 
    // Store the values to global memory.
    DeviceRegToGlobal<NT, VT>(destCount, values, tid, output_global + 
range.x);
}

The first half of KernelIntervalMove is boilerplate. mp_global points to coarse-grained partitioning 
information computed in prior to the launch. CTALoadBalance uses this to subdivide the input and output 
ranges into intervals that fit exactly in CTA shared memory. 

    0:     0    0    0    1    4    4    4    4    4    4
   10:     4    5    5    5    6    6    7    7    7    7
   20:     7    7    7    7    7    7    7    7    7    7
   30:     8    8    8    8    9    9    9    9    9    9



   40:    11   11   12   13   13   13   13   13   14   14
   50:    14   16   16   16   16   16   17   18   18   18
   60:    18   18   18   19   19   22   22   22   22   22
   70:    22   22   22   22   23   23   23   24   24   25
   80:     0    3    4    4    4   11   14   16   30   34
   90:    40   40   42   43   48   51   51   56   57   63
  100:    65   65   65   74   77   79 

CTALoadBalance fills shared memory with two non-descending sequences: references to the generating 
source object for each destination object (in green), and the scan of source object item counts (in black). It's 
not coincidence that the array of source references is exactly the same as the array of gather indices 
computed by Thrust's expand function. With one boilerplate call we've already solved the problem! 
CTALoadBalance does as much work as the scatter_if and inclusive scan on maximum, yet never 
has to materialize intermediates into global memory, and so requires no storage.

The kernel moves on to load the source indices into register, freeing up shared memory. It cooperatively 
loads the referenced source values into shared memory (the 26 letters of the alphabet). Each thread uses the 
source indices to gather up to VT values, then stores them to the output array. This implementation is much 
faster than the scan-based version using Thrust. As developers gain more exposure to the patterns involved, 
code written in this idiom will become easy to read and write.

Because it delivers gather indices so readily, it may seem that CTALoadBalance is just part of an expand 
implementation. In fact, the load-balancing search is a highly general tool. Rather than solving the expand 
problem, it simply partitions for this class of problems. In the case of expand, the code that is executed per 
work-item is trivial (we just copy from the source to the destination). Interval move (a vectorized memcpy) 
and relational join (including full outer join) are implemented with the same partitioning boilerplate, but with 
additional problem-specific logic. The load-balancing search introduces a new pattern for GPU computing, 
one that I hope will push out the frontier and allow users to run more ambitious calculations.

Algorithms

The algorithms in this project primarily operate on multiple sorted inputs and produce one sorted output. The 
collection comprises an attempt at addressing the lack of data structures on GPU. Although we don't have 
self-balancing trees to serve as a data store, we can use multiset union and intersection to add and remove 
records by key. We can use bulk insert and bulk remove for fine-grained modification of arrays given sorted 
indices. Vectorized sorted search is a high-throughput search with desirable work-complexity characteristics 
to help locate records quickly.

Although many of these functions take sorted inputs, this is not an impractical requirement. The load-
balancing search pattern, introduced in the expand example, takes a sorted array, but this is typically 
generated by scanning a sequence of non-negative work-item counts.

Modern GPU covers ten functions:

1. Reduce and Scan   - Standard reduce and scan with a few special features. This shows where we've 
been and the following sections show where we're going with GPU computing.

2. Bulk Remove and Bulk Insert   - The first routines that use coarse-grained partitioning. Remove and 
insert items given a sorted sequence of indices. Merge Path partitioning is introduced to serve bulk 
insert.

3. Merge   - Uses Merge Path for fine-grained partitioning. The first routine that does not use scan. 

http://nvlabs.github.io/moderngpu/merge.html
http://nvlabs.github.io/moderngpu/bulkinsert.html
http://nvlabs.github.io/moderngpu/scan.html
http://nvlabs.github.io/moderngpu/join.cuh
http://nvlabs.github.io/moderngpu/intervalmove.cuh


Develops many patterns for the routines that follow.

4. Mergesort   - Recursively merge sorted sequences. Develops a useful and reusable CTA blocksort. 
Mergesort's throughput is usually beaten by radix sort for uniform random inputs, but the highly-
organized structure of mergesort allows for optimizations on conditioned inputs.

5. Segmented Sort and Locality Sort   - Segmented sort is probably the most versatile GPU sort. This 
allows us to sort many variable-length arrays in parallel. A list of segment head indices or an array of 
head flag bits is provided to define segment intervals. Segmented sort is fast: not only is segmentation 
supported for negligible cost, the function takes advantage of early-exit opportunities to improve 
throughput over vanilla mergesort. Locality sort is a useful variant that detects regions of approximate 
sortedness without requiring annotations.

6. Vectorized Sorted Search   - Run many concurrent searches where both the needles and haystack 
arrays are sorted. This input condition lets us recast the function as a sequential process resembling 
merge, rather than as a traditional binary search. Complexity improves from A log B to A + B, and 
because we touch every input, a search can retrieve not just the lower-bound of A into B but 
simultaneously the upper-bound of B into A, plus flags for all elements indicating if matches in the 
other array exist.

7. Load-Balancing Search   - Load-balancing search is a specialization of vectorized sorted search. It 
coordinates output items with the input objects that generated them. The CTA load-balancing search 
is a fundamental tool for partitioning irregular problems.

8. IntervalExpand and IntervalMove   - Schedule multiple variable-length fill, gather, scatter, or move 
operations. Partitioning is handled by load-balancing search. Small changes in problem logic enable 
different behaviors. These functions are coarse-grained counterparts to Bulk Remove and Bulk Insert.

9. Relational joins   - Sort-merge joins supporting inner, left, right, and outer variants. Uses vectorized 
sorted search to match keys between input arrays and load-balancing search to manage Cartesian 
products.

10.Multisets   - Replace Merge Path partitioning with the sophisticated Balanced Path to search for key-
rank matches. The new partitioning strategy is combined with four different serial set operations to 
support CUDA analogs of std::set_intersection, set_union, set_difference, and 
set_symmetric_difference.

http://nvlabs.github.io/moderngpu/sets.html
http://nvlabs.github.io/moderngpu/join.html
http://nvlabs.github.io/moderngpu/intervalmove.html
http://nvlabs.github.io/moderngpu/loadbalance.html
http://nvlabs.github.io/moderngpu/sortedsearch.html
http://nvlabs.github.io/moderngpu/segsort.html
http://nvlabs.github.io/moderngpu/mergesort.html


2. Performance

Occupancy and latency

Latency-oriented systems (CPUs) use large caches, branch prediction, and speculative fetching to avoid stalls 
on data dependencies. GPUs, by contrast, are throughput-oriented systems that use massive parallelism to 
hide latency. Occupancy is a measure of thread parallelism in a CUDA program. Instruction-level 
Parallelism is a measure of parallelism within threads. The higher the occupancy and ILP, the more 
opportunities an SM has to put compute and load/store units to work each cycle. Threads waiting on data 
dependencies and barriers are taken out of consideration until their hazards are resolved.

Kernels may be limited by DRAM to L2 cache bandwidth, L2 to SM bandwidth, texture bandwidth, FFMA, 
DFMA, integer, or shared memory performance. In some sense, being limited by bandwidth is a victory 
condition for optimization: a routine is fully exploiting at least one capability of the device. Performance can 
then only be improved by redesigning the code to use less of the limiting resource.

More often, and especially for codes that involve many subsystems, a kernel is latency limited. This state of 
under-occupancy occurs when there is insufficient parallelism to hide instruction latency. Many highly-
optimized routines become latency limited as work efficiency improves. As code becomes leaner the GPU 
has more free units than it has instructions to execute, a consequence of optimizing away unnecessary 
operations.

Important: The performance of latency-limited kernels is difficult to reason about. Optimizations that reduce 
work (or improve work efficiency) might not improve—and could even hinder—throughput. Focus on 
reducing latency on the most-congested path.

There are five resource limits that cap occupancy: 

 sm_20 sm_30 sm_35

Max Threads (SM) 1536 2048 2048

Max CTAs (SM) 8 16 16

Shared Memory Capacity (SM) 48 KB 48 KB 48 KB

Register File Capacity (SM) 128 KB 256 KB 256 KB

Max Registers (Thread) 63 63 255

1. Max Threads - You may be under-occupied even with 100% occupancy (1536 or 2048 threads 
running concurrently per SM). This is likely caused by poor ILP: increase the program's parallelism 
by register blocking to process multiple elements per thread. In MGPU most kernels are register 
blocked with grain size VT. You may also want to reduce the CTA size, so that barriers don't stall as 
many threads: smaller CTAs lead to better overlapped execution than larger ones.

2. Max CTAs - If you launch small blocks, your occupancy will be constrained by a hardware limit on 
resident CTAs. On Kepler, blocks must be at least 128 threads wide to hit maximum occupancy (16 
CTAs/SM). Using smaller blocks reduces occupancy; larger blocks compromise overlapped 



execution. In the absence of performance data, start with blocks of 128 or 256 threads.

3. Shared Memory Capacity - Many optimized register-blocked kernels are limited by shared memory 
capacity. Fermi has a 2.7:1 ratio of register file to shared memory. Kepler SMs have higher arithmetic 
throughput and latency (the two often increase together), but hasn't increased shared memory 
capacity, giving a ratio of 5.3:1. At 100% occupancy, a thread has 32 registers (128 bytes) but only 24 
bytes of shared memory. Register blocking for eight values per thread, with shared memory sized to 
accommodate all the values at once (for key exchange and other critical operations), implies no higher 
than 1536 threads/SM occupancy (75%). More aggressive register blocking drops this further. 
Operating at less than maximum occupancy does not imply under-occupancy, as ILP may be 
sufficient to cover latencies.

4. Register File Capacity - Register file is more copious than shared memory, and in the inverted cache 
hierarchy that GPUs are designed with, it's larger than even L2 cache. Still, code may be limited by 
RF capacity. Do mental live analysis while writing your kernel to reduce register usage. If your kernel 
uses more registers than you expect, try re-ordering load and store procedures to move out results 
before reading more inputs.

5. Max Registers - sm_20 and sm_30 devices have a limit of 63 registers per thread. If the back-end 
code generator cannot fit the working set of the kernel into 63 registers, it provisions local memory 
(driver-managed global memory) to spill state. Kernels with spill assume additional latency. sm_35 
devices have access to 255 registers per thread. While this relieves a register pressure problem for 
many procedures, it may also cause an additional drop in occupancy. sm_30 kernels that are limited 
by RF capacity will run at 50% occupancy (63 registers/thread). The same kernel running on sm_35 
may only achieve 12.5% occupancy, because each thread now consumes four times as much of the 
RF.

For all performance-critical kernels, compile with -Xptxas="-v". This passes a request through the 
NVVM compiler to the PTX assembler to output register, shared memory, and spill information for all 
kernels on each target architecture.

ptxas : info : Compiling entry function 'KernelFoo' for 'sm_20' 
ptxas : info : Function properties for KernelFoo
      48 bytes stack frame, 48 bytes spill stores, 36 bytes spill loads
ptxas : info : Used 63 registers, 11264 bytes smem, 64 bytes cmem[0]

ptxas : info : Compiling entry function 'KernelFoo' for 'sm_35'
ptxas : info : Function properties for KernelFoo
      0 bytes stack frame, 0 bytes spill stores, 0 bytes spill loads
ptxas : info : Used 80 registers, 11264 bytes smem, 352 bytes cmem[0]

The above kernel is too-aggressively register blocked. It hits the 63 register limit and spills on Fermi, 
achieving 25% occupancy. The function on Fermi is limited by RF capacity - launching four 128-thread 
CTAs consumes the entire register file.

Because of sm_35's per-thread register increase, the same code doesn't spill on GK110 Kepler. Thanks to the 
doubled RF capacity, it not limited by that, either. However, the code still runs at only 25% occupancy, 
because it's limited by shared memory capacity. Each CTA uses 11KB of shared memory, and since the SMs 
only have 48KB to share, only four 128-thread CTAs may be scheduled per SM (25% occupancy).

ptxas : info : Compiling entry function 'KernelFoo' for 'sm_20'
ptxas : info : Function properties for KernelFoo
      0 bytes stack frame, 0 bytes spill stores, 0 bytes spill loads
ptxas : info : Used 48 registers, 6144 bytes smem, 64 bytes cmem[0]

http://nvlabs.github.io/moderngpu/http%7C//en.wikipedia.org/wiki/Live_variable_analysis


ptxas : info : Compiling entry function 'KernelFoo' for 'sm_35'
ptxas : info : Function properties for KernelFoo
      0 bytes stack frame, 0 bytes spill stores, 0 bytes spill loads
ptxas : info : Used 48 registers, 6144 bytes smem, 352 bytes cmem[0]

Reducing the grain size VT improves occupancy by tackling less work per thread, requiring less state, and 
thereby consuming less resource. Five CTAs per SM are scheduled on Fermi - the kernel is RF capacity 
limited (five kernel use 30720/32768 registers). Both sm_30 and sm_35 fare better here. Eight CTAs are 
scheduled per SMX, limited by shared memory capacity (eight CTAs use all 49152 bytes).

Important: If a kernel spills even after you decrease grain size, you may be inadvertently dynamically 
indexing into an array that you intended to have reside in register. Use only literals, constant expressions, and 
unrolled loop iterators to index into register arrays. A compiler warning about an "unknown pragma" that 
refers back to a #pragma unroll attribute indicates that some construct is preventing the loop from 
unrolling, turning static indexes into dynamic ones, and likely causing spill. Although spilling may help 
performance by increasing occupancy in complex kernels, you should never allow spill that's caused by 
inadvertent dynamic indexing; this always hurts performance.

Launch bounds

CTA size and shared memory consumption are specified by the programmer; these are easily adjusted. RF 
usage, on the other hand, is a consequence of choices made by the register allocator in the back-end code 
generator. The __launch_bounds__ kernel attribute gives the user more control over occupancy by 
providing a cap on per-thread register usage. Tag the kernel with the CTA size and the desired number of 
CTAs per SM. The code generator now caps register usage by re-ordering instructions to reduce live 
variables. It spills the overflow.

ptxas : info : Compiling entry function 'KernelFoo' for 'sm_20'
ptxas : info : Function properties for KernelFoo
      40 bytes stack frame, 40 bytes spill stores, 24 bytes spill loads
ptxas : info : Used 36 registers, 6144 bytes smem, 64 bytes cmem[0]

ptxas : info : Compiling entry function 'KernelFoo' for 'sm_35'
ptxas : info : Function properties for KernelFoo
      0 bytes stack frame, 0 bytes spill stores, 0 bytes spill loads
ptxas : info : Used 48 registers, 6144 bytes smem, 352 bytes cmem[0]

The previous tuning of KernelFoo was tagged with __launch_bounds__(128, 7) to guarantee that 
7 CTAs run on each SM. The function now spills on Fermi, but uses only 36 registers per thread 
(32256/32768 registers per SM). The generated code is unchanged on Kepler, which remains limited to 8 
CTAs/SMX by shared memory capacity.

Getting more performance from MGPU

Most MGPU kernels are parameterized over NT (number of threads per CTA) and VT (values per thread, the 
grain size). The product of these two, NV (number of values per CTA), is the tile size. Increasing tile size 
amortizes the cost of once-per-thread and once-per-CTA operations, improving work efficiency. On the other 
hand, increasing grain size consumes more shared memory and registers, reducing occupancy and the ability 
to hide latency.

The constants NT, VT, and the __launch_bounds__ argument OCC (for occupancy, the minimum 
number of CTAs per SM) are tuning parameters. Finding optimal tuning parameters is an empirical 



process. Different hardware architectures, data types, input sizes and distributions, and compiler versions all 
effect parameter selection. User-tuning of MGPU library functions may improve throughput by 50% 
compared to the library's hard-coded defaults.

MGPU library functions have hard-coded and somewhat arbitrary parameters. The tuning space for type- and 
behavior-parameterized kernels is simply too large to explore in a project with Modern GPU's goals and 
scope.

// Copy and modify the host function to expose parameters for easier 
tuning.
template<int NT, int VT, int OCC, typename GatherIt, typename ScatterIt, 
    typename InputIt, typename OutputIt>
MGPU_HOST void IntervalMove2(ScatterIt scatter_global, const int* 
scan_global,
    int intervalCount, int moveCount, InputIt input_global, 
    OutputIt output_global, CudaContext& context) {
    // Parameters NT, VT, and OCC are passed in and override the host 
    // defaults.
      ...

Identify performance-critical routines and the contexts from which they are invoked. Copy the MGPU host 
functions that launch the relevant kernels and edit them to expose tuning parameters to the caller. Run the 
code on actual data and deployment hardware through the included benchmark programs, testing over a 
variety of parameters, to understand the performance space. Use the optimal settings to create specialized 
entry points to get the best throughput from your GPU.

Important: The omnipresent grain size parameter VT is almost always an odd integer in MGPU code. This 
choice allows us to step over bank-conflict issues that would otherwise need to be resolved with padding and 
complicated index calculations. Kernels execute correctly on even grain sizes, but expect diminished 
performance on these. Omit even grain sizes when searching the tuning space.



Mergesort tuning benchmark from tests/benchmarklaunchbox.cu

We've benchmarked MGPU's keys-only mergesort on GTX 480 and GTX Titan, specialized for both 32- and 
64-bit integer types. You can see the function's sensitivity to tuning parameters. Note the best-performing 
configurations for each device and data-type:

 GTX 480 (Fermi) GTX Titan (Kepler)

32-bit int 128x23 256x11

64-bit int 128x11 256x5

Parallelism (occupancy) decreases and work-efficiency increases as the grain size goes up. Kepler parts 
require much higher occupancy than Fermi to reach top performance—the SM is much wider (6 FFMA units 
up from 2 on Fermi), but per-SM shared memory remains the same at 48KB, badly underoccupying the 
device at larger grain sizes. Because Fermi runs well at lower occupancy, it benefits from scheduling 128-
thread blocks, even at the cost of an additional merge pass. Smaller blocks improve effective occupancy by 
stalling fewer threads at a synchronization.

Due to the GPU's raw power even untuned codes run excellently, often an order-of-magnitude beyond what 
you can achieve on the CPU. Tuning benchmarks may be the easiest way to squeeze 20-30% out of your 
code. It also informs the programmer of the performance landscape, leading to more productive profiling if 
they decide to go the extra mile. It's always good to tune before profiling so that you don't waste time 
optimizing code just to get to a local maxima.

Important: The hard-coded tuning parameters for MGPU functions were selected for decent performance on 
large arrays of 32- and 64-bit data types running on Kepler architecture. Users targeting Fermi devices may 
want to increase grain sizes significantly, as codes on that generation run better at lower occupancies. In all 

https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarklaunchbox.cu


cases, finding optimal tuning parameters for specific data types and distributions, input sizes, and 
architectures, is the easiest way to improve performance of well-written code.

LaunchBox

Once suitable tuning parameters have been discovered, specialize the host functions over data type (with 
function overloading or partial template specialization) and input size (with a runtime check and multiple 
launches). Specializing tuning parameters for different device architectures is more difficult. 

The NVCC front-end strips out __device__ and __host__ tags. It calls the host compiler (Visual C++ 
or GCC) once for the host code and invokes the NVVM back-end once for each architecture. If we were to 
specialize tuning for each device by branching over compute capability (determined at runtime with 
cudaDeviceProp::major and minor) and making a different chevron launch for each one, we'd end 
up compiling the cross product of all tunings over all architectures. Even though an sm_20 client will never 
actually launch kernels specialized with sm_30 or sm_35 tunings, that code would still be included with your 
binary because the multi-pass build process can't use dead code elimination on launches contingent on a 
runtime check.

CUDA defines a __CUDA_ARCH__ macro for the compute capability of the architecture that is currently 
being built in the multi-pass system. Although this macro is not available to the host at runtime, it can be 
used from inside a kernel's source to allow the kernel to change its own behavior. The macro is typically used 
to let a kernel opt into features that aren't available on all devices: e.g. the popc instruction counts the set 
bits in word, replacing a loop, when __CUDA_ARCH__ >= 200. We use this macro to guide the build 
process and only generate device code for the architectures it will run on.

MGPU introduces LaunchBox, a structure that specializes kernels for different compute capabilities without 
generating the cross product of each parameterization and architecture. Think of LaunchBox as concatenating 
the tuning parameters for all architectures into a single type and specializing the kernel over that type. The 
kernel reads the relevant parameters from the LaunchBox and changes its own implementation accordingly.

tests/testlaunchbox.cuh

// LaunchBox-specialized kernel Foo uses MGPU_LAUNCH_BOUNDS kernel to 
control
// register usage. The typename Tuning is obligatory, and must be 
chosen for 
// the MGPU_LAUNCH_BOUNDS and MGPU_LAUNCH_PARAMS macros to work.
template<typename Tuning>
MGPU_LAUNCH_BOUNDS void Foo() {
    typedef MGPU_LAUNCH_PARAMS Params;
    if(!blockIdx.x && !threadIdx.x)
        printf("Launch Foo<<<%d, %d>>> with NT=%d VT=%d OCC=%d\n", 
            gridDim.x, blockDim.x, Params::NT, Params::VT, 
Params::OCC);
}
 
// Use the built-in LaunchBoxVT type to specialize for NT, VT, and 
OCC.
void LaunchFoo(int count, CudaContext& context) {
    typedef LaunchBoxVT<
        128, 7, 4,          // sm_20  NT=128, VT=7,  OCC=4

https://github.com/NVlabs/moderngpu/blob/master/tests/testlaunchbox.cu

        256, 9, 5,          // sm_30  NT=256, VT=9,  OCC=5
        256, 15, 3          // sm_35  NT=256, VT=15, OCC=3
    > Tuning;
     
    // GetLaunchParamaters returns (NT, VT) for the arch vesion of the 
provided
    // CudaContext. The product of these is the tile size.
    int2 launch = Tuning::GetLaunchParams(context);
    int NV = launch.x * launch.y;
    int numBlocks = MGPU_DIV_UP(count, NV);
 
    Foo<Tuning><<<numBlocks, launch.x>>>();
}

Opt into LaunchBox by templating your kernel over Tuning (Tuning is an obligatory name). The 
MGPU_LAUNCH_BOUNDS macro, which includes the __global__ tag, generates the 
__launch_bounds__ attribute with the NT and OCC parameters specified at the launch site. The macro 
uses __CUDA_ARCH__ to discriminate between compute capabilities, binding to the static values 
corresponding to the currently-executing NVVM compiler pass. Typedef the MGPU_LAUNCH_PARAMS 
macro to access the tuning parameters inside the kernel.

Use LaunchBox to specialize the mergesort benchmark above:

• Specialize 32-bit mergesort for Fermi and Kepler with LaunchBoxVT<128, 23, 0, 256, 
11, 0>.

• Specialize 64-bit mergesort for Fermi and Kepler with LaunchBoxVT<128, 11, 0, 256, 5, 
0>.

These tunings for sm_20 and sm_30 are inherited by other platforms (sm_21 inherits sm_20; sm_35+ inherits 
sm_30). We choose not to constrain register count by leaving the occupancy parameter zero.

Most users will only need the LaunchBoxVT structure, a specialization that makes tuning more succinct. 
Specialize this template over (NT, VT, OCC) for sm_20, sm_30, and sm_35. Default template arguments 
inherit parameters from the earlier-generation architecture, so LaunchBoxVT<128, 7> is equivalent to 
LaunchBoxVT<128, 7, 0, 128, 7, 0, 128, 7, 0>.

Use the static method GetLaunchParams, passing the CudaContext object, to return the (NT, VT) 
arguments for the compute capability of the currently-selected device. The product of these is the tile size. 
Use it to calculate the launch's grid size. Finally, specialize the kernel over Tuning and launch the grid with 
launch.x (NT) threads.

tests/testlaunchbox.cuh

// LaunchBox-specialized kernel Bar introduces its own set of launch 
parameters.
template<int NT_, int VT_, int OCC_, int P1_, typename T1_>
struct BarParams {
    enum { NT = NT_, VT = VT_, OCC = OCC_, P1 = P1_ };
    typedef T1_ T1;
};
template<typename Tuning>

https://github.com/NVlabs/moderngpu/blob/master/tests/testlaunchbox.cu

MGPU_LAUNCH_BOUNDS void Bar() {
    typedef MGPU_LAUNCH_PARAMS Params;
    if(!blockIdx.x && !threadIdx.x) {
        printf("Launch Bar<<<%d, %d>>> with NT=%d VT=%d OCC=%d\n",
            gridDim.x, blockDim.x, Params::NT, Params::VT, 
Params::OCC);
        printf("\t\tP1 = %d  sizeof(TT1) = %d\n", Params::P1, 
            sizeof(typename Params::T1));
    }   
}
 
void LaunchBar(int count, CudaContext& context) {
    typedef LaunchBox<
        BarParams<128, 7, 4, 20, short>,  // sm_20
        BarParams<256, 9, 5, 30, float>,  // sm_30
        BarParams<256, 15, 3, 35, double> // sm_35
    > Tuning;
    int2 launch = Tuning::GetLaunchParams(context);
 
    int nv = launch.x * launch.y;
    int numBlocks = MGPU_DIV_UP(count, nv);
    Bar<Tuning><<<numBlocks, launch.x>>>();
}

LaunchBoxVT inherits the more generic LaunchBox type and provides some syntactic sugar. LaunchBox 
takes three types as arguments and typedefs those to Sm20, Sm30, and Sm35. When devices based on the 
Maxwell architecture are released, LaunchBox will add additional typedefs. The LaunchBox technique 
inherits parameter tunings to avoid versioning difficulties. Although LaunchBox puts no restrictions on its 
specialization types, constants NT and VT must be included if the host code wishes to use 
LaunchBox::GetLaunchParams (the client can elect not to use this), and NT and OCC must be 
included to support MGPU_LAUNCH_BOUNDS (ditto).

GeForce GTX 570 : 1464.000 Mhz   (Ordinal 0)
15 SMs enabled. Compute Capability sm_20
FreeMem:    778MB   TotalMem:   1279MB.
Mem Clock: 1900.000 Mhz x 320 bits   (152.000 GB/s)
ECC Disabled

Launching Foo with 1000000 inputs:
Launch Foo<<<1117, 128>>> with NT=128 VT=7 OCC=4

Launching Bar with 1000000 inputs:
Launch Bar<<<1117, 128>>> with NT=128 VT=7 OCC=4
                P1 = 20  sizeof(TT1) = 2

Launching Foo and Bar prints the above. The host function correctly coordinates the launch with the 
statically- specialized kernel.

There is one small caveat: if LaunchBox is used to specialize kernels for architectures that are not compiled 
with -gencode on the command line, LaunchBox::GetLaunchParams could return a different set of 
tuning parameters than those that the kernel actually gets specialized over. If, for example, the program is 
compiled for targets sm_20 and sm_30 but is executed on an sm_35 device, the kernel that is launched would 
be for sm_30 (the largest targeted architectured not greater than the architecture of the device), however the 
host side would configure the launch with the tuning parameters for sm_35.



To properly coordinate between the static device-side and dynamic host-side interfaces, we implement 
GetLaunchParams so that it uses the highest targeted architecture not greater than the device's compute 
capability when selecting dynamic launch parameters.

include/device/launchbox.cuh

// Returns (NT, VT) from the sm version.
template<typename Derived>
struct LaunchBoxRuntime {
    static int2 GetLaunchParams(CudaContext& context) {
        int sm = context.CompilerVersion();
        if(sm < 0) {
            cudaFuncAttributes attr;
            cudaFuncGetAttributes(&attr, LBVerKernel);
            sm = 10 * attr.ptxVersion;
            context.Device().SetCompilerVersion(sm);
        }
 
        // TODO: Add additional architectures as devices are released.
        if(sm >= 350) 
            return make_int2(Derived::Sm35::NT, Derived::Sm35::VT);
        else if(sm >= 300) 
            return make_int2(Derived::Sm30::NT, Derived::Sm30::VT);
        else
            return make_int2(Derived::Sm20::NT, Derived::Sm20::VT);
    }
};

The first time GetLaunchParams is called, CudaDevice::CompilerVersion is unset, and we call 
cudaFuncGetAttributes on a place-holder method LBVerKernel. (Although it is never launched, 
taking the address of this kernel keeps it in our program.) This bit of CUDA introspection returns the target 
architecture of the compiled kernel (and presumably all other kernels in the program) through 
cudaFuncAttributes::ptxVersion. LaunchBox allows the CUDA runtime to coordinate the host 
and device sides of the call. All subsequent LaunchBox invocations are ready to go with the compiler 
version.

https://github.com/NVlabs/moderngpu/blob/master/include/device/launchbox.cuh


3. The Library

Framework

To ease development MGPU includes a sample framework, defined in util/mgpucontext.h. At the start of 
your program create a CudaContext object. This encapsulates an event, a timer, a stream, and an allocator. 
Allocations made through this context are recycled after being freed, reducing calls to cudaMalloc.

include/util/mgpucontext.h

ContextPtr CreateCudaDevice(int ordinal);
ContextPtr CreateCudaDevice(int argc, char** argv, bool printInfo = 
false);
 
ContextPtr CreateCudaDeviceStream(int ordinal);
ContextPtr CreateCudaDeviceStream(int argc, char** argv, 
    bool printInfo = false);

Call CreateCudaDevice to create a context on the default stream or CreateCudaDeviceStream to 
create a context on the new stream. The (argc, argv) overloads parse the command-line arguments for a 
device ordinal. You can pass true for printInfo to print device attributes:

int main(int argc, char** argv) {
    ContextPtr context = CreateCudaDevice(argc, argv, true);
    return 0;
}

GeForce GTX 480 : 1401.000 Mhz   (Ordinal 1)
15 SMs enabled. Compute Capability sm_20
FreeMem:   1086MB   TotalMem:   1535MB.
Mem Clock: 1848.000 Mhz x 384 bits   (177.408 GB/s)
ECC Disabled

MGPU context and device objects are managed with the reference-counting pointer types ContextPtr and 
DevicePtr. MGPU-allocated memory is reference counted with intrusive_ptr< 
CudaDeviceMem<type> > which is bound to the MGPU_MEM(type) macro for ease of use.

#include "moderngpu.cuh"
 
using namespace mgpu;
 
int main(int argc, char** argv) {
    ContextPtr context = CreateCudaDevice(argc, argv);
 
    MGPU_MEM(uint) data = context->Malloc<uint>(1000);
 
    MGPU_MEM(int) a = context->FillAscending<int>(50, 0, 5);
    MGPU_MEM(float) b = context->GenRandom<float>(50, 0.0f, 10.0f);
    MGPU_MEM(double) c = context->SortRandom<double>(50, 0.0, 20.0);

https://github.com/NVlabs/moderngpu/blob/master/include/util/mgpucontext.h
https://github.com/NVlabs/moderngpu/blob/master/include/util/mgpucontext.h


13
14
15
16
17
18
19
20
21
22
23
24

 
    printf("A:\n");
    PrintArray(*a, "%6d", 10);
 
    printf("\nB:\n");
    PrintArray(*b, "%6.2lf", 10);
 
    printf("\nC:\n");
    PrintArray(*c, "%6.2lf", 10);
 
    return 0;
}

A:
    0:       0      5     10     15     20     25     30     35     40     45
   10:      50     55     60     65     70     75     80     85     90     95
   20:     100    105    110    115    120    125    130    135    140    145
   30:     150    155    160    165    170    175    180    185    190    195
   40:     200    205    210    215    220    225    230    235    240    245

B:
    0:    8.15   1.35   9.06   8.35   1.27   9.69   9.13   2.21   6.32   3.08
   10:    0.98   5.47   2.78   1.88   5.47   9.93   9.58   9.96   9.65   9.68
   20:    1.58   7.26   9.71   9.81   9.57   1.10   4.85   7.98   8.00   2.97
   30:    1.42   0.05   4.22   1.12   9.16   6.40   7.92   8.78   9.59   5.04
   40:    6.56   7.98   0.36   3.61   8.49   2.12   9.34   6.81   6.79   3.99

C:
    0:    0.64   0.69   0.73   0.92   1.02   1.94   2.50   2.52   2.98   3.42
   10:    3.48   3.74   4.20   5.54   6.04   6.33   6.34   7.63   7.84   8.17
   20:    8.44   8.77   8.91   9.16   9.50   9.75   9.80   9.81  12.93  13.11
   30:   13.27  13.90  14.12  14.19  14.81  14.86  15.09  15.15  15.28  15.31
   40:   15.88  15.90  15.95  16.15  16.44  16.47  17.45  18.42  19.00  19.88

CudaContext::Malloc allocates memory from its caching allocator. The class supports a variety of 
methods to fill device memory with data to accelerate testing and debugging. FillAscending, 
GenRandom, and SortRandom are demonstrated above. PrintArray prints CudaDeviceMem arrays 
to the console using printf-style format specifiers.

When MGPU_MEM-wrapped objects fall out of scope, the underlying device memory is inserted into a 
weighted least-recently-used cache. Subsequent queries check the pool and reuse allocations of a similar size 
before calling cudaMalloc. Once a program gets hot, the client can make small requests from 
CudaContext with high confidence that the call will return immediately.

Users can opt-out of the default caching allocator by deriving CudaAlloc and providing their own 
implementation, or simply by using CudaAllocSimple, which calls cudaFree immediately on device 
memory falling out of scope.

include/util/mgpucontext.h

https://github.com/NVlabs/moderngpu/blob/master/include/util/mgpucontext.h
http://en.wikipedia.org/wiki/Cache_algorithms#Least_Recently_Used

class CudaAlloc : public CudaBase {
public:
    virtual cudaError_t Malloc(size_t size, void** p) = 0;
    virtual bool Free(void* p) = 0;
    virtual ~CudaAlloc() { }
 
    const CudaDevice& Device() const { return *_device; }
    CudaDevice& Device() { return *_device; }
protected:
    CudaAlloc(CudaDevice* device) : _device(device) { }
    DevicePtr _device;
};

CudaAlloc is an interface that defines two abstract methods for users to implement: Malloc allocates 
size bytes and returns the pointer in p. Free releases memory allocated by Malloc.

1
2
3
4

int main(int argc, char** argv) {
    ContextPtr context = CreateCudaDevice(argc, argv, true);
    AllocPtr standardAlloc(new CudaAllocSimple(&context->Device()));
    context->SetAllocator(standardAlloc);

Instantiate your allocator and associate it with the device context with CudaContext::SetAllocator. 
The provided caching allocator is not optimal for all applications; use the simple allocator to get back to a 
baseline.

1
2
3
4
5
6
7
8
9

int main(int argc, char** argv) {
    ContextPtr context = CreateCudaDevice(argc, argv, true);
 
    // Cast CudaAlloc* to CudaAllocBuckets*
    CudaAllocBuckets* buckets = dynamic_cast<CudaAllocBuckets*>
        (context->GetAllocator());
 
    // Set the capacity of the LRU cache to 500MB.
    buckets->SetCapacity(500000000);

You can set the capacity of the LRU cache dynamically. CudaContext::GetAllocator returns a 
CudaContext* pointer to the currently-selected allocator. Because we know it's a caching allocator, we 
use RTTI's dynamic_cast to retrieve a CudaAllocBuckets* pointer. We call SetCapacity with a 
request of 500MB to set the cache size. If the size of an allocation request plus the size of items allocated in 
the cache exceeds 500MB, the caching allocator frees the least-recently-used requests to make space for the 
new memory.

include/kernels/bulkremove.cuh

    KernelBulkRemove<Tuning><<<numBlocks, launch.x, 0, 
context.Stream()>>>(
        source_global, sourceCount, indices_global, indicesCount, 
        partitionsDevice->get(), dest_global);

The context object attempts to support CUDA streams in as non-obtrusive a manner as possible. All MGPU 
host functions take a CudaContext object by reference and pass the stream handle to the launch chevrons. 
This enqueues the kernel launch into the stream that attached to the context.

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/bulkremove.cuh


Some MGPU functions—namely reduce, join, and some variants of scan and vectorized sorted search—use 
cudaMemcpyDeviceToHost to move kernel outputs into host memory. This is a synchronizing function; 
it will cause the thread to wait on the transfer, preventing it from queueing launches on other streams. If this 
creates scheduling inefficiences, the programmer can split apart the host function, use cudaMemcpyAsync 
to asynchronously move data into CPU-pinned memory, and overlap scheduling of operations on other 
threads. This is an invasive and application-specific way to program, so it is not directly support by the 
MGPU library.

Load/store functions

MGPU functions are aggressively register-blocked. Register blocking amortizes per-CTA and per-thread 
costs by increasing the number of items processed per thread. To improve clarity and reduce errors, common 
routines for moving portions of data between memory spaces (global memory, shared memory, and register) 
have been factored into functions in the include/device/loadstore.cuh header.

The common template argument VT is the kernel's grain size; it specifies the number of values processed per 
thread. The argument NT is the number of threads in the CTA.

Most of these functions operate in strided order, in which elements are assigned to threads according to NT 
* i + tid, where i is the index of the element in the register and tid is the thread ID. Data should be loaded and 
stored in strided order, as this organizes warp transfers into cache lines, which maximizes data throughput.

Many MGPU algorithms work with data in thread order, in which elements are assigned to threads 
according to VT * tid + i. Each thread has access to VT consecutive elements which makes performing 
sequential operations like scan and merge very easy. However data should not be loaded or stored to global 
memory in thread order, as warp transfers would touch VT different cache lines, wasting memory bandwidth.

By choosing an odd number for VT we avoid bank conflicts that would otherwise be incurred when re-
ordering data between strided and thread orders. Within a warp, all banks (VT * tid + i) % 32 are accessed 
exactly once for each step i when VT is odd. If VT is a power-of-two, you can expect VT-way conflicts at 
each step.

Load/store function prototypes are found in mgpudevice.cuh. Most functions have names matching the 
pattern Device[Source]To[Dest]:

include/mgpudevice.cuh

76
77
78
79
80
81
82

// For 0 <= i < VT: 
//      index = NT * i + tid;
//      if(index < count) reg[i] = data[index];
// Synchronize after load.
template<int NT, int VT, typename InputIt, typename T>
MGPU_DEVICE void DeviceSharedToReg(int count, InputIt data, int tid, 
    T* reg, bool sync = true);

Functions of this form are parameterized over NT and VT arguments—these are typically communicated to 
the kernel using the LaunchBox mechanism. The first argument is the count of items to move across the 
entire CTA. If NT * VT == count, an optimized implementation may be used which eliminates per-item 
predication to reduce latency and promote parallelism. The second argument is the source data, and its 
memory space should match the [Source] part of the function name. The third argument is the thread ID. The 
fourth argument is the destination data and its memory space should match the [Dest] part of the function 

http://nvlabs.github.io/moderngpu/performance.html#launchbox
https://github.com/NVlabs/moderngpu/blob/master/include/mgpudevice.cuh
https://github.com/NVlabs/moderngpu/blob/master/include/device/loadstore.cuh


name. The final argument is used to hit a __syncthreads after the operation. Data movement functions 
with Shared in the name synchronize by default; other functions do not.

Data can be loaded from shared memory into registers in thread order with DeviceSharedToThread. 
Data can be stored to shared from registers in thread order with DeviceThreadToShared. A common 
practice is to:

1. Cooperatively load data into register in strided order and store to shared memory with 
DeviceGlobalToShared.

2. Read out values in thread order into register with DeviceSharedToThread.

3. Operate on data that is now sequentially ordered by thread. Scan works in this manner.

4. Store results from register in thread order into shared memory with DeviceThreadToShared.

5. Cooperatively load data from shared memory into register in strided order and store to global 
memory with DeviceSharedToGlobal.

Regimented application of these utility functions to move data between global memory, shared memory, and 
register helps highlight the novel aspects of the kernel (the stuff in step 3).

Task range

Work can be divided into CTAs with two strategies:

1. Process one tile per CTA and make the grid a function of your data size. The hardware schedules 
blocks and fine-grained load balancing keeps the SMs busy.

2. Evenly divide your data into a fixed number of CTAs (ideally a function of device width) and 
dynamically loop through tiles. Oversubscribe CTAs to SMs to remedy warp-biasing artifacts of the 
hardware. This technique requires more tweaking than the one-tile-per-CTA division, but may also 
enable algorithmic improvements. Atomic counters can be integrated to facilitate work-stealing, 
resulting in the "persistent CTAs" idiom of scheduling.

I prefer the first strategy for work division unless there is a compelling algorithmic opportunity to assume 
tile-scheduling responsibilities. The reduce kernel, covered on the next page, provides such an opportunity. 
Because the version of KernelReduce which exploits the commutative property only runs a CTAReduce 
at the end of each CTA rather than the end of each tile, it makes sense to amortize this cost by running many 
tiles through each CTA.

Multiple tiles can be evenly balanced over a fixed number of CTAs using an easy trick which requires no 
division on the device (division is very slow on the GPU).

include/kernels/reduce.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/reduce.cuh
http://nvlabs.github.io/moderngpu/scan.html

// We have a large reduction so balance over many CTAs. Launch up to 
25
// per CTA for oversubscription.
const int NT = 128;
const int VT = 9;
typedef LaunchBoxVT<NT, VT> Tuning;
int2 launch = Tuning::GetLaunchParams(context);
const int NV = launch.x * launch.y;
 
int numTiles = MGPU_DIV_UP(count, NV);
int numBlocks = std::min(context.NumSMs() * 25, numTiles);
int2 task = DivideTaskRange(numTiles, numBlocks);
 
// Reduce on the GPU.
MGPU_MEM(T) reductionDevice = context.Malloc<T>(numBlocks);
KernelReduce<Tuning><<<numBlocks, launch.x, 0, context.Stream()>>>(
    data_global, count, task, reductionDevice->get(), op);

Data is divided into tiles of size NV (number of Values per Thread). The number of CTAs is set to a small 
multiple of the number of SMs or the number of tiles, whichever is smaller. We use the C runtime function 
div() to compute both the quotient and remainder of the tile count over the CTA count. Each CTA 
reconstructs its range of tiles from this pair of terms. The number of tiles assigned to each CTA differ by no 
more than one, with the earlier CTAs being assigned the additional tiles.

After setting tuning parameters, the host calls LaunchBox::GetLaunchParams to retrieve (NT, VT). 
The product of these is the tile size, and it's used to compute the number of tiles in the launch. A call to 
DivideTaskRange evenly distributes tiles over CTAs. The returned task structure is passed to the kernel.

include/device/deviceutil.cuh

MGPU_HOST int2 DivideTaskRange(int numItems, int numWorkers) {
    div_t d = div(numItems, numWorkers);
    return make_int2(d.quot, d.rem);
}
 
MGPU_HOST_DEVICE int2 ComputeTaskRange(int block, int2 task) {
    int2 range;
    range.x = task.x * block;
    range.x += min(block, task.y);
    range.y = range.x + task.x + (block < task.y);
    return range;
}
 
MGPU_HOST_DEVICE int2 ComputeTaskRange(int block, int2 task, int 
blockSize,
    int count) {
    int2 range = ComputeTaskRange(block, task);
    range.x *= blockSize;
    range.y = min(count, range.y * blockSize);
    return range;
}

https://github.com/NVlabs/moderngpu/blob/master/include/device/deviceutil.cuh


NumTiles = 1234.    NumCTAs = 20.    d.quot = 61.    d.rem = 14.
CTA   begin   end  count
   0: (   0,   62:   62)
   1: (  62,  124:   62)
   2: ( 124,  186:   62)
   3: ( 186,  248:   62)
   4: ( 248,  310:   62)
   5: ( 310,  372:   62)
   6: ( 372,  434:   62)
   7: ( 434,  496:   62)
   8: ( 496,  558:   62)
   9: ( 558,  620:   62)
  10: ( 620,  682:   62)
  11: ( 682,  744:   62)
  12: ( 744,  806:   62)
  13: ( 806,  868:   62)
  14: ( 868,  929:   61)
  15: ( 929,  990:   61)
  16: ( 990, 1051:   61)
  17: (1051, 1112:   61)
  18: (1112, 1173:   61)
  19: (1173, 1234:   61)

A problem with 1234 tiles and 20 CTAs calls div() to compute the quot/rem pair(61, 14), which is passed 
to the kernel as task. Call DivideTaskRange to handle the division and casting; the int2 it returns 
may be passed directly to the kernel.

ComputeTaskRange returns the interval of tiles for the block. The four-argument overload accepts the 
block size and item count to compute a range in values rather than tiles. This is called from inside a kernel to 
establish the kernel's operating range.

include/kernels/reduce.cuh

// Run a high-throughput reduction over multiple CTAs. Used as the 
upsweep phase
// for global reduce and global scan.
template<typename Tuning, typename InputIt, typename Op>
MGPU_LAUNCH_BOUNDS void KernelReduce(InputIt data_global, int count, 
    int2 task, typename Op::value_type* reduction_global, Op op) {
 
    typedef MGPU_LAUNCH_PARAMS Params;
    const int NT = Params::NT;
    const int VT = Params::VT;
    const int NV = NT * VT;
 
    KERNEL BOILERPLATE HERE
    ...
 
    int tid = threadIdx.x;
    int block = blockIdx.x;
    int first = VT * tid;
 
    int2 range = ComputeTaskRange(block, task, NV, count);
 

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/reduce.cuh

    // total is the sum of encountered elements. It's undefined on the 
first 
    // loop iteration.
    value_type total = op.Extract(op.Identity(), -1);
    bool totalDefined = false;
     
    // Loop through all tiles defined by ComputeTaskRange.
    while(range.x < range.y) {
        int count2 = min(NV, count - range.x);
 
        // Read tile data into register.
        input_type inputs[VT];
        DeviceGlobalToReg<NT, VT>(count2, data_global + range.x, tid, 
inputs);
         
        KERNEL BODY HERE
        ...
         
        // Update range pointer to advance to next tile.
        range.x += NV;
        totalDefined = true;
    }

On the device side, MGPU_LAUNCH_PARAMS identifies the nested tuning-parameters type for the 
compute capability. The tile size, NV, is the product of the number of threads per CTA, NT, and the number 
of values per thread, VT. ComputeTaskRange takes the task structure and returns the range of tiles for the 
CTA to process.



4. Reduce and Scan
Reduce and scan are core primitives of parallel computing. This implementation supports user-defined binary 
operations and defines an interface for handling different input, intermediate, and result types.

Algorithmically, reduce and scan show where we've been with GPU computing—the following sections 
show where we're going.

Benchmark and usage

Scan benchmark from tests/benchmarkscan.cu

Scan demonstration from tests/demo.cu

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarkscan.cu

void DemoScan(CudaContext& context) {
    printf("\n\nREDUCTION AND SCAN DEMONSTRATION:\n\n");
 
    // Generate 100 random integers between 0 and 9.
    int N = 100;
    MGPU_MEM(int) data = context.GenRandom<int>(N, 0, 9);
    printf("Input array:\n");
    PrintArray(*data, "%4d", 10);
     
    // Run a global reduction.
    int total = Reduce(data->get(), N, context);
    printf("Reduction total: %d\n\n", total);
     
    // Run an exclusive scan.
    total = Scan<MgpuScanTypeExc>(data->get(), N, context);
    printf("Exclusive scan:\n");
    PrintArray(*data, "%4d", 10);
    printf("Scan total: %d\n", total);
}

REDUCTION AND SCAN DEMONSTRATION:

Input array:
    0:     8    1    9    8    1    9    9    2    6    3
   10:     0    5    2    1    5    9    9    9    9    9
   20:     1    7    9    9    9    1    4    7    8    2
   30:     1    0    4    1    9    6    7    8    9    5
   40:     6    7    0    3    8    2    9    6    6    3
   50:     7    7    7    4    3    4    6    1    1    3
   60:     7    7    0    3    2    8    0    1    0    9
   70:     8    8    6    1    3    7    9    4    0    6
   80:     4    1    3    2    7    0    7    0    1    4
   90:     4    4    4    4    6    7    7    9    7    8
Reduction total: 492

Exclusive scan:
    0:     0    8    9   18   26   27   36   45   47   53
   10:    56   56   61   63   64   69   78   87   96  105
   20:   114  115  122  131  140  149  150  154  161  169
   30:   171  172  172  176  177  186  192  199  207  216
   40:   221  227  234  234  237  245  247  256  262  268
   50:   271  278  285  292  296  299  303  309  310  311
   60:   314  321  328  328  331  333  341  341  342  342
   70:   351  359  367  373  374  377  384  393  397  397
   80:   403  407  408  411  413  420  420  427  427  428
   90:   432  436  440  444  448  454  461  468  477  484
Scan total: 492



Max-index benchmark from tests/benchmarkscan.cu

Max-index demonstration from tests/demo.cu

void DemoMaxIndex(CudaContext& context) {
    printf("\n\nMAX-INDEX DEMONSTRATION:\n\n");
 
    // Generate 100 random integers between 0 and 9.
    int N = 100;
    MGPU_MEM(int) data = context.GenRandom<nt>(N, 0, N);
    printf("Input array:\n");
    PrintArray(*data, "%4d", 10);
 
    // Run a global reduction.
    typedef ScanOpMaxIndex<int> Op;
    Op::Pair pair = Reduce(data->get(), N, Op(), context);
    printf("Max-index reduction: %d at position %d\n", pair.value, 
pair.index);
 
    // Run an exclusive scan.
    Scan<MgpuScanTypeExc>(data->get(), N, data->get(), Op(), &pair, 
false,
        context);
    printf("\nMax-index exclusive scan:\n");
    PrintArray(*data, "%4d", 10);
    printf("Scan total: %d at position %d\n", pair.value, pair.index);
}

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarkscan.cu


MAX-INDEX DEMONSTRATION:

Input array:
    0:    27   71   68    0   66   71   16   65   12   46
   10:    50   78   96   57   34   88   59   81   22    1
   20:    75   82   25   82   51   94   70   41   89   42
   30:    96   58   55   15   14   76   15   23   26   81
   40:    84   99   25   33   82   30   24    1   93   21
   50:    35   91   19   85   25   96   62   78   47   99
   60:    35    6   83   80   59   60   55   74   92   70
   70:    28   68   76   39   76   56   38   21   57   53
   80:     7   40    5   35   53   59   78   35   94   97
   90:    13   15   57   39   47   39    1   73   34   39
Max-index reduction: 99 at position 41

Max-index exclusive scan:
    0:    -1    0    1    1    1    1    1    1    1    1
   10:     1    1   11   12   12   12   12   12   12   12
   20:    12   12   12   12   12   12   12   12   12   12
   30:    12   12   12   12   12   12   12   12   12   12
   40:    12   12   41   41   41   41   41   41   41   41
   50:    41   41   41   41   41   41   41   41   41   41
   60:    41   41   41   41   41   41   41   41   41   41
   70:    41   41   41   41   41   41   41   41   41   41
   80:    41   41   41   41   41   41   41   41   41   41
   90:    41   41   41   41   41   41   41   41   41   41
Scan total: 99 at position 41

Host functions

include/mgpuhost.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/mgpuhost.cuh

//////////////////////////////////////////////////////////////////////
//////////
// kernels/reduce.cuh
 
// Reduce input and return variable in host memory. Note that this 
calls 
// cudaMemcpyDeviceToHost, a synchronous operation that may interrupt 
queueing
// on streams.
template<typename InputIt, typename Op>
MGPU_HOST typename Op::value_type Reduce(InputIt data_global, int count, 
Op op, 
    CudaContext& context);
 
// T = std::iterator_traits<InputIt>::value_type.
// Reduce with Op = ScanOp<ScanOpTypeAdd, T>.
template<typename InputIt>
MGPU_HOST typename std::iterator_traits<InputIt>::value_type
Reduce(InputIt data_global, int count, CudaContext& context);
 
 
//////////////////////////////////////////////////////////////////////
//////////
// kernels/scan.cuh
 
// Scan inputs in device memory.
// MgpuScanType may be:
//      MgpuScanTypeExc (exclusive) or
//      MgpuScanTypeInc (inclusive).
// If total is non-zero, the reduction of the input is returned in 
host memory.
//      This incurs a synchronization so may not be appropriate for 
programs
//      using stream.
// If totalAtEnd is true, the reduction is stored at 
dest_global[count].
template<MgpuScanType Type, typename InputIt, typename OutputIt, 
typename Op>
MGPU_HOST void Scan(InputIt data_global, int count, OutputIt 
dest_global, Op op,
    typename Op::value_type* total, bool totalAtEnd, CudaContext& 
context);
 
// Specialization Scan:
// Returns the reduction as a variable in host memory.
// Uses Op = ScanOp<ScanOpTypeAdd, T>
// totalAtEnd = false
template<MgpuScanType Type, typename InputIt>
MGPU_HOST typename std::iterator_traits<InputIt>::value_type
Scan(InputIt data_global, int count, CudaContext& context);
 
// Specialization with MgpuScanTypeExc.



template<typename InputIt>
MGPU_HOST typename std::iterator_traits<InputIt>::value_type
Scan(InputIt data_global, int count, CudaContext& context);

Algorithm

Further Reading: For a detailed introduction to reduction networks read NVIDIA's Mark Harris on 
Optimizing Parallel Reduction in CUDA.

Reduce and scan (prefix sum) are core primitives in parallel computing. Reduce is the sum of elements in an 
input sequence:

Input:      1   7   4   0   9   4   8   8   2   4   5   5   1   7   1   1   5   2   7   
6
Reduction: 87

Scan generalizes this operator, reducing inputs for every interval from the start to the current element. 
Exclusive scan is the sum of all inputs from the beginning to the element before the current element. 
Inclusive scan is the exclusive scan plus the current element. You can convert from inclusive to exclusive 
scan with a component-wise subtraction of the input array, or by shifting the scan one element to the right:

Input:      1   7   4   0   9   4   8   8   2   4   5   5   1   7   1   1   5   2   7   
6
Exclusive:  0   1   8  12  12  21  25  33  41  43  47  52  57  58  65  66  67  72  74  
81
Inclusive:  1   8  12  12  21  25  33  41  43  47  52  57  58  65  66  67  72  74  81  
87

Note that the last element of inclusive scan is the reduction of the inputs.

Parallel evaluation of reduce and scan requires cooperative scan networks which sacrifice work efficiency to 
expose parallelism. Consider the cooperative scan network based on the Kogge-Stone adder. n inputs are 
processed in log(n) passes. On pass 0, element i - 1 (if in range) is added into element i, in parallel. On pass 
1, element i - 2 is added into element i. On pass 2, element i - 4 is added into element i, and so on.

Input:      1   7   4   0   9   4   8   8   2   4   5   5   1   7   1   1   5   2   7   
6

Inclusive scan network by offset:
    1:      1   8  11   4   9  13  12  16  10   6   9  10   6   8   8   2   6   7   9  
13
    2:      1   8  12  12  20  17  21  29  22  22  19  16  15  18  14  10  14   9  15  
20
    4:      1   8  12  12  21  25  33  41  42  39  40  45  37  40  33  26  29  27  29  
30
    8:      1   8  12  12  21  25  33  41  43  47  52  57  58  65  66  67  71  66  69  
75
   16:      1   8  12  12  21  25  33  41  43  47  52  57  58  65  66  67  72  74  81  
87

Exclusive:  0   1   8  12  12  21  25  33  41  43  47  52  57  58  65  66  67  72  74  
81

On each pass, the element in red (column 17 - offset) is added into the element in green (column 17) and 
stored at column 17 on the next line. By the last pass this chain of adders has communicated the sum of all 

http://en.wikipedia.org/wiki/Kogge%E2%80%93Stone_adder
http://en.wikipedia.org/wiki/Prefix_sum
http://nvlabs.github.io/moderngpu/http%7C//developer.download.nvidia.com/assets/cuda/files/reduction.pdf


elements between columns 0 and 16 into column 17.

The sequential implementation runs in O(n) operations with O(n) latency. The parallel version has O(n log n) 
work efficiency, but by breaking the serial dependencies, improves latency to O(log n) on machines with n 
processors. Most workloads have many more inputs than the device has processors. We amortize the super-
linear cooperative scan network cost by processing subsequences of the input with work-efficient serial 
functions. The partials are run through the parallel scan network, which costs O(p log p), where the number 
of processors p « n. Finally the scanned partials are added back into the inputs in linear time to complete 
operation.

This outlines a common pattern in GPU computing:

1. Upsweep to send partial reductions to the spine. 
2. Scan the spine of partials with a low-latency cooperative function. 
3. Downsweep to distribute the scanned partials from the spine to each of the inputs. 

Input array: 20 threads and 5 elements/thread:
            1   7   4   0   9   4   8   8   2   4   5   5   1   7   1   1   5   2   7   
6
            1   4   2   3   2   2   1   6   8   5   7   6   1   8   9   2   7   9   5   
4
            3   1   2   3   3   4   1   1   3   8   7   4   2   7   7   9   3   1   9   
8
            6   5   0   2   8   6   0   2   4   8   6   5   0   9   0   0   6   1   3   
8
            9   3   4   4   6   0   6   6   1   8   4   9   6   3   7   8   8   2   9   
1

Partial reduction by threads (Upsweep):
           21  26  19  21  12  22  31  27  12  17  27  30  21  20  20  18  26  21  29  
28

Parallel scan of partials (Spine):
    1:     21  47  45  40  33  34  53  58  39  29  44  57  51  41  40  38  44  47  50  
57
    2:     21  47  66  87  78  74  86  92  92  87  83  86  95  98  91  79  84  85  94 
104
    4:     21  47  66  87  99 121 152 179 170 161 169 178 187 185 174 165 179 183 185 
183
    8:     21  47  66  87  99 121 152 179 191 208 235 265 286 306 326 344 349 344 354 
361
   16:     21  47  66  87  99 121 152 179 191 208 235 265 286 306 326 344 370 391 420 
448

Exclusive scan of partials:
            0  21  47  66  87  99 121 152 179 191 208 235 265 286 306 326 344 370 391 
420

Add exclusive scan of partials into exclusive sequential scan of input array 
(Downsweep):
            0   1   8  12  12  21  25  33  41  43  47  52  57  58  65  66  67  72  74  
81
           87  88  92  94  97  99 101 102 108 116 121 128 134 135 143 152 154 161 170 
175
          179 182 183 185 188 191 195 196 197 200 208 215 219 221 228 235 244 247 248 
257
          265 271 276 276 278 286 292 292 294 298 306 312 317 317 326 326 326 332 333 
336
          344 353 356 360 364 370 370 376 382 383 391 395 404 410 413 420 428 436 438 
447



The figure above uses 20 threads to cooperatively scan 100 inputs. During the upsweep phase, each thread 
reduces five inputs using a work-efficient serial loop. The 20 partials are then scanned in parallel using five 
Kogge-Stone passes. During the downsweep phase, each thread sequentially adds its five inputs into the 
scanned partial from the spine. By increasing the grain size (the parameter VT in MGPU kernels) we do more 
linear-complexity work to amortize the O(n log n) scan network cost.

Scan operators

Modern GPU defines a scan operator interface to increase the flexibility of the Reduce and Scan functions. 
The library provides several implementations to satisfy common needs. Note that participating operations do 
not need to support the commutative property (A + B need not necessarily equal B + A), although they must 
be associative, otherwise parallelism is impossible.

include/device/ctascan.cuh

struct ScanOpInterface {
    enum { Commutative }; // False to require non-commutative 
treatment.
 
    typedef X input_type;
    typedef Y value_type;
    typedef Z result_type;
 
    // Extract() takes inputs loaded from global memory and converts 
to
    // value_type.
    MGPU_HOST_DEVICE value_type Extract(input_type t, int index);
 
    // Plus() operates on two value_types. Reduce and Scan do not rely 
on the 
    // Plus function being commutative - value t1 always represents 
values that
    // occur earlier in the input stream than t2.
    MGPU_HOST_DEVICE value_type Plus(value_type t1, value_type t2);
 
    // Combine() prepares a value for storage. Values are combined 
with the 
    // original input_type element at the same slot. Combine() is not 
used with
    // Reduce, as Reduce only returns value_types.
    MGPU_HOST_DEVICE result_type Combine(input_type t1, value_type 
t2);
 
    // Identity() returns an input_type that interacts benignly with 
any other
    // value_type in Plus(). The Identity() value_type is always 
extracted with
    // the index -1. Identity() elements appear at the end of the 
stream (in the
    // partial last tile) or are returned as the first element for an 
exclusive

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctascan.cuh


    // scan.
    MGPU_HOST_DEVICE input_type Identity();
};

ScanOpInterface is a blueprint for customizing the behavior of MGPU Reduce and Scan. Do not 
derive this type, simply make a new type that implements its three typedefs and four methods. Specialize 
CTAReduce, Reduce, CTAScan, and Scan over this new type. Although fancy iterators can provide 
much of the flexibility of ScanOpInterface, MGPU defines a pipeline that makes the process easier to 
reason about. 

When dealing with data inside a CTA (CTAReduce and CTAScan), only the Plus and Identity 
methods are used. All data is typed to value_type. Combine() supports operators that don't have the 
commutative property; if the operator defines Commutative = false, the first argument always 
represents expressions formed from inputs that come before the inputs of the second argument's expression. 
Guaranteeing the order of operands requires a CTA-wide reduction for every tile of data processed. This adds 
considerable overhead. Set Commutative = true unless your operator is emphatically non-
commutative. The ScanOpMaxIndex<> operator demonstrated in the benchmark and usage section is non-
commutative for purposes of demonstration, not of performance.

The scan interface defines three types: input_type, value_type, result_type. Inputs loaded from 
device memory are kept in register. Extract() is called on each element and passed its index. It returns 
value_type. The Extract() method allows the user to bind an index to the value for implementing 
max-index/min-index reductions, or to strip out flag bits that aren't subject to the scan logic. All folding 
operations are performed on value_type. Scanned values are converted to result_type before being 
stored to device memory. The Combine() method is passed the scanned value along with the original 
input_type value. The user can discard the value from a max-index operation or re-apply the flags that 
were masked out during the Extract().

For type consistency, the Identity() method returns input_type. It is extracted with index -1. By 
providing the identity as input_type, we can Combine() the reduction value_type with the identity 
and store it to the last element plus one.

ScanOp

include/device/ctascan.cuh

// Basic scan operators.
enum ScanOpType {
    ScanOpTypeAdd,
    ScanOpTypeMul,
    ScanOpTypeMin,
    ScanOpTypeMax
};
 
template<ScanOpType OpType, typename T>
struct ScanOp {
    enum { Commutative = true };
    typedef T input_type;
    typedef T value_type;
    typedef T result_type;
 
    MGPU_HOST_DEVICE value_type Extract(input_type t, int index) 
{ return t; }
    MGPU_HOST_DEVICE value_type Plus(value_type t1, value_type t2) { 
        switch(OpType) {
            case ScanOpTypeAdd: t1 += t2; break;
            case ScanOpTypeMul: t1 *= t2; break;
            case ScanOpTypeMin: t1 = min(t1, t2); break;
            case ScanOpTypeMax: t1 = max(t1, t2); break;
        }
        return t1;
    }
    MGPU_HOST_DEVICE result_type Combine(input_type t1, value_type t2) 
{
        return t2;
    }
    MGPU_HOST_DEVICE input_type Identity() { return _ident; }
     
    MGPU_HOST_DEVICE ScanOp(input_type ident) : _ident(ident) { }
    MGPU_HOST_DEVICE ScanOp() {
        switch(OpType) {
            case ScanOpTypeAdd: _ident = 0; break;
            case ScanOpTypeMul: _ident = 1; break;
            case ScanOpTypeMin: _ident = numeric_limits<T>::max(); 
break;
            case ScanOpTypeMax: _ident = numeric_limits<T>::lowest(); 
break;
        }
    }
 
    input_type _ident;
};
typedef ScanOp<ScanOpTypeAdd, int> ScanOpAdd;

ScanOp is a basic operator that performs add, mul, min, or max functions. MGPU includes a subset of 
std::numeric_limits<> (in the mgpu namespace) with methods __device__-tagged to support 
specialization of ScanOp over built-in types. The typedef ScanOpAdd is provided for convenience - a 
partial sum on integers is by far the most common scan operator in MGPU kernels.



ScanOpIndex

template<typename T>
struct ScanOpIndex {
    enum { Commutative = false };
    struct Pair { int index; T value; };
 
    typedef T input_type;
    typedef Pair value_type;
    typedef int result_type;
    MGPU_HOST_DEVICE value_type Extract(T t, int index) {
        Pair p = { index, t };
        return p;
    }
    MGPU_HOST_DEVICE int Combine(T t1, value_type t2) {
        return t2.index;
    }
    MGPU_HOST_DEVICE input_type Identity() { 
        return _identity;
    }
    MGPU_HOST_DEVICE ScanOpIndex(T identity) : _identity(identity) { }
    T _identity;
};
 
template<typename T>
struct ScanOpMinIndex : ScanOpIndex<T> {
    typedef typename ScanOpIndex<T>::value_type value_type;
    MGPU_HOST_DEVICE value_type Plus(value_type t1, value_type t2) {
        if(t2.value < t1.value) t1 = t2;
        return t1;
    }
    MGPU_HOST_DEVICE ScanOpMinIndex(T max_ = numeric_limits<T>::max()) 
:
        ScanOpIndex<T>(max_) { }
};
template<typename T>
struct ScanOpMaxIndex : ScanOpIndex<T> {
    typedef typename ScanOpIndex<T>::value_type value_type;
    MGPU_HOST_DEVICE value_type Plus(value_type t1, value_type t2) {
        if(t2.value > t1.value) t1 = t2;
        return t1;
    }
    MGPU_HOST_DEVICE ScanOpMaxIndex(T min_ = 
numeric_limits<T>::lowest()) :
        ScanOpIndex<T>(min_) { }
};

ScanOpIndex is the base class for min-index and max-index operators. Commutative = false is a 
request for the reduction kernel to treat the operator as non-commutative. The operator uses type-morphing to 
change type T input to integer output. On Extract(), the value and index are moved into a pair which is 

returned. The Plus() operator only compares the values - because the inputs of the left expression come 
before the inputs of the right expression, the user is guaranteed, in case of multiple min- or max-elements, to 
receive the index of the left-most one. Combine() discards the value and returns the left-most index of an 
extremum.

Important: MGPU kernels aggressively union together types to conserve shared memory. Don't specialize 
MGPU kernels over types that have non-trivial constructors or destructors. C++ 11 relaxes this restriction, as 
will this library when CUDA supports the new union semantics. If your type cannot be unioned, cast to an 
equivalent POD type.

CTAReduce

CTAReduce recursively folds inputs. Specialize with a power-of-two number of threads and a scan operator. 
This is a work-efficient operation requiring n total additions and log(n) passes.

Inputs in thread order:  0   1   2   3   4   5   6   7   8   9  10  11  12  13  14  15

Fold elements 8-15       0   1   2   3   4   5   6   7
into 0-7:                8   9  10  11  12  13  14  15

Fold elements 4-7        0   1   2   3
into 0-3:                8   9  10  11
                         4   5   6   7
                        12  13  14  15

Fold elements 2-3        0   1
into 0-1:                8   9
                         4   5
                        12  13
                         2   3
                        10  11
                         6   7
                        14  15

Thread 0 folds in element 1:
Reduction = 0 + 8 + 4 + 12 + 2 + 10 + 6 + 14 + 1 + 9 + 5 + 13 + 3 + 11 + 7 + 15

Folding the data in thread order correctly reduces inputs only for operations that support the commutative 
property. Prefix sum would work, but max-index would not necessarily return the left-most extrema. We 
need to permute the inputs such that the reduction is in ascending order. Scatter element 0 into position 0, 
element 1 into position 8, element 2 into position 4, etc.

Permuted input order:    0   8   4  12   2  10   6  14   1   9   5  13   3  11   7  15

Fold elements 8-15       0   8   4  12   2  10   6  14
into 0-7:                1   9   5  13   3  11   7  15

Fold elements 4-7        0   8   4  12
into 0-3:                1   9   5  13
                         2  10   6  14
                         3  11   7  15

Fold elements 2-3        0   8
into 0-1:                1   9
                         2  10
                         3  11
                         4  12



                         5  13
                         6  14
                         7  15

Thread 0 folds in element 1:
Reduction = 0 + 1 + 2 + 3 + 4 + 5 + 6 + 7 + 8 + 9 + 10 + 11 + 12 + 13 + 14 + 15

Fortunately this permutation is trivial to calculate. We simply reverse the bits in the thread ID and scatter. 
Devices of compute capability 2.0 (Fermi) and later support a hardware bit reverse intrinsic __brev.

include/device/ctascan.cuh

template<int NT, typename Op = ScanOpAdd>
struct CTAReduce {
    typedef typename Op::value_type T;
    enum { Size = NT };
    enum { Capacity = NT + NT / WARP_SIZE };
    struct Storage { T shared[Capacity]; };
 
    MGPU_DEVICE static T Reduce(int tid, T x, Storage& storage, Op op = 
Op()) {
        // Reverse the bits of the source thread ID and make a 
conflict-free
        // store using a 33-stride spacing.
        int dest = brev(tid)>> (32 - sLogPow2<NT>::value);
        storage.shared[dest + dest / WARP_SIZE] = x;
        __syncthreads();
 
        // Fold the data in half with each pass.
        int src = tid + tid / WARP_SIZE;
        #pragma unroll
        for(int destCount = NT / 2; destCount >= 1; destCount /= 2) {
            if(tid < destCount) {
                // On the first pass, read this thread's data out of 
shared 
                // memory.
                if(NT / 2 == destCount) x = storage.shared[src];
                int src2 = destCount + tid;
                x = op.Plus(x, storage.shared[src2 + src2 / 
WARP_SIZE]);
                storage.shared[src] = x;
            }
            __syncthreads();
        }
        T total = storage.shared[0];
        __syncthreads();
        return total;
    }
};

CTA reduction is defined as a type to unite a storage structure and a static method in the same scope. Callers 
should union CTAReduce<>::Storage into their shared memory structure to optimize occupancy. The 
bit-reversal permutation causes bank conflicts by mapping multiple lanes in a warp to the same bank which 

this method avoids by adding a padding index every 32 slots.

Important: When mapping adjacent threads into the same bank, avoid conflicts by adding a padding element 
every 32 slots. Set index = index + index / WARP_SIZE and reserve NT + NT / 
WARP_SIZE shared memory slots.

Simulation of bank-conflict resolution on 16 banks:
      0   1   2   3   4   5   6   7   8   9  10  11  12  13  14  15
      x  16  17  18  19  20  21  22  23  24  25  26  27  28  29  30
     31   x  32  33  34  35  36  37  38  39  40  41  42  43  44  45
     46  47   x  48  49  50  51  52  53  54  55  56  57  58  59  60
     61  62  63   x  64  65  66  67  68  69  70  71  72  73  74  75
     76  77  78  79   x  80  81  82  83  84  85  86  87  88  89  90
     91  92  93  94  95   x  96  97  98  99 100 101 102 103 104 105
    106 107 108 109 110 111   x 112 113 114 115 116 117 118 119 120
    121 122 123 124 125 126 127   x 128 129 130 131 132 133 134 135
    136 137 138 139 140 141 142 143   x 144 145 146 147 148 149 150
    151 152 153 154 155 156 157 158 159   x 160 161 162 163 164 165
    166 167 168 169 170 171 172 173 174 175   x 176 177 178 179 180
    181 182 183 184 185 186 187 188 189 190 191   x 192 193 194 195
    196 197 198 199 200 201 202 203 204 205 206 207   x 208 209 210
    211 212 213 214 215 216 217 218 219 220 221 222 223   x 224 225
    226 227 228 229 230 231 232 233 234 235 236 237 238 239   x 240
    241 242 243 244 245 246 247 248 249 250 251 252 253 254 255   x

CTAScan and shfl scan

include/device/ctascan.cuh

template<int NT, typename Op = ScanOpAdd>
struct CTAScan {
    typedef typename Op::value_type T;
    enum { Size = NT, Capacity = 2 * NT + 1 };
    struct Storage { T shared[Capacity]; };
 
    MGPU_DEVICE static T Scan(int tid, T x, Storage& storage, T* total,
        MgpuScanType type = MgpuScanTypeExc, Op op = Op()) {
 
        storage.shared[tid] = x;
        int first = 0;
        __syncthreads();
 
        #pragma unroll
        for(int offset = 1; offset < NT; offset += offset) {
            if(tid >= offset)
                x = op.Plus(storage.shared[first + tid - offset], x);
            first = NT - first;
            storage.shared[first + tid] = x;
            __syncthreads();
        }
        *total = storage.shared[first + NT - 1];
        if(MgpuScanTypeExc == type) 
            x = tid ? 
                storage.shared[first + tid - 1] : 
                op.Extract(op.Identity(), -1);
        __syncthreads();
 
        return x;
    }
    MGPU_DEVICE static T Scan(int tid, T x, Storage& storage) {
        T total;
        return Scan(tid, x, storage, &total, MgpuScanTypeExc, Op());
    }
};

CTAScan is a basic implemenation that uses double buffering to reduce synchronization. To further reduce 
latency we utilize the shfl instruction available on Kepler. This feature supports inter-lane communication 
inside a warp with a only a single trip over the shared memory cross-bar. Although CUDA C++ includes a 
__shfl intrinsic, we choose to access the instruction using inline PTX to save the returned predicate flag. 
The current CUDA backend copies predicate flags into registers when they are returned as bool types, 
resulting in wasted instructions if we were to build a shuffle intrinsic that returned both value and predicate. 
Best performance is achieved by executing both the shfl and the add in inline PTX.

include/device/intrinsics.cuh

MGPU_DEVICE int shfl_add(int x, int offset, int width = WARP_SIZE) {
    int result = 0;
#if __CUDA_ARCH__ >= 300
    int mask = (WARP_SIZE - width)<< 8;
    asm(
        "{.reg .s32 r0;"
        ".reg .pred p;"
        "shfl.up.b32 r0|p, %1, %2, %3;"
        "@p add.s32 r0, r0, %4;"
        "mov.s32 %0, r0; }"
        : "=r"(result) : "r"(x), "r"(offset), "r"(mask), "r"(x));
#endif
    return result;
}

The mov instruction is elided by the compiler, creating a warp scan in a tight sequence of five shfl and five 
predicated add instructions. The shfl_add function scans multiple segments within a warp, where width 
is a power-of-two segment size.

include/device/ctascan.cuh

#if __CUDA_ARCH__ >= 300
 
template<int NT>
struct CTAScan<NT, ScanOpAdd> {
    // Define WARP_SIZE segments that are NT / WARP_SIZE large.
    // Each warp makes log(SegSize) shfl_add calls.
    // The spine makes log(WARP_SIZE) shfl_add calls.
    enum { Size = NT, NumSegments = WARP_SIZE, SegSize = NT / 
NumSegments };
    enum { Capacity = NumSegments + 1 };
    struct Storage { int shared[Capacity + 1]; };
 
    MGPU_DEVICE static int Scan(int tid, int x, Storage& storage, int* 
total,
        MgpuScanType type = MgpuScanTypeExc, ScanOpAdd op = 
ScanOpAdd()) {
 
        int lane = (SegSize - 1) & tid;
        int segment = tid / SegSize;
 
        // Scan each segment using shfl_add.
        int scan = x;
        #pragma unroll
        for(int offset = 1; offset < SegSize; offset *= 2)
            scan = shfl_add(scan, offset, SegSize);
 
        // Store the reduction (last element) of each segment into 
storage.
        if(SegSize - 1 == lane) storage.shared[segment] = scan;
        __syncthreads();
 
        // Warp 0 does a full shfl warp scan on the partials. The 
total is
        // stored to shared[NumSegments]. (NumSegments = WARP_SIZE)
        if(tid < NumSegments) {
            int y = storage.shared[tid];
            int scan = y;
            #pragma unroll
            for(int offset = 1; offset < NumSegments; offset *= 2)
                scan = shfl_add(scan, offset, NumSegments);
            storage.shared[tid] = scan - y;
            if(NumSegments - 1 == tid) storage.shared[NumSegments] = 
scan;
        }
        __syncthreads();
 
        // Add the scanned partials back in and convert to exclusive 
scan.
        scan += storage.shared[segment];
        if(MgpuScanTypeExc == type) scan -= x;
        *total = storage.shared[NumSegments];
        __syncthreads();
 



The CTA shuffle scan implementation takes the form of warp-synchronous programming but without the 
need for volatile memory qualifiers. We choose to divide the input into 32 equal segments. For 256 threads, 
we have a segment size of eight, and this is scanned in three calls to shfl_add. The last thread in each 
segment stores the partial sum to shared memory. After a barrier the partials are warp-scanned with five 
invocations of shfl_add.

The choice to scan small segments in the upsweep (8 threads/segment) and scan large segments in the spane 
(32 threads/segment) has significant consequence for work efficiency: in the 256-thread example, each of the 
eight warps makes three calls to shfl_add in the upsweep, and the spine warp makes five calls, for 29 
shuffles in all. By contrast, setting the segment size to 32 performs a five-pass warp scan in the upsweep and 
a three-pass scan over the eight partials in the spine, calling shfl_add 43 times. Changing the fan-out of 
scan networks can have implications for both the latency and efficiency. 

Reduce kernel

include/kernels/reduce.cuh

// Run a high-throughput reduction over multiple CTAs. Used as the 
upsweep phase
// for global reduce and global scan.
template<typename Tuning, typename InputIt, typename Op>
MGPU_LAUNCH_BOUNDS void KernelReduce(InputIt data_global, int count, 
    int2 task, typename Op::value_type* reduction_global, Op op) {
 
    typedef MGPU_LAUNCH_PARAMS Params;
    const int NT = Params::NT;
    const int VT = Params::VT;
    const int NV = NT * VT;
    typedef typename Op::input_type input_type;
    typedef typename Op::value_type value_type;
    typedef CTAReduce<NT, Op> R;
 
    union Shared {
        typename R::Storage reduce;
        input_type inputs[NV];
    };
    __shared__ Shared shared;
 
    int tid = threadIdx.x;
    int block = blockIdx.x;
    int first = VT * tid;
 
    int2 range = ComputeTaskRange(block, task, NV, count);
 
    // total is the sum of encountered elements. It's undefined on the 
first 
    // loop iteration.
    value_type total = op.Extract(op.Identity(), -1);
    bool totalDefined = false;
 
    while(range.x < range.y) {
        int count2 = min(NV, count - range.x);
 
        // Read terms into register.
        input_type inputs[VT];
        DeviceGlobalToReg<NT, VT>(count2, data_global + range.x, tid, 
inputs);
 
        if(Op::Commutative) {
            // This path exploits the commutative property of the 
operator.
            #pragma unroll
            for(int i = 0; i < VT; ++i) {
                int index = NT * i + tid;
                if(index < count2) {
                    value_type x = op.Extract(inputs[i], range.x + 
index);
                    total = (i || totalDefined) ? op.Plus(total, x) : 
x;
                }



We cover the Reduce kernel in detail, as most MGPU kernels are similarly constructed. The first template 
argument is Tuning, indicating the kernel uses the LaunchBox tuning mechanism. The 
MGPU_LAUNCH_BOUNDS macro expands to a __launch_bounds__ attribute and __global__ 
tag. Input data is accepted with a template argument InputIt to support iterators in addition to pointers. 
MGPU provides counting_iterator and step_iterator, and users can pass their own or any of 
the custom iterators included with Thrust.

The tuning parameters NT and VT are pulled from the Tuning argument by way of the 
MGPU_LAUNCH_PARAMS macro. As described in the library overview, the grain size VT should be odd to 
avoid bank conflicts when reading out values from shared memory in thread order.

CTAReduce is specialized over the scan operator and its storage is made part of the union. The task range is 
computed and the CTA loops—sizing the grid to the device rather than the data results in a smaller spine 
with lower-latency scan.

Data is loaded from the input iterator into shared memory using DeviceGlobalToShared. All device 
methods in loadstore.cuh that move to or from shared memory include an implicit synchronization—this can 
be disabled by passing false to the default last argument. After the data is loaded, threads read VT terms 
starting at VT * tid and call Extract() to change the data type from input_type to value_type. 
The Plus method on the scan operator combines values into x. The larger the grain size, the smaller the 
relative cost of the cooperative CTA reduction which follows. The reduction for the iteration, x, is added into 
the cumulative reduction for the CTA, total, and the loop continues until the task is complete. The CTA 
reduction is stored to global memory as a value_type. This may be copied to the host, combined with the 
other partials, and returned. Or it may be fed into another kernel to compute a global scan.

Scan kernel

include/kernels/scan.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/scan.cuh
https://github.com/NVlabs/moderngpu/blob/master/include/device/loadstore.cuh
http://nvlabs.github.io/moderngpu/library.html#taskrange
http://nvlabs.github.io/moderngpu/library.html#loadstore

template<int NT, int VT, MgpuScanType Type, typename InputIt, typename 
OutputIt, 
    typename Op>
__global__ void KernelParallelScan(InputIt cta_global, int count, Op 
op, 
    typename Op::value_type* total_global, typename Op::result_type* 
end_global,
    OutputIt dest_global) {
 
    typedef typename Op::input_type input_type;
    typedef typename Op::value_type value_type;
    typedef typename Op::result_type result_type;
    const int NV = NT * VT;
 
    typedef CTAScan<NT, Op> S;
    union Shared {
        typename S::Storage scan;
        input_type inputs[NV];
        result_type results[NV];
    };
    __shared__ Shared shared;
 
    int tid = threadIdx.x;
     
    // total is the sum of encountered elements. It's undefined on the 
first 
    // loop iteration.
    value_type total = op.Extract(op.Identity(), -1);
    bool totalDefined = false;
    int start = 0;
    while(start < count) {
        // Load data into shared memory.
        int count2 = min(NV, count - start);
        DeviceGlobalToShared<NT, VT>(count2, cta_global + start, tid, 
            shared.inputs);
 
        // Transpose data into register in thread order. Reduce terms 
serially.
        input_type inputs[VT];
        value_type values[VT];
        value_type x = op.Extract(op.Identity(), -1);
        #pragma unroll
        for(int i = 0; i < VT; ++i) {
            int index = VT * tid + i;
            if(index < count2) {
                inputs[i] = shared.inputs[index];
                values[i] = op.Extract(inputs[i], start + index);
                x = i ? op.Plus(x, values[i]) : values[i];
            }
        }
        __syncthreads();
                 
        // Scan the reduced terms.



KernelParallelScan is launched on a single block and completely scans its input. It is structured as a 
persistent CTA, looping until all inputs have been processed. This spine function takes actions that resemble 
a global scan, but in miniature:

1. DeviceGlobalToShared cooperatively loads elements as input_type and stores into shared 
memory.

2. Threads loop through the values in a serial reduction, calling Extract() to convert inputs to 
value_type and generating partials with application of Plus().

3. The CTA cooperatively scans the partials with CTAScan. Note that the storage for this method is 
unioned into the Shared structure.

4. Threads loop through the values in a serial downsweep, calling Combine() to add values with the 
previous scan result, producing an array of result_type which are stored in shared memory. Note 
that inclusive scans sum before storing, and exclusive scans store before summing.

5. DeviceSharedToGlobal cooperatively stores result_type data into global memory.

ScanOpValue

include/device/ctascan.cuh

// Override the Extract and Combine behavior of the base operator. 
This prevents
// the Scan kernel from extracting or combining values twice.
template<typename Base>
struct ScanOpValue : public Base {
    typedef typename Base::value_type input_type;
    typedef typename Base::value_type value_type;
    typedef typename Base::value_type result_type;
    MGPU_HOST_DEVICE value_type Extract(value_type t, int index) 
{ return t; }
    MGPU_HOST_DEVICE value_type Combine(value_type t1, value_type t2) 
{
        return t2;     
    }
    MGPU_HOST_DEVICE value_type Identity() {
        return Base::Extract(Base::Identity(), -1);
    }
    MGPU_HOST_DEVICE ScanOpValue(Base base) : Base(base) { }
};

Because KernelParallelScan does double-duty both spine-scanning partials for large inputs and 
completely scanning elements for small inputs, we could potentially run into problems with the type system. 
The kernel calls Extract() and Combine() to convert between types, but when used to scan the spine of 
partials, inputs are already in value_type (but we expect them in input_type). Before launching 
KernelParallelScan to scan the spine of partials, the host encapsulates the user-provided scan operator 
in ScanOpValue to suppress type morphing. Extract() and Combine() are overridden to pass 
value_type arguments straight throughe. This shim class inherits its Commutative property from the 
base class.

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctascan.cuh


KernelScanDownsweep, which runs the downsweep for large input arrays, operates in much the same 
way as KernelParallelScan. 



5. Bulk Remove and Bulk Insert
Bulk Remove and Bulk Insert are high-throughput functions for fine-grained parallel array surgery. Bulk 
Remove compacts an array by removing all elements specified by index. Bulk Insert merges two arrays in 
parallel by insertion index. Merge Path partitioning for load balancing is introduced here.

Benchmark and usage

Bulk Remove benchmark from tests/benchmarkinsert.cu

Bulk Remove demonstration from tests/demo.cu

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarkinsert.cu

void DemoBulkRemove(CudaContext& context) {
    printf("\n\nBULK REMOVE DEMONSTRATION:\n\n");
 
    // Use CudaContext::FillAscending to generate 100 integers between 
0 and 99.
    int N = 100;
    MGPU_MEM(int) data = context.FillAscending<int>(N, 0, 1);
 
    printf("Input array:\n");
    PrintArray(*data, "%4d", 10);
 
    // Remove every 3rd element from the exclusive scan. Use
    // CudaContext::FillAscending to generate removal indices for 
every 3rd
    // integer between 0 and 99.
    int RemoveCount = MGPU_DIV_UP(N, 3);
    MGPU_MEM(int) remove = context.FillAscending(RemoveCount, 0, 3);
    MGPU_MEM(int) data2 = context.Malloc<int>(N - RemoveCount);
 
    BulkRemove(data->get(), N, remove->get(), RemoveCount, data2-
>get(),
        context);
    printf("\nRemoving every 3rd element:\n");
    PrintArray(*data2, "%4d", 10);
}

BULK REMOVE DEMONSTRATION:

Input array:
    0:     0    1    2    3    4    5    6    7    8    9
   10:    10   11   12   13   14   15   16   17   18   19
   20:    20   21   22   23   24   25   26   27   28   29
   30:    30   31   32   33   34   35   36   37   38   39
   40:    40   41   42   43   44   45   46   47   48   49
   50:    50   51   52   53   54   55   56   57   58   59
   60:    60   61   62   63   64   65   66   67   68   69
   70:    70   71   72   73   74   75   76   77   78   79
   80:    80   81   82   83   84   85   86   87   88   89
   90:    90   91   92   93   94   95   96   97   98   99

Removing every 3rd element:
    0:     1    2    4    5    7    8   10   11   13   14
   10:    16   17   19   20   22   23   25   26   28   29
   20:    31   32   34   35   37   38   40   41   43   44
   30:    46   47   49   50   52   53   55   56   58   59
   40:    61   62   64   65   67   68   70   71   73   74
   50:    76   77   79   80   82   83   85   86   88   89
   60:    91   92   94   95   97   98



Bulk Insert benchmark from tests/benchmarkinsert.cu

Bulk Insert demonstration from tests/demo.cu

void DemoBulkInsert(CudaContext& context) {
    printf("\n\nBULK INSERT DEMONSTRATION:\n\n");
 
    // Use CudaContext::FillAscending to generate 100 integers between 
0 and 99.
    int N = 100;
    MGPU_MEM(int) data = context.FillAscending<int>(N, 0, 1);
 
    printf("Input array:\n");
    PrintArray(*data, "%4d", 10);
 
    // Insert new elements before every 5 input starting at index 2. 
    // Use step_iterator for insertion positions and content.
    int InsertCount = MGPU_DIV_UP(N - 2, 5);
    MGPU_MEM(int) data2 = context.Malloc<int>(N + InsertCount);
    mgpu::step_iterator<int> insertData(1000, 10);
    mgpu::step_iterator<int> insertIndices(2, 5);
 
    BulkInsert(insertData, insertIndices, InsertCount, data->get(), N, 
        data2->get(), context);
 
    printf("\nInserting before every 5th element starting at item 
2:\n");
    PrintArray(*data2, "%4d", 10);

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarkinsert.cu


}

BULK INSERT DEMONSTRATION:

Input array:
    0:     0    1    2    3    4    5    6    7    8    9
   10:    10   11   12   13   14   15   16   17   18   19
   20:    20   21   22   23   24   25   26   27   28   29
   30:    30   31   32   33   34   35   36   37   38   39
   40:    40   41   42   43   44   45   46   47   48   49
   50:    50   51   52   53   54   55   56   57   58   59
   60:    60   61   62   63   64   65   66   67   68   69
   70:    70   71   72   73   74   75   76   77   78   79
   80:    80   81   82   83   84   85   86   87   88   89
   90:    90   91   92   93   94   95   96   97   98   99

Inserting before every 5th element starting at item 2:
    0:     0    1 1000    2    3    4    5    6 1010    7
   10:     8    9   10   11 1020   12   13   14   15   16
   20:  1030   17   18   19   20   21 1040   22   23   24
   30:    25   26 1050   27   28   29   30   31 1060   32
   40:    33   34   35   36 1070   37   38   39   40   41
   50:  1080   42   43   44   45   46 1090   47   48   49
   60:    50   51 1100   52   53   54   55   56 1110   57
   70:    58   59   60   61 1120   62   63   64   65   66
   80:  1130   67   68   69   70   71 1140   72   73   74
   90:    75   76 1150   77   78   79   80   81 1160   82
  100:    83   84   85   86 1170   87   88   89   90   91
  110:  1180   92   93   94   95   96 1190   97   98   99

Host functions

include/mgpuhost.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/mgpuhost.cuh

//////////////////////////////////////////////////////////////////////
//////////
// kernels/bulkremove.cuh
 
// Compact the elements in source_global by removing elements 
identified by
// indices_global. indices_global must be unique, sorted, and range 
between 0
// and sourceCount - 1. The number of outputs is sourceCount - 
indicesCount.
 
// IndicesIt should resolve to an integer type. iterators like 
step_iterator
// are supported.
 
// If sourceCount = 10, indicesCount = 6, and indices = (1, 3, 4, 5, 
7, 8), then
// dest = A0 A2 A6 A9. (All indices between 0 and sourceCount - 1 
except those
// in indices_global).
template<typename InputIt, typename IndicesIt, typename OutputIt>
MGPU_HOST void BulkRemove(InputIt source_global, int sourceCount,
    IndicesIt indices_global, int indicesCount, OutputIt dest_global,
    CudaContext& context);
 
 
//////////////////////////////////////////////////////////////////////
//////////
// kernels/bulkinsert.cuh
 
// Combine aCount elements in a_global with bCount elements in 
b_global.
// Each element a_global[i] is inserted before position 
indices_global[i] and
// stored to dest_global. The insertion indices are relative to the B 
array,
// not the output. Indices must be sorted but not necessarily unique. 
 
// If aCount = 5, bCount = 3, and indices = (1, 1, 2, 3, 3), the 
output is:
// B0 A0 A1 B1 A2 B2 A3 A4.
template<typename InputIt1, typename IndicesIt, typename InputIt2,
    typename OutputIt>
MGPU_HOST void BulkInsert(InputIt1 a_global, IndicesIt indices_global, 
    int aCount, InputIt2 b_global, int bCount, OutputIt dest_global,
    CudaContext& context);

Bulk Remove algorithm

Bulk Remove and Bulk Insert are intermediate forms between functions of the scan idiom and functions of 
MGPU's two-phase idiom. These functions search over their inputs to establish coarse-grained partitionings. 



Sorted indices are loaded into CTAs and scatter and scan operations remove or insert items with fine-grained 
control.

Data:
    0:     0    1    2    3    4    5    6    7    8    9
   10:    10   11   12   13   14   15   16   17   18   19
   20:    20   21   22   23   24   25   26   27   28   29
   30:    30   31   32   33   34   35   36   37   38   39
   40:    40   41   42   43   44   45   46   47   48   49
   50:    50   51   52   53   54   55   56   57   58   59
   60:    60   61   62   63   64   65   66   67   68   69
   70:    70   71   72   73   74   75   76   77   78   79
   80:    80   81   82   83   84   85   86   87   88   89
   90:    90   91   92   93   94   95   96   97   98   99

Remove indices:
    0:     1    4    5    7   10   14   15   16   18   19
   10:    27   29   31   32   33   36   37   39   50   59
   20:    60   61   66   78   81   83   85   90   91   96
   30:    97   98   99

For Bulk Remove, we start with a full tile of values plus the interval of remove indices that map into the tile.

Flags:
    0:     1    0    1    1    0    0    1    0    1    1
   10:     0    1    1    1    0    0    0    1    0    0
   20:     1    1    1    1    1    1    1    0    1    0
   30:     1    0    0    0    1    1    0    0    1    0
   40:     1    1    1    1    1    1    1    1    1    1
   50:     0    1    1    1    1    1    1    1    1    0
   60:     0    0    1    1    1    1    0    1    1    1
   70:     1    1    1    1    1    1    1    1    0    1
   80:     1    0    1    0    1    0    1    1    1    1
   90:     0    0    1    1    1    1    0    0    0    0

A tile-sized buffer of flags is initialized to 1. Poking in 0 will remove the value at that position. Threads 
scatter 0s for all elements listed in the remove indices array.

Scan of flags:
   0:      0    1    1    2    3    3    3    4    4    5 
  10:      6    6    7    8    9    9    9    9   10   10
  20:     10   11   12   13   14   15   16   17   17   18
  30:     18   19   19   19   19   20   21   21   21   22
  40:     22   23   24   25   26   27   28   29   30   31 
  50:     32   32   33   34   35   36   37   38   39   40
  60:     40   40   40   41   42   43   44   44   45   46 
  70:     47   48   49   50   51   52   53   54   55   55 
  80:     56   57   57   58   58   59   59   60   61   62 
  90:     63   63   63   64   65   66   67   67   67   67

Reduced data:
    0:     0    2    3    6    8    9   11   12   13   17
   10:    20   21   22   23   24   25   26   28   30   34
   20:    35   38   40   41   42   43   44   45   46   47
   30:    48   49   51   52   53   54   55   56   57   58
   40:    62   63   64   65   67   68   69   70   71   72
   50:    73   74   75   76   77   79   80   82   84   86
   60:    87   88   89   92   93   94   95

An exclusive scan is run on the flags. Flags that were 1 prior to the scan reserve one slot for their values; 



flags that were 0 reserve no space. The scanned offsets in green had set flags. Values at the green locations in 
the tile scatter their values to shared memory at their corresponding scanned locations. In this example, 
element 0 is stored to location 0; element 1 is skipped (its flag is 0); element 2 is stored to location 1; etc.

BinarySearchPartitions

How do we find the interval of removal indices to load into the CTA, and where does each CTA store its 
Bulk Remove output? By insisting on sorted indices (and no duplicate indices) both questions have simple 
answers:

1. Remove indices are partioned with a lower-bound binary search, where the keys are the starting 
offsets for each tile, i.e., multiplies of the block size.

2. Destination offsets for each tile are inferred from the the tile's input offset (a multiple of the block 
size) and the removal index offset from 1. Because Bulk Remove requires indices both sorted and 
unique, a CTA knows where to stream given only its block ID and the location of the first remove 
index in its range. If NV = 1000, block 8 that loads removal indices (2113, 2423) streams exactly 690 
values (1000 - (2423 - 2113)) starting at output position 5887 (8 * 1000 - 2113). 

include/kernels/search.cuh

template<int NT, MgpuBounds Bounds, typename It, typename Comp>
__global__ void KernelBinarySearch(int count, It data_global, int 
numItems,
    int nv, int* partitions_global, int numSearches, Comp comp) {
 
    int gid = NT * blockIdx.x + threadIdx.x;
    if(gid < numSearches) {
        int p = BinarySearch<Bounds>(data_global, numItems, 
            min(nv * gid, count), comp);
        partitions_global[gid] = p;
    }
}
 
template<MgpuBounds Bounds, typename It1, typename Comp>
MGPU_MEM(int) BinarySearchPartitions(int count, It1 data_global, int 
numItems,
    int nv, Comp comp, CudaContext& context) {
 
    const int NT = 64;
    int numBlocks = MGPU_DIV_UP(count, nv);
    int numPartitionBlocks = MGPU_DIV_UP(numBlocks + 1, NT);
    MGPU_MEM(int) partitionsDevice = context.Malloc<int>(numBlocks + 
1);
 
    KernelBinarySearch<NT, Bounds>
        <<<numPartitionBlocks, NT, 0, context.Stream()>>>(count, 
data_global, 
        numItems, nv, partitionsDevice->get(), numBlocks + 1, comp);
    return partitionsDevice;

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/search.cuh


}

Many host functions call BinarySearchPartitions to partition problems before calling the logic 
kernel. If the CTA size is 1000, this launch finds intervals in the argument array with values between 0 and 
999, 1000 and 1999, etc. For BulkRemove it maps the correct remove indices into each tile.

Thrust includes a function remove_if that compacts inputs given a corresponding array of predicates. This 
interface requires multiple trips through the input: the first trip simply counts the number of elements not to 
remove. Modern GPU functions usually accept sorted arrays of indices (to remove, to insert, or to mark 
segment heads) because it eliminates a pass that would otherwise need to run just to calculate output 
locations and sizes.

KernelBulkRemove

include/kernels/bulkremove.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/bulkremove.cuh
http://thrust.github.io/doc/group__stream__compaction.html#ga5760a32d1a99d89732206f48b75138ea

template<typename Tuning, typename InputIt, typename IndicesIt, 
    typename OutputIt>
MGPU_LAUNCH_BOUNDS void KernelBulkRemove(InputIt source_global, int 
sourceCount, 
    IndicesIt indices_global, int indicesCount, const int* p_global,
    OutputIt dest_global) {
 
    typedef MGPU_LAUNCH_PARAMS Params;
    typedef typename std::iterator_traits<InputIt>::value_type T;
    const int NT = Params::NT;
    const int VT = Params::VT;
    const int NV = NT * VT;
    typedef CTAScan<NT, ScanOpAdd> S;
    union Shared {
        int indices[NV];
        typename S::Storage scan;
    };
    __shared__ Shared shared;
 
    int tid = threadIdx.x;
    int block = blockIdx.x;
    int gid = block * NV;
    sourceCount = min(NV, sourceCount - gid);
 
    // Search for begin and end iterators of interval to load.
    int p0 = p_global[block];
    int p1 = p_global[block + 1];
 
    // Set the flags to 1. The default is to copy a value.
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        int index = NT * i + tid;
        shared.indices[index] = index < sourceCount;
    }
    __syncthreads();
 
    // Load the indices into register.
    int begin = p0;
    int indexCount = p1 - p0;
    int indices[VT];
    DeviceGlobalToReg<NT, VT>(indexCount, indices_global + p0, tid, 
indices);
 
    // Set the counter to 0 for each index we've loaded.
    #pragma unroll
    for(int i = 0; i < VT; ++i)
        if(NT * i + tid < indexCount) 
            shared.indices[indices[i] - gid] = 0;
    __syncthreads();
 
    // Run a raking scan over the flags. We count the set flags - this 
is the 
    // number of elements to load in per thread.



After partitioning, the host launches KernelBulkRemove. The results of the binary search are loaded into 
p0 and p1. Flags are initialized to 1 as described, except those for last-tile elements which extend past the end 
of the input: these are set to 0. DeviceGlobalToReg loads indices (p0, p1) into register. Because the 
indices are unique, there cannot be more than NV indices in the interval (p0, p1), so we can safely load into 
register with an unrolled loop. We use the local removal indices (the offset of the start of the tile is subtracted 
from each removal index) and poke 0s into shared memory.

Each thread loads VT flags starting at VT * tid. These are summed and passed to CTAReduce. Scanned 
flags with corresponding set indices (the scan of flags in green from three figures above) are compacted by 
conditionally streaming not to VT * tid but to each thread's count scan. Because we had to load flags in 
thread order (VT * tid + i) rather than strided order (NT * i + tid), it is more efficient to invert the problem 
from a scatter to a gather.

Finally, the threads cooperatively load sourceCount - indexCount indices in strided order, gather 
the values from global memory, and store directly to output at dest_global + gid - p0.

Bulk Insert partitioning

The complementary function requires a bit more sophistication to write. Remove is easy, because the array 
can only get smaller. Insert is harder because the array may grow arbitrarily large, causing load-balancing 
problems. If we were to assign a fixed amount of inputs (say, 1000) to BulkInsert as we do with 
BulkRemove, we may not be able to hold the input indices or the expanded array in the CTA's register or 
shared memory resources.

The caller may want to insert 1000 or even 1,000,000 elements before the first input. How does the kernel 
handle thousands of consecutive 0s in the insert indices array? Unlike the remove case, we cannot disallow 
duplicate insert indices without loss of functionality. If BinarySearchPartition mapped 10,000 insert 
requests into a single tile, the kernel would have to dynamically loop through these requests to fulfill them 
within the capacity constraints of the device. This is expressly against our philosophy that the logic code 
should not handle scheduling - that is the domain of the partioning phase.

The solution is to generalize the partitioning code to search over two inputs simultaneously: we want to 
balance the source data with the insertion requests. For NV = 1000, loading 712 inputs implies we can load 
no more than 288 indices. This way both the new and the old values get merged without overflowing the 
resources available for the block.

Searching on two sorted arrays is fortunately very easy. For each tile index i, we look for splitters Ai and Bi 

such that Ai + Bi = NV * i. CTA i loads input from the interval (Ai, Ai+1) and insertion indices from the 

interval (NV * i - Ai, NV * (i + 1) - Ai+1). The total number of loaded items is clearly NV. If we can simply 

calculate the partitions Ai we'll have solved Bulk Insert.

Merge Path

include/device/ctasearch.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctasearch.cuh

template<MgpuBounds Bounds, typename It1, typename It2, typename Comp>
MGPU_HOST_DEVICE int MergePath(It1 a, int aCount, It2 b, int bCount, 
int diag,
    Comp comp) {
 
    typedef typename std::iterator_traits<It1>::value_type T;
    int begin = max(0, diag - bCount);
    int end = min(diag, aCount);
 
    while(begin < end) {
        int mid = (begin + end)>> 1;
        T aKey = a[mid];
        T bKey = b[diag - 1 - mid];
        bool pred = (MgpuBoundsUpper == Bounds) ? 
            comp(aKey, bKey) : 
            !comp(bKey, aKey);
        if(pred) begin = mid + 1;
        else end = mid;
    }
    return begin;
}

MergePath is a binary search over two sorted arrays that uses a constraint, diag, to reduce the search to 
one dimension. The search resembles the basic binary search in that we establish an interval over the array a, 
and loop while begin < end. The primary difference is that instead of comparing a[mid] to a fixed 
query, we utilize the constraint and compare to b[diag - 1 - mid].

This routine is sound analytically but makes little intuitive sense. Fortunately a 2012 paper Merge Path - 
Parallel Merging Made Simple by Saher Odeh, Oded Green, Zahi Mwassi, Oz Shmuli, and Yitzhak Birk, 
provides a visualization for this search to help understand and apply it in diverse situations.

Further Reading: Read Merge Path - Parallel Merging Made Simple to better understand a partitioning 
strategy that's employed repeatedly in the MGPU algorithms.

http://www.cc.gatech.edu/~ogreen3/_docs/Merge_Path_-_Parallel_Merging_Made_Simple.pdf
http://www.cc.gatech.edu/~ogreen3/_docs/Merge_Path_-_Parallel_Merging_Made_Simple.pdf


Consider merging the array on the top with the array on the right. Start with the cursor (a pair of pointers to 
the head of each list) at the upper-left corner. Compare the heads of the lists and advance in the direction of 
the smaller key; advance to the right if the keys are equal. The curve in green is the Merge Path. It is a 
visualization of the history of decisions in executing a serial merge.

We draw red cross-diagonals over the Merge Path curve. The portion of the path enclosed by consecutive 
cross-diagonals map the intervals from each source array that contribute to one interval of the merge output. 



In this example, the first partition is assigned keys 0, 0 from A and 0, 0 from B; the second interval is 
assigned key 1 from A and keys 1, 1, 1 from B; etc. All partitions are assigned exactly four inputs (and 
because of the nature of the merge function, four outputs), and each partition can be processed in parallel 
without communication. If keys in A and B match, the use of use of lower-bound semantics in the 
MergePath binary search assigns the matching keys in A to the earlier partitions and the matching keys in 
B to the later partitions, which agrees with the behavior of std::merge.

This illustration shows a merge divided into six partitions of equal size plus one partial partition at the lower-
right. The eight cross-diagonals (in red) intersect the Merge Path to establish tile partitions (in black). 
Because the Merge Path can only move down and right (and not diagonal), the length of the curve bounded 
by two consecutive cross-diagonals is constant and equal to the inter-diagonal spacing. The intersection of 
the Merge Path projects upwards to define the interval of values in A inside the partition; it projects to the 
right to define the interval of values in B. The number of steps required for the binary search is the log of the 
cross-diagonal length.

Important: Merge Path is the history of comparisons made during a sequential merge operation. We want to 
find the intersection of the Merge Path with regularly-spaced cross-diagonals without actually constructing 
the Merge Path. We binary search along cross-diagonals, sampling and comparing elements from inputs A 
and B until we've ascertained where the Merge Path would be, if we had constructed it.

In the CUDA code above, the constraint parameter diag is the distance from the origin that the cross-
diagonal intersects the x-axis. Consider tile 3's partition in this figure. A search along cross-diagonal 3 
returns the pair (a3, b3), and the search along cross-diagonal 4 returns (a4, b4). Because of the constraint that 

ai + bi = diag, we need only deal with the coverage interval of A: (a3, a4). When needed, the coverage 

interval of B is can be computed as (diag3 - a3, diag4 - a4).

How is this applicable to load-balancing Bulk Insert? Consider the array across the top (A) as the sorted list 
of insertion indices; the array on the right (B) is a counting_iterator<int> representing the positions 
of the source elements. Bulk Insert is essentially a merge operation where the keys of A are the insertion 
indices and the keys of B are the natural numbers. Computing the intersection of the Merge Path curve with 
the cross-diagonal using a constrained binary search solves Bulk Insert's partioning needs.



Bulk insert algorithm

After mapping a source and insert elements into each tile we can run the actual Bulk Insert logic. This is a 
scan-oriented kernel and follows the patterns of Bulk Remove.

Insert indices:
    0:     1   12   13   14   14   18   20   38   39   44
   10:    45   50   50   50   54   56   59   63   68   69
   20:    74   75   84   84   88  111  111  119  121  123
   30:   126  127  144  153  157  159  163  169  169  175
   40:   178  183  190  194  195  196  196  201  219  219
   50:   253  256  259  262  262  266  272  273  278  283
   60:   284  291  296  297  302  303  306  306  317  318
   70:   318  319  319  320  320  323  326  329  330  334
   80:   340  349  352  363  366  367  369  374  381  383
   90:   383  384  386  388  388  389  393  398  398  399
Tile size = 100   ACount = 22   BCount = 78

Counters:
    0:     0    1    0    0    0    0    0    0    0    0
   10:     0    0    0    1    0    1    0    1    1    0
   20:     0    0    0    1    0    0    1    0    0    0
   30:     0    0    0    0    0    0    0    0    0    0
   40:     0    0    0    0    0    1    0    1    0    0
   50:     0    0    0    1    0    1    0    0    0    0
   60:     0    1    1    1    0    0    0    0    1    0
   70:     0    1    0    0    0    1    0    0    0    0
   80:     1    0    0    0    0    0    1    0    1    0
   90:     0    0    0    0    1    0    1    0    0    0 

Our example starts with 400 source elements and 100 insertions. A Merge Path search determines that the 
first tile of 100 elements (NV = 100) should load 22 from A (the insertion array) and 78 from B (the source 
array). This should be clear from the figure—the insert indices range from 1 to 75 and the implicit source 
indices range from 0 to 77. Adjusting the Merge Path either way would invalidate the ordering. For example, 
setting ACount to 23 and BCount to 77 would attempt an insert at position 84 when only source elements 0 
to 76 have been loaded.

We allocate and initialize 100 flags to zero. Ones are poked in for each insert index at the index's location 
plus its value. That is, indices[0] = 1 sets the flag at location 1 (0 + 1); indices[1] = 12 sets the flag at location 
13 (1 + 12); indices[2] = 13 sets the flag at location 15 (2 + 13); and so on. Duplicate insertion indices are 
admissible, as they won't set flags at the same location, because their ranks are different.

 Scan of counters:
    0:     0    0    1    1    1    1    1    1    1    1
   10:     1    1    1    1    2    2    3    3    4    5
   20:     5    5    5    5    6    6    6    7    7    7
   30:     7    7    7    7    7    7    7    7    7    7
   40:     7    7    7    7    7    7    8    8    9    9
   50:     9    9    9    9   10   10   11   11   11   11
   60:    11   11   12   13   14   14   14   14   14   15
   70:    15   15   16   16   16   16   17   17   17   17
   80:    17   18   18   18   18   18   18   19   19   20
   90:    20   20   20   20   20   21   21   22   22   22

Gather indices:
    0:    22    0   23   24   25   26   27   28   29   30
   10:    31   32   33    1   34    2   35    3    4   36
   20:    37   38   39    5   40   41    6   42   43   44
   30:    45   46   47   48   49   50   51   52   53   54



   40:    55   56   57   58   59    7   60    8   61   62
   50:    63   64   65    9   66   10   67   68   69   70
   60:    71   11   12   13   72   73   74   75   14   76
   70:    77   15   78   79   80   16   81   82   83   84
   80:    17   85   86   87   88   89   18   90   19   91
   90:    92   93   94   95   20   96   21   97   98   99

As with Bulk Remove, an exclusive scan is computed across the flags. If the flag was set (indicating an 
insertion at this position) then we'll gather at that scan offset. If the flag was cleared we subtract the scan 
from the output rank and add aCount (the number of inserted vertices). This segregates the gather into two 
forms: inserted values (in green) are referenced by their ranks; source values (in black) are referenced by 
their ranks plus the number of inserted values. 

The loadstore.cuh function DeviceTransferMergeValues interprets gather indices in this format and 
performs the data transfer. This mechanism is used by other merge-like functions (merge, mergesort, 
segmented sort, multisets) to move the values in key/value pair operations.

Bulk insert host function and kernel

include/kernels/bulkinsert.cuh

template<typename InputIt1, typename IndicesIt, typename InputIt2,
    typename OutputIt>
MGPU_HOST void BulkInsert(InputIt1 a_global, IndicesIt indices_global, 
    int aCount, InputIt2 b_global, int bCount, OutputIt dest_global,
    CudaContext& context) {
 
    const int NT = 128;
    const int VT = 7;
    typedef LaunchBoxVT<NT, VT> Tuning;
    int2 launch = Tuning::GetLaunchParams(context);
    const int NV = launch.x * launch.y;
 
    MGPU_MEM(int) partitionsDevice = 
MergePathPartitions<MgpuBoundsLower>(
        indices_global, aCount, mgpu::counting_iterator<int>(0), 
bCount, NV, 0,
        mgpu::less<int>(), context);
 
    int numBlocks = MGPU_DIV_UP(aCount + bCount, NV);
    KernelBulkInsert<Tuning><<<numBlocks, launch.x, 0, 
context.Stream()>>>(
        a_global, indices_global, aCount, b_global, bCount, 
        partitionsDevice->get(), dest_global);
}

The host function creates a LaunchBoxVT type to provide tuning parameters to the kernel. We choose a 
configuration with 128 threads and 7 values per thread as a default, but users who want performance are 
encouraged to customize this type for their own architectures, data types, and array sizes.

The call to MergePathPartitions runs a MergePath binary search to assign intervals of insert 
indices and intervals of source data to each CTA. You can imagine the insert indices written along the top of 

http://nvlabs.github.io/moderngpu/mergesort.html#mergepathpartitions
http://nvlabs.github.io/moderngpu/performance.html#launchbox
https://github.com/NVlabs/moderngpu/blob/master/include/kernels/bulkinsert.cuh
https://github.com/NVlabs/moderngpu/blob/master/include/device/loadstore.cuh


the Merge Path diagram and the natural numbers (corresponding to the positions of the source data) written 
along the right. A cross-diagonal every NV elements finds an equal partitioning of insert and source elements 
for each CTA. The intersections of the cross-diagonals with the Merge Path are stored in 
partitionsDevice, which is passed to the kernel as mp_global.

include/kernels/bulkinsert.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/bulkinsert.cuh

// Insert the values from a_global into the positions marked by 
indices_global.
template<typename Tuning, typename InputIt1, typename IndicesIt, 
    typename InputIt2, typename OutputIt>
MGPU_LAUNCH_BOUNDS void KernelBulkInsert(InputIt1 a_global, 
    IndicesIt indices_global, int aCount, InputIt2 b_global, int 
bCount, 
    const int* mp_global, OutputIt dest_global) {
 
    typedef MGPU_LAUNCH_PARAMS Params;
    typedef typename std::iterator_traits<InputIt1>::value_type T;
    const int NT = Params::NT;
    const int VT = Params::VT;
    const int NV = NT * VT;
 
    typedef CTAScan<NT, ScanOpAdd> S;
    union Shared {
        int indices[NV];
        typename S::Storage scan;
    };
    __shared__ Shared shared;
 
    int tid = threadIdx.x;
    int block = blockIdx.x;
 
    int4 range = ComputeMergeRange(aCount, bCount, block, 0, NV, 
mp_global);
    int a0 = range.x;       // A is array of values to insert.
    int a1 = range.y;
    int b0 = range.z;       // B is source array.
    int b1 = range.w;
    aCount = a1 - a0;
    bCount = b1 - b0;
 
    // Initialize the indices to 0.
    #pragma unroll
    for(int i = 0; i < VT; ++i)
        shared.indices[NT * i + tid] = 0;
    __syncthreads();
 
    // Load the indices.
    int indices[VT];
    DeviceGlobalToReg<NT, VT>(aCount, indices_global + a0, tid, 
indices);
 
    // Set the counters for all the loaded indices. This has the 
effect of 
    // pushing the scanned values to the right, causing the B data to 
be 
    // inserted to the right of each insertion point.
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        int index = NT * i + tid;



KernelBulkInsert is a straight-forward implementation of the routine described in the algorithm 
section. Because this function was written in the two-phase decomposition methodology, the important 
business of mapping work to CUDA cores has already been performed by the time this kernel is launched. 
ComputeMergeRange reassembles source intervals from the mp_global array, communicating the 
partitoining of the previous launch to the work-logic kernel.

Kernels are tuned by increasing VT to amortize per-CTA and per-thread costs, improving work-efficiency. 
Consequently most programs run underoccupied, especially on Kepler. A nice property of both Bulk Remove 
and Bulk Insert is that only the 32-bit indices are staged in shared memory. Specializing on 64-bit data-types 
doesn't decrease occupancy on Kepler, which has sufficient RF capacity to accommodate the larger types. 
Working on 64-bit types increases the atom of transfer to and from global memory while keeping the cost of 
compute constant. This results in significantly higher throughput (up to 220 GB/s for Bulk Remove on GTX 
Titan) on 64-bit types compared to 32-bit types (158 GB/s).

http://nvlabs.github.io/moderngpu/mergesort.html#mergepartitioning
http://nvlabs.github.io/moderngpu/bulkinsert.html#bulkinsertalgorithm
http://nvlabs.github.io/moderngpu/bulkinsert.html#bulkinsertalgorithm


6. Merge
Merge two sorted sequences in parallel. This implementation supports custom iterators and comparators. It 
achieves throughputs greater than half peak bandwidth. MGPU's two-phase approach to scheduling is 
developed here.

Benchmark and usage

Merge Keys benchmark from tests/benchmarkmerge.cu

Merge keys demonstration from tests/demo.cu

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarkmerge.cu


142
143
144
145
146
147
148
149
150
151
152
153
154
155
156
157
158
159
160
161
162
163

void DemoMergeKeys(CudaContext& context) {
    printf("\n\nMERGE KEYS DEMONSTRATION:\n\n");
     
    // Use CudaContext::SortRandom to generate 100 sorted random 
integers 
    // between 0 and 99.
    int N = 100;
    MGPU_MEM(int) aData = context.SortRandom<int>(N, 0, 99);
    MGPU_MEM(int) bData = context.SortRandom<int>(N, 0, 99);
 
    printf("A:\n");
    PrintArray(*aData, "%4d", 10);
    printf("\nB:\n");
    PrintArray(*bData, "%4d", 10);
 
    // Merge the two sorted sequences into one.
    MGPU_MEM(int) cData = context.Malloc<int>(2 * N);
    MergeKeys(aData->get(), N, bData->get(), N, cData->get(), 
mgpu::less<int>(),
        context);
 
    printf("\nMerged array:\n");
    PrintArray(*cData, "%4d", 10);
}

MERGE KEYS DEMONSTRATION:

A:
    0:     0    0    3    4    4    7    7    7    8    8
   10:     9   10   11   12   13   13   13   14   14   15
   20:    16   16   18   18   19   22   23   23   25   25
   30:    26   26   28   31   34   34   35   36   38   39
   40:    40   43   43   43   44   44   45   46   47   49
   50:    50   50   50   51   52   52   53   53   54   54
   60:    55   57   60   60   62   62   62   65   66   67
   70:    68   68   71   72   74   74   76   77   79   80
   80:    80   81   82   82   85   85   85   86   86   86
   90:    91   91   91   92   96   97   97   98   98   99

B:
    0:     1    3    4    4    4    5    5    8    9   10
   10:    11   12   13   16   16   18   18   21   22   23
   20:    24   24   25   27   28   29   30   30   30   31
   30:    32   33   34   34   35   36   36   36   37   37
   40:    38   38   39   40   40   41   43   43   44   45
   50:    45   48   48   48   49   49   49   49   50   51
   60:    54   54   55   57   62   62   64   64   65   66
   70:    68   71   73   74   75   75   77   78   78   79
   80:    80   81   81   81   82   82   87   87   88   90
   90:    90   90   91   91   92   94   94   95   95   98

Merged array:
    0:     0    0    1    3    3    4    4    4    4    4
   10:     5    5    7    7    7    8    8    8    9    9
   20:    10   10   11   11   12   12   13   13   13   13
   30:    14   14   15   16   16   16   16   18   18   18



   40:    18   19   21   22   22   23   23   23   24   24
   50:    25   25   25   26   26   27   28   28   29   30
   60:    30   30   31   31   32   33   34   34   34   34
   70:    35   35   36   36   36   36   37   37   38   38
   80:    38   39   39   40   40   40   41   43   43   43
   90:    43   43   44   44   44   45   45   45   46   47
  100:    48   48   48   49   49   49   49   49   50   50
  110:    50   50   51   51   52   52   53   53   54   54
  120:    54   54   55   55   57   57   60   60   62   62
  130:    62   62   62   64   64   65   65   66   66   67
  140:    68   68   68   71   71   72   73   74   74   74
  150:    75   75   76   77   77   78   78   79   79   80
  160:    80   80   81   81   81   81   82   82   82   82
  170:    85   85   85   86   86   86   87   87   88   90
  180:    90   90   91   91   91   91   91   92   92   94
  190:    94   95   95   96   97   97   98   98   98   99

Merge Pairs benchmark from tests/benchmarkmerge.cu

Merge pairs demonstration from tests/demo.cu

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarkmerge.cu

void DemoMergePairs(CudaContext& context) {
    printf("\n\nMERGE PAIRS DEMONSTRATION:\n\n");
 
    int N = 100;
    MGPU_MEM(int) aKeys = context.SortRandom<int>(N, 0, 99);
    MGPU_MEM(int) bKeys = context.SortRandom<int>(N, 0, 99);
    MGPU_MEM(int) aVals = context.FillAscending<int>(N, 0, 1);
    MGPU_MEM(int) bVals = context.FillAscending<int>(N, N, 1);
 
    printf("A:\n");
    PrintArray(*aKeys, "%4d", 10);
    printf("\nB:\n");
    PrintArray(*bKeys, "%4d", 10);
 
    // Merge the two sorted sequences into one.
    MGPU_MEM(int) cKeys = context.Malloc<int>(2 * N);
    MGPU_MEM(int) cVals = context.Malloc<int>(2 * N);
    MergePairs(aKeys->get(), aVals->get(), N, bKeys->get(), bVals-
>get(), N,
        cKeys->get(), cVals->get(), mgpu::less<int>(), context);
 
    printf("\nMerged keys:\n");
    PrintArray(*cKeys, "%4d", 10);
    printf("\nMerged values (0-99 are A indices, 100-199 are B 
indices).\n");
    PrintArray(*cVals, "%4d", 10);
}

MERGE PAIRS DEMONSTRATION:

A:
    0:     1    1    2    4    8    8   10   11   11   11
   10:    13   14   14   16   16   17   18   18   19   19
   20:    19   20   21   22   22   22   23   23   23   24
   30:    24   25   26   26   26   28   29   30   31   31
   40:    32   34   35   35   37   38   40   42   42   43
   50:    43   43   44   44   45   47   47   47   48   50
   60:    53   54   54   55   57   58   58   59   60   62
   70:    63   64   64   65   68   70   71   72   73   76
   80:    77   78   79   79   80   81   83   84   87   88
   90:    90   90   92   92   93   94   96   97   99   99

B:
    0:     0    1    1    2    3    3    6    9    9   10
   10:    12   13   15   16   17   18   18   19   22   23
   20:    23   23   23   24   25   26   26   28   29   29
   30:    31   31   32   32   33   33   33   35   36   38
   40:    39   40   40   41   42   47   47   47   48   48
   50:    48   49   50   50   50   50   51   51   52   54
   60:    57   58   59   60   60   61   61   62   63   65
   70:    67   67   68   69   71   71   71   72   74   74
   80:    76   76   77   79   80   84   85   88   88   88
   90:    89   90   90   91   93   95   96   96   97   98

Merged keys:



    0:     0    1    1    1    1    2    2    3    3    4
   10:     6    8    8    9    9   10   10   11   11   11
   20:    12   13   13   14   14   15   16   16   16   17
   30:    17   18   18   18   18   19   19   19   19   20
   40:    21   22   22   22   22   23   23   23   23   23
   50:    23   23   24   24   24   25   25   26   26   26
   60:    26   26   28   28   29   29   29   30   31   31
   70:    31   31   32   32   32   33   33   33   34   35
   80:    35   35   36   37   38   38   39   40   40   40
   90:    41   42   42   42   43   43   43   44   44   45
  100:    47   47   47   47   47   47   48   48   48   48
  110:    49   50   50   50   50   50   51   51   52   53
  120:    54   54   54   55   57   57   58   58   58   59
  130:    59   60   60   60   61   61   62   62   63   63
  140:    64   64   65   65   67   67   68   68   69   70
  150:    71   71   71   71   72   72   73   74   74   76
  160:    76   76   77   77   78   79   79   79   80   80
  170:    81   83   84   84   85   87   88   88   88   88
  180:    89   90   90   90   90   91   92   92   93   93
  190:    94   95   96   96   96   97   97   98   99   99

Merged values (0-99 are A indices, 100-199 are B indices)
    0:   100    0    1  101  102    2  103  104  105    3
   10:   106    4    5  107  108    6  109    7    8    9
   20:   110   10  111   11   12  112   13   14  113   15
   30:   114   16   17  115  116   18   19   20  117   21
   40:    22   23   24   25  118   26   27   28  119  120
   50:   121  122   29   30  123   31  124   32   33   34
   60:   125  126   35  127   36  128  129   37   38   39
   70:   130  131   40  132  133  134  135  136   41   42
   80:    43  137  138   44   45  139  140   46  141  142
   90:   143   47   48  144   49   50   51   52   53   54
  100:    55   56   57  145  146  147   58  148  149  150
  110:   151   59  152  153  154  155  156  157  158   60
  120:    61   62  159   63   64  160   65   66  161   67
  130:   162   68  163  164  165  166   69  167   70  168
  140:    71   72   73  169  170  171   74  172  173   75
  150:    76  174  175  176   77  177   78  178  179   79
  160:   180  181   80  182   81   82   83  183   84  184
  170:    85   86   87  185  186   88   89  187  188  189
  180:   190   90   91  191  192  193   92   93   94  194
  190:    95  195   96  196  197   97  198  199   98   99

Host functions

include/mgpuhost.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/mgpuhost.cuh

//////////////////////////////////////////////////////////////////////
//////////
// kernels/merge.cuh
 
// MergeKeys merges two arrays of sorted inputs with C++-comparison 
semantics.
// aCount items from aKeys_global and bCount items from bKeys_global 
are merged
// into aCount + bCount items in keys_global.
// Comp is a comparator type supporting strict weak ordering.
// If !comp(b, a), then a is placed before b in the output.
template<typename KeysIt1, typename KeysIt2, typename KeysIt3, typename 
Comp>
MGPU_HOST void MergeKeys(KeysIt1 aKeys_global, int aCount, KeysIt2 
bKeys_global,
    int bCount, KeysIt3 keys_global, Comp comp, CudaContext& context);
 
// MergeKeys specialized with Comp = mgpu::less<T>.
template<typename KeysIt1, typename KeysIt2, typename KeysIt3>
MGPU_HOST void MergeKeys(KeysIt1 aKeys_global, int aCount, KeysIt2 
bKeys_global,
    int bCount, KeysIt3 keys_global, CudaContext& context);
 
// MergePairs merges two arrays of sorted inputs by key and copies 
values.
// If !comp(bKey, aKey), then aKey is placed before bKey in the 
output, and
// the corresponding aData is placed before bData. This corresponds to 
*_by_key
// functions in Thrust.
template<typename KeysIt1, typename KeysIt2, typename KeysIt3, typename 
ValsIt1,
    typename ValsIt2, typename ValsIt3, typename Comp>
MGPU_HOST void MergePairs(KeysIt1 aKeys_global, ValsIt1 aVals_global, 
    int aCount, KeysIt2 bKeys_global, ValsIt2 bVals_global, int bCount,
    KeysIt3 keys_global, ValsIt3 vals_global, Comp comp, CudaContext& 
context);
 
// MergePairs specialized with Comp = mgpu::less<T>.
template<typename KeysIt1, typename KeysIt2, typename KeysIt3, typename 
ValsIt1,
    typename ValsIt2, typename ValsIt3>
MGPU_HOST void MergePairs(KeysIt1 aKeys_global, ValsIt1 aVals_global, 
    int aCount, KeysIt2 bKeys_global, ValsIt2 bVals_global, int bCount,
    KeysIt3 keys_global, ValsIt3 vals_global, CudaContext& context);

Two-stage design

Further Reading: Read GPU Merge Path - A GPU Merging Algorithm by Oded Green, Robert McColl, and 
David A. Bader for another discussion on using Merge Path partitioning to implement merge with CUDA.

http://www.cc.gatech.edu/~ogreen3/_docs/GPU%20Merge%20Path%20-%20A%20GPU%20Merging%20Algorithm.pdf


CPU Merge implementation

template<typename T, typename Comp>
void CPUMerge(const T* a, int aCount, const T* b, int bCount, T* dest,
    Comp comp) {
 
    int count = aCount + bCount;
    int ai = 0, bi = 0;
    for(int i = 0; i < count; ++i) {
        bool p;
        if(bi >= bCount) p = true;
        else if(ai >= aCount) p = false;
        else p = !comp(b[bi], a[ai]);
 
        dest[i] = p ? a[ai++] : b[bi++];
    }
}

Merge is the simplest function that is constructed in the two-phase style promoted by this project. Developing 
algorithms in the two-phase style begins with writing down a serial implementation. CPUMerge is a good 
point of reference because it consumes one input and emits one output per iteration. Our goal is to:

1. Divide the domain into partitions of exactly the same size. We use the Merge Path ideas covered on 
the previous page to assist with partitioning and scheduling. A coarse-grained search over the inputs 
in global memory breaks the problem into tiles with workloads of constant size. A fine-grained search 
over the inputs in shared memory breaks the problem into threads with workloads of constant size.

2. Develop a serial merge, like CPUMerge above, that is executed in parallel and in isolation by each 
thread to process distinct intervals of the problem. Rather than running over the entire input, as in 
CPUMerge, each thread performs exactly VT iterations, consuming VT input and emitting VT 
output. This strategy has the same linear work efficiency as a standard sequential merge (parallel 
algorithms often choose to sacrifice work efficiency to gain concurrency).

By decoupling scheduling and work, the two-phase strategy assists the programmer in developing readable 
and composable algorithms. We'll show in a future page how to replace the serial portion of the parallel 
merge to execute high-throughput vectorized sorted searches.

Algorithm

include/device/ctamerge.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctamerge.cuh
http://nvlabs.github.io/moderngpu/sortedsearch.html
http://nvlabs.github.io/moderngpu/bulkinsert.html


205
206
207
208
209
210
211
212
213
214
215
216
217
218
219
220
221
222
223
224
225
226
227
228
229
230
231
232
233
234

template<int NT, int VT, typename It1, typename It2, typename T, typename 
Comp>
MGPU_DEVICE void DeviceMergeKeysIndices(It1 a_global, It2 b_global, 
int4 range,
    int tid, T* keys_shared, T* results, int* indices, Comp comp) {
 
    int a0 = range.x;
    int a1 = range.y;
    int b0 = range.z;
    int b1 = range.w;
    int aCount = a1 - a0;
    int bCount = b1 - b0;
 
    // Load the data into shared memory.
    DeviceLoad2ToShared<NT, VT, VT>(a_global + a0, aCount, b_global + 
b0,
        bCount, tid, keys_shared);
 
    // Run a merge path to find the start of the serial merge for each 
thread.
    int diag = VT * tid;
    int mp = MergePath<MgpuBoundsLower>(keys_shared, aCount,
        keys_shared + aCount, bCount, diag, comp);
 
    // Compute the ranges of the sources in shared memory.
    int a0tid = mp;
    int a1tid = aCount;
    int b0tid = aCount + diag - mp;
    int b1tid = aCount + bCount;
 
    // Serial merge into register.
    SerialMerge<VT, true>(keys_shared, a0tid, a1tid, b0tid, b1tid, 
results,
        indices, comp);
}

MGPU Merge merges two sorted inputs with C++ std::merge ordering semantics. As in Bulk Insert, the 
source inputs are partitioned into equal size-interval pairs by calling MergePathPartitions. We 
double-down on this divide-and-conquer strategy by calling MergePath a second time, locally searching 
over the keys in a tile. 

DeviceMergeKeysIndices is a re-usable CTA-level function that merges keys provided in shared 
memory. The caller secifies the tile's intervals over A and B in the range argument. range is derived by 
ComputeMergeRange using the intersections of the tile's cross-diagonals with the Merge Path, as 
illustrated here. DeviceLoad2ToShared performs an optimized, unrolled, cooperative load of a variable 
number of contiguous elements from two input arrays. Loaded keys are stored in shared memory: A's 
contributions in (0, aCount) and B's contributions in (aCount, aCount + bCount).

http://nvlabs.github.io/moderngpu/bulkinsert.html#mergepath
http://nvlabs.github.io/moderngpu/mergesort.html#mergepartitioning
http://nvlabs.github.io/moderngpu/mergesort.html#mergepathpartitions
http://nvlabs.github.io/moderngpu/bulkinsert.html#bulkinsert


MergePath is called by all threads in parallel to find their individual partitions. This is a faster search than 
the global partitioning search because shared memory has much lower latency, and intra-CTA cross-
diagonals are much shorter than global cross-diagonals, resulting in binary searches that converge after fewer 
iterations. The intra-CTA Merge Path searches are conducted in the tile's local coordinate system. Cross-
diagonals are given indices VT * tid.

The starting cursor for each thread (a0tid and b0tid) is handed to SerialMerge, which loads keys from 
shared memory, merges them, and returns a fragment of the result in register.

include/device/ctamerge.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctamerge.cuh

template<int VT, bool RangeCheck, typename T, typename Comp>
MGPU_DEVICE void SerialMerge(const T* keys_shared, int aBegin, int aEnd,
    int bBegin, int bEnd, T* results, int* indices, Comp comp) { 
 
    T aKey = keys_shared[aBegin];
    T bKey = keys_shared[bBegin];
 
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        bool p;
        if(RangeCheck) 
            p = (bBegin >= bEnd) || ((aBegin < aEnd) && !comp(bKey, 
aKey));
        else
            p = !comp(bKey, aKey);
 
        results[i] = p ? aKey : bKey;
        indices[i] = p ? aBegin : bBegin;
 
        if(p) aKey = keys_shared[++aBegin];
        else bKey = keys_shared[++bBegin];
    }
    __syncthreads();
}

Partitioning doesn't really differentiate merge from similar functions, as all it does is handle scheduling. The 
soul of this function is SerialMerge. Incredible throughput is achieved because Merge Path isn't simply a 
very good partitioning; it's an exact partition. The merge kernel is tuned to a specific (odd) number of values 
per thread. For a CTA with 128 threads (NT) and 11 values per thread (VT), each tile loads and merges 1408 
inputs (NV). These inputs aren't simply merged cooperatively, though. They are merged independently by the 
128 threads, 11 per thread, which is far better.

Because each thread merges precisely 11 elements, the SerialMerge routine can unroll its loop. Accesses 
to the output arrays results and indices are now static (the iterator for unrolled loops is treated as a 
constant by the compiler). Because we're using only static indexing, the outputs can be stored in register 
rather than shared memory. RF capacity is much higher than shared memory capacity, and the performance 
tuning strategy of increasing grain size to amortize partitioning costs always results in underoccupied kernels. 
Storing outputs in register cuts the kernel's shared memory footprint in half, doubling occupancy, and 
boosting performance.

Important: Structure your code to only dynamically index either the sources or the destinations (not both). 
Use loop unrolling to statically index the complementary operations in register, then synchronize and swap. 
Exact partitioning facilitates this pattern, which doubles occupancy to improve latency-hiding.

Keys are returned into results. indices (the locations of keys in shared memory) are also returned to 
facilitate a value gather for sort-by-key. For key-only merge, operations involving indices should be 
eliminated by the compiler.

Note that the next item in each sequence is fetched prior to the start of the next iteration. This reduces two 
shared loads per thread to just one, which reduces bank conflicts across the warp. Unfortunately it may also 
cause us to read off the end of the B array. To prevent an illegal access failure in the kernel, allocate at leats 
one extra slot in shared memory. This doesn't compromise occupancy at all, because we use odd numbered 



VT parameters—we can reserve up to a full additional slot per thread before the extra provisioning reduces 
the number of concurrent CTAs per SM.

Important: If fetching the next iteration's data at the end of the loop body, allocate an extra slot in shared 
memory to prevent illegal access violations.

include/device/ctamerge.cuh

template<int NT, int VT, bool HasValues, typename KeysIt1, typename 
KeysIt2,
    typename KeysIt3, typename ValsIt1, typename ValsIt2, typename 
KeyType,
    typename ValsIt3, typename Comp>
MGPU_DEVICE void DeviceMerge(KeysIt1 aKeys_global, ValsIt1 
aVals_global, 
    KeysIt2 bKeys_global, ValsIt2 bVals_global, int tid, int block, 
int4 range,
    KeyType* keys_shared, int* indices_shared, KeysIt3 keys_global,
    ValsIt3 vals_global, Comp comp) {
 
    KeyType results[VT];
    int indices[VT];
    DeviceMergeKeysIndices<NT, VT>(aKeys_global, bKeys_global, range, 
tid, 
        keys_shared, results, indices, comp);
 
    // Store merge results back to shared memory.
    DeviceThreadToShared<VT>(results, tid, keys_shared);
 
    // Store merged keys to global memory.
    int aCount = range.y - range.x;
    int bCount = range.w - range.z;
    DeviceSharedToGlobal<NT, VT>(aCount + bCount, keys_shared, tid, 
        keys_global + NT * VT * block);
 
    // Copy the values.
    if(HasValues) {
        DeviceThreadToShared<VT>(indices, tid, indices_shared);
        DeviceTransferMergeValues<NT, VT>(aCount + bCount,
          aVals_global + range.x, bVals_global + range.z, aCount,
          indices_shared, tid, vals_global + NT * VT * block);
    }
}

DeviceMerge, one level closer to the kernel, invokes DeviceMergeKeysIndices and receives the 
merged results and indices in register. Each thread uses DeviceThreadtoShared to store its merged 
keys to shared memory at VT * tid + i, synchronizes, and calls DeviceSharedToGlobal to 
cooperatively make coalesced stores to the destination array. DeviceTransferMergeValues 
(discussed here) uses the indices to gather values from global memory and store them back, coalesced, to 
vals_global.

DeviceMerge does the heavy lifting for both MGPU's merge and mergesort kernels.

http://nvlabs.github.io/moderngpu/bulkinsert.html#bulkinsert
https://github.com/NVlabs/moderngpu/blob/master/include/device/ctamerge.cuh


To recap merge:

1. Prior to the merge kernel, use MergePathPartitions for coarse-grained exact partitioning.

2. At the start of the kernel, ComputeMergeRange determines the intervals to load from arrays A and 
B. DeviceLoad2ToShared loads these into shared memory; first A, then B.

3. MergePath searches keys in shared memory to find a fine-grained partitioning of data, with VT 
items per thread.

4. Each thread makes VT trips through an unrolled loop, dynamically indexing into shared memory 
retrieving keys, comparing them, and emitting the smaller key to an array in register, using the static 
loop iterator.

5. After synchronization each thread writes its values back at VT * tid + i (thread order). The values are 
cooperatively transferred to the destination in global memory using coalesced stores.

6. Indices are stored to shared memory (writing from thread order into strided order). 
DeviceTransferMergeValues uses these to gather merged values from the input. It makes 
coalesced stores to the destination.

Much of the MGPU Merge implementation is shared with Mergesort—these portions are covered on the next 
page.



7. Mergesort
A high-throughput mergesort that is perfectly load-balanced over all threads. Develops partitioning and 
scheduling functions that are used throughout these pages. This mergesort is the basis for high-performance 
segmented and locality sorts that work with structured data (i.e. non-uniformly random).

Benchmark and usage

Sort Keys benchmark from tests/benchmarksort.cu

Sort keys demonstration from tests/demo.cu

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarksort.cu

void DemoSortKeys(CudaContext& context) {
    printf("\n\nSORT KEYS DEMONSTRATION:\n\n");
 
    // Use CudaContext::GenRandom to generate 100 random integers 
between 0 and
    // 199.
    int N = 100;
    MGPU_MEM(int) data = context.GenRandom<int>(N, 0, 99);
     
    printf("Input:\n");
    PrintArray(*data, "%4d", 10);
 
    // Mergesort keys.
    MergesortKeys(data->get(), N, mgpu::less<int>(), context);
 
    printf("\nSorted output:\n");
    PrintArray(*data, "%4d", 10);
}

SORT KEYS DEMONSTRATION:

Input:
    0:     5   95   68   53    4   87    7   93   52   66
   10:     9   28   81    6   81   23   72   70   14   19
   20:    65   42   51   93   97   14   64   64   80   47
   30:    45   43   43   24   82   50    8   90   13    7
   40:    17   71   39   61   83   18   80   39    6   27
   50:    39   85   52   90   41   61   65   18   62   51
   60:    29   82   43   35    1   81   98   29   16   17
   70:    10   49   37   19   19   86   48   20   33   61
   80:    95   87   92   39    5   94   73   16   26   97
   90:    42   56   54   59   94   13   41   56   98   55

Sorted output:
    0:     1    4    5    5    6    6    7    7    8    9
   10:    10   13   13   14   14   16   16   17   17   18
   20:    18   19   19   19   20   23   24   26   27   28
   30:    29   29   33   35   37   39   39   39   39   41
   40:    41   42   42   43   43   43   45   47   48   49
   50:    50   51   51   52   52   53   54   55   56   56
   60:    59   61   61   61   62   64   64   65   65   66
   70:    68   70   71   72   73   80   80   81   81   81
   80:    82   82   83   85   86   87   87   90   90   92
   90:    93   93   94   94   95   95   97   97   98   98

.



Sort Pairs benchmark from tests/benchmarksort.cu

Sort pairs demonstration from tests/demo.cu

void DemoSortPairs(CudaContext& context) {
    printf("\n\nSORT PAIRS DEMONSTRATION:\n\n");
 
    // Use CudaContext::GenRandom to generate 100 random integers 
between 0 and
    // 99.
    int N = 100;
    MGPU_MEM(int) keys = context.GenRandom<int>(N, 0, 99);
    MGPU_MEM(int) vals = context.FillAscending<int>(N, 0, 1);
 
    printf("Input keys:\n");
    PrintArray(*keys, "%4d", 10);
 
    // Mergesort pairs.
    MergesortPairs(keys->get(), vals->get(), N, mgpu::less<int>(), 
context);
 
    printf("\nSorted keys:\n");
    PrintArray(*keys, "%4d", 10);
 
    printf("\nSorted values:\n");
    PrintArray(*vals, "%4d", 10);
}

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarksort.cu


Input:
    0:    30   31   70   12   66   73   53   24   69   82
   10:    66   18   17   31   12   88   99   67   17   73
   20:     3    6   56   13   88    8   66    0   19   45
   30:    36   63   46   52   98   49   15   33   85   25
   40:    64   23   37   17   19   59   42   72   48   87
   50:    12   70   58   23   22   47   38    1   58   74
   60:    25   65   29    7   61   47   26   99   82   53
   70:    98   89   73   77   34   20   58   90   10   37
   80:    90   84   87   32   81   32   26   65   59   58
   90:     2    4   42   76   31   49   16   48   17   42

Sorted keys:
    0:     0    1    2    3    4    6    7    8   10   12
   10:    12   12   13   15   16   17   17   17   17   18
   20:    19   19   20   22   23   23   24   25   25   26
   30:    26   29   30   31   31   31   32   32   33   34
   40:    36   37   37   38   42   42   42   45   46   47
   50:    47   48   48   49   49   52   53   53   56   58
   60:    58   58   58   59   59   61   63   64   65   65
   70:    66   66   66   67   69   70   70   72   73   73
   80:    73   74   76   77   81   82   82   84   85   87
   90:    87   88   88   89   90   90   98   98   99   99

Sorted values:
    0:    27   57   90   20   91   21   63   25   78    3
   10:    14   50   23   36   96   12   18   43   98   11
   20:    28   44   75   54   41   53    7   39   60   66
   30:    86   62    0    1   13   94   83   85   37   74
   40:    30   42   79   56   46   92   99   29   32   55
   50:    65   48   97   35   95   33    6   69   22   52
   60:    58   76   89   45   88   64   31   40   61   87
   70:     4   10   26   17    8    2   51   47    5   19
   80:    72   59   93   73   84    9   68   81   38   49
   90:    82   15   24   71   77   80   34   70   16   67

Host functions

include/mgpuhost.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/mgpuhost.cuh


159
160
161
162
163
164
165
166
167
168
169
170
171
172
173
174
175
176
177
178
179
180
181
182
183
184
185
186
187
188
189
190
191
192
193

//////////////////////////////////////////////////////////////////////
//////////
// kernels/mergesort.cuh
 
// MergesortKeys sorts data_global using comparator Comp.
// If !comp(b, a), then a comes before b in the output. The data is 
sorted
// in-place.
template<typename T, typename Comp>
MGPU_HOST void MergesortKeys(T* data_global, int count, Comp comp,
    CudaContext& context);
 
// MergesortKeys specialized with Comp = mgpu::less<T>.
template<typename T>
MGPU_HOST void MergesortKeys(T* data_global, int count, CudaContext& 
context);
 
// MergesortPairs sorts data by key, copying data. This corresponds to 
// sort_by_key in Thrust.
template<typename KeyType, typename ValType, typename Comp>
MGPU_HOST void MergesortPairs(KeyType* keys_global, ValType* 
values_global,
    int count, Comp comp, CudaContext& context);
 
// MergesortPairs specialized with Comp = mgpu::less<KeyType>.
template<typename KeyType, typename ValType>
MGPU_HOST void MergesortPairs(KeyType* keys_global, ValType* 
values_global,
    int count, CudaContext& context);
 
// MergesortIndices is like MergesortPairs where values_global is 
treated as
// if initialized with integers (0 ... count - 1). 
template<typename KeyType, typename Comp>
MGPU_HOST void MergesortIndices(KeyType* keys_global, int* 
values_global,
    int count, Comp comp, CudaContext& context);
 
// MergesortIndices specialized with Comp = mgpu::less<KeyType>.
template<typename KeyType>
MGPU_HOST void MergesortIndices(KeyType* keys_global, int* 
values_global,
    int count, CudaContext& context);

Algorithm

Mergesort recursively merges sorted lists until the sequence is fully sorted.

Input array is treated as sequence of sorted lists of length 1:
   13  90  83  12  96  91  22  63  30   9  54  27  18  54  99  95  
  

http://en.wikipedia.org/wiki/Merge_sort


Merge adjacent pairs of length-1 lists into sequence of length-2 lists:
   13  90  12  83  91  96  22  63   9  30  27  54  18  54  95  99
   
Merge adjacent pairs of length-2 lists into sequence of length-4 lists:
   12  13  83  90  22  63  91  96   9  27  30  54  18  54  95  99

Merge adjacent pairs of length-4 lists into sequence of length-8 lists:
   12  13  22  63  83  90  91  96   9  18  27  30  54  54  95  99
   
Merge adjacent pairs of length-8 lists into final length-16 output:   
    9  12  13  18  22  27  30  54  54  63  83  90  91  95  96  99

Although mergesort takes one unsorted array as in input, thematically it is the same as the functions that 
make up the bulk of MGPU: take two sorted inputs and emit one sorted output. This is clear if treat 
consecutive input elements as two sorted lists of length 1. Mergesort is type of multi-pass vectorized merge: 
the first iteration executes N / 2 merges with inputs of length 1; the second iteration executes N / 4 merges 
with inputs of length 2; etc.

The number of batched merge passes is log(N) and the work per pass is N. This O(N log N) work-efficiency 
hurts mergesort's scalability compared to radix sort. This mergesort implementation runs at about half the 
throughput on large arrays with 32-bit keys as the fastest GPU radix sorts. But it still clocks about 100x faster 
than calling std::stable_sort on an i7 Sandy Bridge. If the sort is truly on a critical path, it may be worth 
pulling a radix sort from B40C/FastSortSm20. Otherwise you can get by just fine with this very hackable 
mergesort, or use one of MGPU's higher-performance derivative sorts (segmented sort or locality sort).

O(N log N) complexity aside, mergesort has some notable advantages over radix sort:

1. Mergesort is a comparison sort. While radix sort requires types to have the same lexicographical order 
as integers, limiting practical use to numeric types like ints and floats, mergesort accepts a user-
defined comparator function. This allows mergesort to efficiently handle types like strings by calling 
strcmp from the comparator.

2. Mergesort scales better as keys get larger. Radix sort's work-efficiency is O(k N), where k is the key 
size in bits. Mergesort's complexity is only dependent on the number of input elements.

3. Mergesort provides extremely fast CTA-level blocksorts. On small inputs (like data mapped into an 
individual tile) the O(log N) penalty is on order with radix sort's O(k) penalty. Mergesort's simpler, 
faster inner loop allows blocksorts that are more flexible, easier to maintain, and often quicker than 
radix blocksorts.

4. Mergesort makes data progressively more sorted, never less. Even when launched on fully-sorted 
inputs, LSB radix sort randomly scatters data each pass, only putting the data into sorted order during 
the final pass. On fully sorted inputs, mergesort simply copies the data log(N) times. This pass-to-pass 
coherence allows detection of sorted intervals and early-exits to reduce unnecessary work. The O(N 
log N) complexity is only for uniform random inputs—data with exploitable structure can be sorted 
with far fewer comparisons. The next page builds special-case mergesorts that detect input structure 
and early-exit out of unnecessary operations.

Mergesort on GPU runs best when written in two distinct stages:

1. A blocksort kernel sorts random inputs into tile-length sorted lists, communicating with low-latency, 
high-bandwidth shared memory .The CTA blocksort forms a convenient re-usable component for 
MGPU's customers.

http://nvlabs.github.io/moderngpu/segsort.html
http://nvlabs.github.io/moderngpu/segsort.html#localitysortbenchmark
http://nvlabs.github.io/moderngpu/segsort.html#segsortbenchmark
https://code.google.com/p/back40computing/source/browse/#svn%2Fbranches%2FFastSortSm20%2Fb40c%2Fradix_sort


2. Multiple launches of MGPU Merge iteratively merge sorted lists, starting with the output of 1, 
communicating between passes with high-latency, high-capacity DRAM. 

Mergesort merge pass with coop = 2

Mergesort merge pass with coop = 4

Both the blocksort and global merge passes follow the structure illustrated above. Pairs of threads (or CTAs 
for the global merge passes) cooperatively merge two VT-length lists (or two NV-length lists) into one list. 
This phase is noted coop = 2, for 2 threads cooperating on each pair of input lists. During the coop = 4 
pass, 4 threads cooperatively merge two lists into one; during coop = 8, 8 threads cooperatively merge 

http://nvlabs.github.io/moderngpu/merge.html


two lists; and so on. Although there are many sorted lists in the data, threads cooperatively merge from only 
two of them at a time.

The figure at the top shows 16 sorted lists (each segment along the top and right of a square is a list). Two 
threads cooperate to merge each pair of lists (a square) into a single list (a segment) for the coop = 4 pass. 
Threads 0 and 1 merge the top and right segments of the first pair in coop = 2 into the top segment of the 
first pair in coop = 4; threads 2 and 3 merge the top and right segments of the second pair in coop = 2 
into the right segment of the first pair in coop = 4; etc. This process continues until only a single sorted 
list remains.

For each global merge pass, a call to MergePathPartitions partitions the input arrays into tile-sized 
chunks. ComputeMergeRange is invoked early in KernelMerge to identify the intervals of the input to 
load. We then hand the intra-CTA merging to DeviceMerge, developed in the previous page.

Sorting networks

MGPU's blocksort loads VT values per thread over NT threads per CTA. Merging requires dynamic 
indexing, which means shared memory. The first few rounds of merges can be replaced by an in-register 
sorting network. Batcher's odd-even mergesort sorts inputs in O(n log2 n time) using only comparisons and 
swaps. The odd-even transposition sort takes O(n2) comparisons but adds stability. These sorting networks 
are relatively inefficient, but expose great amounts of immediate parallelism, making them effective tools for 
sorting small inputs.

Although MGPU includes an implementation for Batcher's odd-even mergesort (in sortnetwork.cuh), the 
slower odd-even transposition sort is preferred, because it is stable. It takes more comparisons to sort a thread 
using this network, but the cost is small compared to the cost of the many recursive merge passes that follow.

 13   90   83   12   96   91   22   63   30    9   54   27   18   54   99   95

(13   90) (12   83) (91   96) (22   63) ( 9   30) (27   54) (18   54) (95   99)
     (12   90) (83   91) (22   96) ( 9   63) (27   30) (18   54) (54   95)
(12   13) (83   90) (22   91) ( 9   96) (27   63) (18   30) (54   54) (95   99)
     (13   83) (22   90) ( 9   91) (27   96) (18   63) (30   54) (54   95)
(12   13) (22   83) ( 9   90) (27   91) (18   96) (30   63) (54   54) (95   99)
     (13   22) ( 9   83) (27   90) (18   91) (30   96) (54   63) (54   95)
(12   13) ( 9   22) (27   83) (18   90) (30   91) (54   96) (54   63) (95   99)
     ( 9   13) (22   27) (18   83) (30   90) (54   91) (54   96) (63   95)
( 9   12) (13   22) (18   27) (30   83) (54   90) (54   91) (63   96) (95   99)
     (12   13) (18   22) (27   30) (54   83) (54   90) (63   91) (95   96)
( 9   12) (13   18) (22   27) (30   54) (54   83) (63   90) (91   95) (96   99)
     (12   13) (18   22) (27   30) (54   54) (63   83) (90   91) (95   96)
( 9   12) (13   18) (22   27) (30   54) (54   63) (83   90) (91   95) (96   99)
     (12   13) (18   22) (27   30) (54   54) (63   83) (90   91) (95   96)
( 9   12) (13   18) (22   27) (30   54) (54   63) (83   90) (91   95) (96   99)
     (12   13) (18   22) (27   30) (54   54) (63   83) (90   91) (95   96)
                                                                              
  9   12   13   18   22   27   30   54   54   63   83   90   91   95   96   99

Sorting an array of N inputs needs N transposition passes. Stability is gained by only comparing pairs of 
neighboring elements, and only exchanging when the element on the right is smaller. 16 inputs are sorted in 
this illustration. Pairs starting at offset 0 are compared and swapped on even passes; pairs starting at offset 1 
are compared and swapped on odd passes. Items in the same pair are drawn in the same color - observe that 
the second pair of values (83, 12) are swapped into (12, 83) during the first pass. The small 9 that starts in the 
middle is moved peristaltically to the front of the array.

https://github.com/NVlabs/moderngpu/blob/master/include/device/sortnetwork.cuh
http://en.wikipedia.org/wiki/Odd%E2%80%93even_sort
http://en.wikipedia.org/wiki/Batcher_odd%E2%80%93even_mergesort
http://nvlabs.github.io/moderngpu/merge.html#algorithm


We get the problem started by loading and transposing VT elements per thread into register, so that each 
thread has items VT * tid + i in register. Each thread calls OddEvenTransposeSort to sort its own set of 
elements in register. This phase of the blocksort uses no shared memory and has high ILP (all of the 
compare-and-swaps in each row can be performed in parallel).

include/device/sortnetwork.cuh

template<int VT, typename T, typename V, typename Comp>
MGPU_DEVICE void OddEvenTransposeSort(T* keys, V* values, Comp comp) {
    #pragma unroll
    for(int level = 0; level < VT; ++level) {
 
        #pragma unroll
        for(int i = 1 & level; i < VT - 1; i += 2) {
            if(comp(keys[i + 1], keys[i])) {
                mgpu::swap(keys[i], keys[i + 1]);
                mgpu::swap(values[i], values[i + 1]);
            }
        }
    }
}

Odd-even transposition sort has two nested loops: the outer iterates over the number of inputs (each row in 
the figure); the inner iterates over the number of pairs, as shown above. Unfortunately CUDA's #pragma 
unroll feature still has some kinks, and the compiler currently fails to unroll all the static indexing when 
the function is written this way. Spills result.

include/device/sortnetwork.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/sortnetwork.cuh
https://github.com/NVlabs/moderngpu/blob/master/include/device/sortnetwork.cuh

template<int I, int VT>
struct OddEvenTransposeSortT {
    // Sort segments marked by head flags. If the head flag between i 
and i + 1
    // is set (so that (2<< i) & flags is true), the values belong to 
different 
    // segments and are not swapped.
    template<typename K, typename V, typename Comp>
    static MGPU_DEVICE void Sort(K* keys, V* values, int flags, 
        Comp comp) {
        #pragma unroll
        for(int i = 1 & I; i < VT - 1; i += 2)
            if((0 == ((2<< i) & flags)) && comp(keys[i + 1], keys[i])) 
{
                mgpu::swap(keys[i], keys[i + 1]);
                mgpu::swap(values[i], values[i + 1]);
            }
        OddEvenTransposeSortT<I + 1, VT>::Sort(keys, values, flags, 
comp);
    }
};
template<int I> struct OddEvenTransposeSortT<I, I> {
    template<typename K, typename V, typename Comp>
    static MGPU_DEVICE void Sort(K* keys, V* values, int flags,
        Comp comp) { }
};
 
template<int VT, typename K, typename V, typename Comp>
MGPU_DEVICE void OddEvenTransposeSort(K* keys, V* values, Comp comp) {
    OddEvenTransposeSortT<0, VT>::Sort(keys, values, 0, comp);
}
template<int VT, typename K, typename V, typename Comp>
MGPU_DEVICE void OddEvenTransposeSortFlags(K* keys, V* values, int 
flags, 
    Comp comp) {
    OddEvenTransposeSortT<0, VT>::Sort(keys, values, flags, comp);
}

We bend to pragmatism and write the code like this. Template loop unrolling replaces the #pragma 
unroll nesting, allowing the sorting network to compile correctly. This implementation takes a bitfield of 
segment head flags to support the segmented sort (we'll revisit this part on the next page). For standard 
mergesort, the bitfield is always 0 and the associated logic is eliminated by the compiler.

Blocksort

include/device/ctamerge.cuh

template<int NT, int VT, bool HasValues, typename KeyType, typename 
ValType,
    typename Comp>
MGPU_DEVICE void CTAMergesort(KeyType threadKeys[VT], ValType 

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctamerge.cuh

threadValues[VT],
    KeyType* keys_shared, ValType* values_shared, int count, int tid,
    Comp comp) {
 
    // Stable sort the keys in the thread.
    if(VT * tid < count)
        OddEvenTransposeSort<VT>(threadKeys, threadValues, comp);
 
    // Store the locally sorted keys into shared memory.
    DeviceThreadToShared<VT>(threadKeys, tid, keys_shared);
 
    // Recursively merge lists until the entire CTA is sorted.
    DeviceBlocksortLoop<NT, VT, HasValues>(threadValues, keys_shared, 
        values_shared, tid, count, comp);
}

CTAMergesort is a reusable block-level mergesort. MGPU uses this function for the locality sort function 
in addition to standard mergesort. If the user wants to only sort keys, set HasValues to false and ValType 
to int. Use this function by passing unsorted keys and values in thread order (i.e. VT * tid + i) through 
register. On return, the same register arrays contain fully-sorted data. Shared memory is also filled with the 
sorted keys, making coalesced stores back to global memory convenient.

When sorting a partial tile, pad out the last valid thread (the last thread in the CTA with with in-range values) 
with copies of the largest key in that thread. The actual mergesort can handle partial blocks just fine: this 
padding helps keep the sorting network simple. (We only specialize the sorting network for one size, VT.)

After running the intra-thread sorting network we need to recursively merge sorted lists. Start with pairs of 
threads cooperating on one destination. Call this pass coop = 2, because two threads cooperate on each 
output list. As the list size doubles, so does the number of cooperating threads per list. We loop until only a 
single sorted list remains.

include/device/ctamerge.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctamerge.cuh
http://nvlabs.github.io/moderngpu/segsort.html

template<int NT, int VT, bool HasValues, typename KeyType, typename 
ValType,
    typename Comp>
MGPU_DEVICE void CTABlocksortLoop(ValType threadValues[VT], 
    KeyType* keys_shared, ValType* values_shared, int tid, int count, 
    Comp comp) {
 
    #pragma unroll
    for(int coop = 2; coop <= NT; coop *= 2) {
        int indices[VT];
        KeyType keys[VT];
        CTABlocksortPass<NT, VT>(keys_shared, tid, count, coop, keys,
            indices, comp);
 
        if(HasValues) {
            // Exchange the values through shared memory.
            DeviceThreadToShared<VT>(threadValues, tid, 
values_shared);
            DeviceGather<NT, VT>(NT * VT, values_shared, indices, tid, 
                threadValues);
        }
 
        // Store results in shared memory in sorted order.
        DeviceThreadToShared<VT>(keys, tid, keys_shared);
    }
}

CTABlocksortLoop is called with keys sorted into VT-length lists in shared memory. Values are passed 
in thread order in register (threadValues). Log(NT) loop iterations are made. CTABlocksortPass 
returns merged keys and indices in register. With the merged keys in safely in register, the function writes 
back the new lists with DDeviceThreadToShared. It gathers them back into register with 
DeviceGather.

Note that we only have the input or output data staged in shared memory at any one time, not both. MGPU 
Mergesort is fast because it intelligently manages occupancy this way.

include/device/ctamerge.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctamerge.cuh

template<int NT, int VT, typename T, typename Comp>
MGPU_DEVICE void CTABlocksortPass(T* keys_shared, int tid, 
    int count, int coop, T* keys, int* indices, Comp comp) {
 
    int list = ~(coop - 1) & tid;
    int diag = min(count, VT * ((coop - 1) & tid));
    int start = VT * list;
    int a0 = min(count, start);
    int b0 = min(count, start + VT * (coop / 2));
    int b1 = min(count, start + VT * coop);
 
    int p = MergePath<MgpuBoundsLower>(keys_shared + a0, b0 - a0,
        keys_shared + b0, b1 - b0, diag, comp);
 
    SerialMerge<VT, true>(keys_shared, a0 + p, b0, b0 + diag - p, b1, 
keys, 
        indices, comp);
}

Locating each thread's pair of source lists, destination list, and position within the output is the first task of 
the vectorized merge function CTABlocksortPass. ~(coop - 1) & tid masks out the bits that 
position each thread's cross-diagonal within the destination list. This expression serves as a scaled destination 
list index and is mulitplied by VT to target the start of the A list in shared memory. The expression VT * 
((coop - 1) & tid) locates each thread's cross-diagonal in the local coordinate system of the output 
list.

NT = 8, VT = 7, count = 49 (full tile)

tid         coop = 2                     coop = 4                     coop = 8          
 0:  A=( 0,  7),B=( 7, 14),d= 0   A=( 0, 14),B=(14, 28),d= 0   A=( 0, 28),B=(28, 56),d= 
0
 1:  A=( 0,  7),B=( 7, 14),d= 7   A=( 0, 14),B=(14, 28),d= 7   A=( 0, 28),B=(28, 56),d= 
7
 2:  A=(14, 21),B=(21, 28),d= 0   A=( 0, 14),B=(14, 28),d=14   A=( 0, 28),B=(28, 
56),d=14
 3:  A=(14, 21),B=(21, 28),d= 7   A=( 0, 14),B=(14, 28),d=21   A=( 0, 28),B=(28, 
56),d=21
 4:  A=(28, 35),B=(35, 42),d= 0   A=(28, 42),B=(42, 56),d= 0   A=( 0, 28),B=(28, 
56),d=28
 5:  A=(28, 35),B=(35, 42),d= 7   A=(28, 42),B=(42, 56),d= 7   A=( 0, 28),B=(28, 
56),d=35
 6:  A=(42, 49),B=(49, 56),d= 0   A=(28, 42),B=(42, 56),d=14   A=( 0, 28),B=(28, 
56),d=42
 7:  A=(42, 49),B=(49, 56),d= 7   A=(28, 42),B=(42, 56),d=21   A=( 0, 28),B=(28, 
56),d=49

These intervals illustrate all blocksort passes for a full tile with 8 threads and 7 values per thread. You can 
work them out using the bit-twiddling described above. The length of the cross-diagonal in a Merge Path 
search is constrained by the length of the shorter of the two input arrays. The cross-diagonal length doubles 
each iteration: 7, 14, 28... Correspondingly the depth (and cost) of the binary search increments as we 
progress in the mergesort: 3, 4, 5... This simple iterative approach to blocksort perfectly load balances 
scheduling and merging work over the CTA.



Flexible merge partitioning

After the blocksort we have NV-length lists sorted in global memory. We recursively run a merge on pairs of 
lists until the entire array is sorted. Just like the CTA mergesort uses code from our CTA-level merge, the 
global mergesort uses code from the global merge. 

include/device/ctamerge.cuh

74
75
76
77
78
79
80
81
82
83
84
85
86
87
88
89
90
91
92
93
94
95
96
97
98
99
100
101
102
103
104
105

// Returns (offset of a, offset of b, length of list).
MGPU_HOST_DEVICE int3 FindMergesortFrame(int coop, int block, int nv) {
    // coop is the number of CTAs or threads cooperating to merge two 
lists into
    // one. We round block down to the first CTA's ID that is working 
on this
    // merge.
    int start = ~(coop - 1) & block;
    int size = nv * (coop>> 1);
    return make_int3(nv * start, nv * start + size, size);
}
 
// Returns (a0, a1, b0, b1) into mergesort input lists between mp0 and 
mp1.
MGPU_HOST_DEVICE int4 FindMergesortInterval(int3 frame, int coop, int 
block,
    int nv, int count, int mp0, int mp1) {
 
    // Locate diag from the start of the A sublist.
    int diag = nv * block - frame.x;
    int a0 = frame.x + mp0;
    int a1 = min(count, frame.x + mp1);
    int b0 = min(count, frame.y + diag - mp0);
    int b1 = min(count, frame.y + diag + nv - mp1);
     
    // The end partition of the last block for each merge operation is 
computed
    // and stored as the begin partition for the subsequent merge. 
i.e. it is
    // the same partition but in the wrong coordinate system, so its 0 
when it
    // should be listSize. Correct that by checking if this is the 
last block
    // in this merge operation.
    if(coop - 1 == ((coop - 1) & block)) {
        a1 = min(count, frame.x + frame.z);
        b1 = min(count, frame.y + frame.z);
    }
    return make_int4(a0, a1, b0, b1);
}

For clarity and maintainability, we factor out mergesort's list-making logic into FindMergesortFrame 
and FindMergesortInterval. The former finds the start of the A list by masking out the bits below 
coop and multiplying by the grain size (either NV, or VT, depending on context). The latter function uses 

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctamerge.cuh


the intersections of the cross-diagonals with the Merge Path to calculate a CTA's or thread's input range 
within the provided A and B lists.

include/device/ctamerge.cuh

MGPU_HOST_DEVICE int4 ComputeMergeRange(int aCount, int bCount, int 
block, 
    int coop, int NV, const int* mp_global) {
 
    // Load the merge paths computed by the partitioning kernel.
    int mp0 = mp_global[block];
    int mp1 = mp_global[block + 1];
    int gid = NV * block;
 
    // Compute the ranges of the sources in global memory.
    int4 range;
    if(coop) {
        int3 frame = FindMergesortFrame(coop, block, NV);
        range = FindMergesortInterval(frame, coop, block, NV, aCount, 
mp0, 
            mp1);
    } else {
        range.x = mp0;                                          // a0
        range.y = mp1;                                          // a1
        range.z = gid - range.x;                                // b0
        range.w = min(aCount + bCount, gid + NV) - range.y;     // b1
    }
    return range;
}

ComputeMergeRange is the range-calculating entry point for merge, mergesort, segmented and locality 
sorts. This is called at the top of those respective kernels, and the A and B input-list intervals are loaded into 
shared memory. This unified function reduces the number of kernels needed to support this diversity of 
functionality.

include/kernels/merge.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/merge.cuh
https://github.com/NVlabs/moderngpu/blob/master/include/device/ctamerge.cuh

template<typename Tuning, bool HasValues, bool MergeSort, typename 
KeysIt1, 
    typename KeysIt2, typename KeysIt3, typename ValsIt1, typename 
ValsIt2,
    typename ValsIt3, typename Comp>
MGPU_LAUNCH_BOUNDS void KernelMerge(KeysIt1 aKeys_global, ValsIt1 
aVals_global,
    int aCount, KeysIt2 bKeys_global, ValsIt2 bVals_global, int bCount,
    const int* mp_global, int coop, KeysIt3 keys_global, ValsIt3 
vals_global,
    Comp comp) {
 
    typedef MGPU_LAUNCH_PARAMS Params;
    typedef typename std::iterator_traits<KeysIt1>::value_type KeyType;
    typedef typename std::iterator_traits<ValsIt1>::value_type ValType;
 
    const int NT = Params::NT;
    const int VT = Params::VT;
    const int NV = NT * VT;
    union Shared {
        KeyType keys[NT * (VT + 1)];
        int indices[NV];
    };
    __shared__ Shared shared;
 
    int tid = threadIdx.x;
    int block = blockIdx.x;
 
    int4 range = ComputeMergeRange(aCount, bCount, block, coop, NT * 
VT, 
        mp_global);
 
    DeviceMerge<NT, VT, HasValues>(aKeys_global, aVals_global, 
bKeys_global,
        bVals_global, tid, block, range, shared.keys, shared.indices, 
        keys_global, vals_global, comp);
}

KernelMerge is called by both the Merge and Mergesort host functions. Mergesort is considerably more 
involved, but this complexity has been factored out into ComputeMergeRange, allowing the heavy lifting 
for both functions to be defined by DeviceMerge. Note that the keys and indices are unioned in shared 
memory, so as to not waste resources. This is an idiom used throughput Modern GPU, and an important one 
to follow if your goal is high throughput.

MergePathPartitions

include/kernels/search.cuh

template<int NT, MgpuBounds Bounds, typename It1, typename It2, 
typename Comp>
__global__ void KernelMergePartition(It1 a_global, int aCount, It2 

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/search.cuh

b_global, 
    int bCount, int nv, int coop, int* mp_global, int numSearches, Comp 
comp) {
 
    int partition = NT * blockIdx.x + threadIdx.x;
    if(partition < numSearches) {
        int a0 = 0, b0 = 0;
        int gid = nv * partition;
        if(coop) {
            int3 frame = FindMergesortFrame(coop, partition, nv);
            a0 = frame.x;
            b0 = min(aCount, frame.y);
            bCount = min(aCount, frame.y + frame.z) - b0;
            aCount = min(aCount, frame.x + frame.z) - a0;
 
            // Put the cross-diagonal into the coordinate system of 
the input
            // lists.
            gid -= a0;
        }
        int mp = MergePath<Bounds>(a_global + a0, aCount, b_global + 
b0, bCount,
            min(gid, aCount + bCount), comp);
        mp_global[partition] = mp;
    }
}
 
template<MgpuBounds Bounds, typename It1, typename It2, typename Comp>
MGPU_MEM(int) MergePathPartitions(It1 a_global, int aCount, It2 
b_global,
    int bCount, int nv, int coop, Comp comp, CudaContext& context) {
 
    const int NT = 64;
    int numPartitions = MGPU_DIV_UP(aCount + bCount, nv);
    int numPartitionBlocks = MGPU_DIV_UP(numPartitions + 1, NT);
    MGPU_MEM(int) partitionsDevice = context.Malloc<int>(numPartitions 
+ 1);
 
    KernelMergePartition<T, Bounds>
        <<<numPartitionBlocks, NT, 0, context.Stream()>>>(a_global, 
aCount,
        b_global, bCount, nv, coop, partitionsDevice->get(), 
numPartitions + 1, 
        comp);
    return partitionsDevice;
}

KernelMergePartition performs coarse-granularity partitioning for both MGPU Merge and 
Mergesort. It fills out the mp_global Merge Path/cross-diagonal intersections that are consumed by 
ComputeMergeRange. This is a simple and efficient division of labor—coarse-grained scheduling is 
achieved by first calling MergePathPartitions to fill out mp_global, which is subsequently 
provided to ComputeMergeRange in an algorithm's kernel.



The MergePathPartitions function is central to the MGPU library. The argument nv is the granularity 
of the partition and typically is set to the number of values per CTA, NV, a product of LaunchBoxVT 
parameters VT and NT, and a template argument for most of MGPU's kernels. coop is the number of CTAs 
cooperating to merge pairs of sorted lists in the mergesort routine—this is non-zero for all other functions 
(like merge, vectorized sorted search, etc.).

The list of callers of MergePathPartitions is extensive:

1. Bulk Insert  . Bulk Remove uses a standard binary search for global partitioning.

2. Merge  .

3. Mergesort  .

4. Vectorized sorted search   calls both the lower- and upper-bound specializations of 
MergePathPartitions.

5. Load-balancing search   calls the upper-bound function and only specializes on integer types. 
Additionally, all load-balancing search clients use MergePathPartitions indirectly:

a. IntervalExpand  

b. IntervalMove  

c. Relational joins  

Other methods don't call MergePathPartitions, but still opt into this two-phase scheduling and 
sequential-work approach:

6. The high-performance segmented and locality sorts of the next section fuse coarse-grained 
partitioning with work queueing, exploiting the sortedness of inputs to reduce processing.

7. MGPU Multisets   introduce a new partitioning search called Balanced Path which incorporates 
duplicate ranking into key ordering. Four serial set functions, modeled after SerialMerge, perform 
C++-style set intersection, union, difference, and symmetric difference.

Launching from the host

Mergesort is a multi-pass, out-of-place algorithm. The blocksort reduces global memory traffic by sorting 
blocks of NV elements locally, performing key exchange through low-latency shared memory. Subsequent 
global merge passes recursively doubles the length of sorted lists, from NV to 2*NV to 4*NV, etc., until the 
input is fully sorted.

include/kernels/mergesort.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/mergesort.cuh
http://nvlabs.github.io/moderngpu/sets.html
http://nvlabs.github.io/moderngpu/segsort.html#localitysortbenchmark
http://nvlabs.github.io/moderngpu/segsort.html#segsortbenchmark
http://nvlabs.github.io/moderngpu/join.html
http://nvlabs.github.io/moderngpu/intervalmove.html#intervalmove
http://nvlabs.github.io/moderngpu/intervalmove.html#intervalexpand
http://nvlabs.github.io/moderngpu/loadbalance.html
http://nvlabs.github.io/moderngpu/sortedsearch.html
http://nvlabs.github.io/moderngpu/mergesort.html
http://nvlabs.github.io/moderngpu/merge.html
http://nvlabs.github.io/moderngpu/bulkinsert.html

template<typename T, typename Comp>
MGPU_HOST void MergesortKeys(T* data_global, int count, Comp comp,
    CudaContext& context) {
 
    const int NT = 256;
    const int VT = 7;
    typedef LaunchBoxVT<NT, VT> Tuning;
    int2 launch = Tuning::GetLaunchParams(context);
     
    const int NV = launch.x * launch.y;
    int numBlocks = MGPU_DIV_UP(count, NV);
    int numPasses = FindLog2(numBlocks, true);
 
    MGPU_MEM(T) destDevice = context.Malloc<T>(count);
    T* source = data_global;
    T* dest = destDevice->get();
 
    KernelBlocksort<Tuning, false>
        <<<numBlocks, launch.x, 0, context.Stream()>>>(source, (const 
int*)0,
        count, (1 & numPasses) ? dest : source, (int*)0, comp);
    if(1 & numPasses) std::swap(source, dest);
 
    for(int pass = 0; pass < numPasses; ++pass) {
        int coop = 2<< pass;
        MGPU_MEM(int) partitionsDevice = 
MergePathPartitions<MgpuBoundsLower>(
            source, count, source, 0, NV, coop, comp, context);
         
        KernelMerge<Tuning, false, true>
            <<<numBlocks, launch.x, 0, context.Stream()>>>(source, 
            (const int*)0, count, source, (const int*)0, 0, 
            partitionsDevice->get(), coop, dest, (int*)0, comp);
        std::swap(dest, source);
    }
}

We allocate a temporary buffer to ping-pong mergesort passes. The number of global passes is the ceil of 
log2 of the tile count. As the user expects results sorted in-place in data_global, we blocksort into 
data_global if numPasses is even and blocksort into the temporary if numPasses is odd. This way, 
sorted data always lands in data_global after the final merge pass without requiring an additional copy. 
The mergesort host function has the same macro structure as CTABlocksortLoop: it loops from coop = 
2 to coop = numBlocks (NT in blocksort). MergePathPartitions searches global memory to find 
the intersection of cross-diagonals and Merge Paths, as identified by the utility function 
FindMergesortFrame.

Idiomatic GPU codes often have this coarse-grained/fine-grained paired structure: coarse-grained partitioning 
and scheduling operates on the full input in global memory; fine-grained partitioning and scheduling operates 
on local tiles in shared memory. This simple two-level hierarchy has algorithmic benefits: the bulk of 
partitioning operations are run over small, constant-sized blocks, helping amortize the cost of global 
partitioning. We also see architecutral benefits: performing most data movement within CTAs rather than 
between them reduces latency and improves the throughput of kernels.



8. Segmented Sort and Locality Sort
Segmented sort and locality sort are high-performance variants of mergesort that operate on non-uniform 
random data. Segmented sort allows us to sort many variable-length arrays in parallel. A list of head indices 
provided to define segment intervals. Segmented sort is fast: not only is segmentation supported for 
negligible cost, the function takes advantage of early-exit opportunities to improve throughput over vanilla 
mergesort. Locality sort detects regions of approximate sortedness without requiring annotations.

Benchmark and usage

Segmented sort keys benchmark from tests/benchmarksegsort.cu

Segmented sort keys demonstration from tests/demo.cu

void DemoSegSortKeys(CudaContext& context) {
    printf("\n\nSEG-SORT KEYS DEMONSTRATION:\n\n");
 
    // Use CudaContext::GenRandom to generate 100 random integers 
between 0 and
    // 9.
    int N = 100;
    MGPU_MEM(int) keys = context.GenRandom<int>(N, 0, 99);
 
    // Define 10 segment heads (for 11 segments in all).
    const int NumSegs = 10;
    const int SegHeads[NumSegs] = { 4, 19, 22, 56, 61, 78, 81, 84, 94, 

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarksegsort.cu

97 };
    MGPU_MEM(int) segments = context.Malloc(SegHeads, 10);
 
    printf("Input keys:\n");
    PrintArray(*keys, "%4d", 10);
 
    printf("\nSegment heads:\n");
    PrintArray(*segments, "%4d", 10);
 
    // Sort within segments.
    SegSortKeysFromIndices(keys->get(), N, segments->get(), NumSegs, 
context);
 
    printf("\nSorted data (segment heads are marked by *):\n");
    PrintArrayOp(*keys, FormatOpMarkArray(" %c%2d", SegHeads, 
NumSegs), 10);
}

SEG-SORT KEYS DEMONSTRATION:

Input keys:
    0:    42   39    9   77   59   97   47   74   69   63
   10:    69    7   63   63    3   52    6   29   31   32
   20:    53   63   65   99   40   51   81   72   71   24
   30:    96   33   53   74   32   68   10   68   61    7
   40:    77   45   42   69    9    6   26    6   15   52
   50:    28   26   44   48   52   13   45    9   87   12
   60:    51   96   94   75   63   26   95   72   24   41
   70:    67   47   28    5   67   61   69   49    6   90
   80:    25   93   22   91   66   30   84   79   34   22
   90:    78   44   67   51    0   23   60   71   38   98

Segment heads:
    0:     4   19   22   56   61   78   81   84   94   97

Sorted data (segment heads are marked by *):
    0:     9   39   42   77  * 3    6    7   29   31   47
   10:    52   59   63   63   63   69   69   74   97  *32
   20:    53   63  * 6    6    7    9   10   13   15   24
   30:    26   26   28   32   33   40   42   44   45   48
   40:    51   52   52   53   61   65   68   68   69   71
   50:    72   74   77   81   96   99  * 9   12   45   51
   60:    87  * 5   24   26   28   41   47   49   61   63
   70:    67   67   69   72   75   94   95   96  * 6   25
   80:    90  *22   91   93  *22   30   34   44   51   66
   90:    67   78   79   84  * 0   23   60  *38   71   98



Segmented sort pairs benchmark from tests/benchmarksegsort.cu

Segmented sort pairs demonstration from tests/demo.cu

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarksegsort.cu

void DemoSegSortPairs(CudaContext& context) {
    printf("\n\nSEG-SORT PAIRS DEMONSTRATION:\n\n");
 
    // Use CudaContext::GenRandom to generate 100 random integers 
between 0 and
    // 9.
    int N = 100;
    MGPU_MEM(int) keys = context.GenRandom<int>(N, 0, 99);
 
    // Fill values with ascending integers.
    MGPU_MEM(int) values = context.FillAscending<int>(N, 0, 1);
 
    // Define 10 segment heads (for 11 segments in all).
    const int NumSegs = 10;
    const int SegHeads[NumSegs] = { 4, 19, 22, 56, 61, 78, 81, 84, 94, 
97 };
    MGPU_MEM(int) segments = context.Malloc(SegHeads, 10);
 
    printf("Input keys:\n");
    PrintArray(*keys, "%4d", 10);
 
    printf("\nSegment heads:\n");
    PrintArray(*segments, "%4d", 10);
 
    // Sort within segments.
    SegSortPairsFromIndices(keys->get(), values->get(), N, segments-
>get(), 
        NumSegs, context);
 
    printf("\nSorted data (segment heads are marked by *):\n");
    PrintArrayOp(*keys, FormatOpMarkArray(" %c%2d", SegHeads, 
NumSegs), 10);
 
    printf("\nSorted indices (segment heads are marked by *):\n");
    PrintArrayOp(*values, FormatOpMarkArray(" %c%2d", SegHeads, 
NumSegs), 10);
}

SEG-SORT PAIRS DEMONSTRATION:

Input keys:
    0:    91   65    0   27   46   46   42    0   46   44
   10:    77   97   32   30   78   21   47   24    3   80
   20:    17   48   72   40   47   21   15   54   34   72
   30:    60   28   19   54   73   75   24   33   91   80
   40:    26   85   76    1   18   88   28   59    9    8
   50:    57   92   68   91   54   98   42   90   64   94
   60:    64   93   67    0   63   77   94    2   20   58
   70:    70   64   23   32   11   11   60   12   45   97
   80:    45   53   66   66   77   70   35    6   66   20
   90:    41   43   84    1   83    6   25   34   61   31

Segment heads:



    0:     4   19   22   56   61   78   81   84   94   97

Sorted data (segment heads are marked by *):
    0:     0   27   65   91  * 0    3   21   24   30   32
   10:    42   44   46   46   46   47   77   78   97  *17
   20:    48   80  * 1    8    9   15   18   19   21   24
   30:    26   28   28   33   34   40   47   54   54   54
   40:    57   59   60   68   72   72   73   75   76   80
   50:    85   88   91   91   92   98  *42   64   64   90
   60:    94  * 0    2   11   11   12   20   23   32   58
   70:    60   63   64   67   70   77   93   94  *45   45
   80:    97  *53   66   66  * 1    6   20   35   41   43
   90:    66   70   77   84  * 6   25   83  *31   34   61

Sorted indices (segment heads are marked by *):
    0:     2    3    1    0  * 7   18   15   17   13   12
   10:     6    9    4    5    8   16   10   14   11  *20
   20:    21   19  *43   49   48   26   44   32   25   36
   30:    40   31   46   37   28   23   24   27   33   54
   40:    50   47   30   52   22   29   34   35   42   39
   50:    41   45   38   53   51   55  *56   58   60   57
   60:    59  *63   67   74   75   77   68   72   73   69
   70:    76   64   71   62   70   65   61   66  *78   80
   80:    79  *81   82   83  *93   87   89   86   90   91
   90:    88   85   84   92  *95   96   94  *99   97   98

Locality sort keys benchmark from tests/benchmarklocalitysort.cu

Locality sort keys demonstration from tests/demo.cu

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarklocalitysort.cu

void DemoLocalitySortKeys(CudaContext& context) {
    printf("\n\nLOCALITY SORT KEYS DEMONSTRATION:\n\n");
 
    // Generate keys that are roughly sorted but with added noise.
    int N = 100;
    std::vector<int> keysHost(N);
    for(int i = 0; i < N; ++i) 
        keysHost[i] = i + Rand(0, 25);
 
    MGPU_MEM(int) keys = context.Malloc(keysHost);
 
    printf("Input keys:\n"); 
    PrintArray(*keys, "%4d", 10);
 
    // Sort by exploiting locality.
    LocalitySortKeys(keys->get(), N, context);
 
    printf("\nSorted data:\n");
    PrintArray(*keys, "%4d", 10);;
}

LOCALITY SORT KEYS DEMONSTRATION:

Input keys:
    0:    15   26   16    9   26   27   12   16   16   28
   10:    13   32   36   18   30   40   28   35   34   44
   20:    34   40   38   28   38   34   44   32   41   50
   30:    55   55   37   52   36   57   38   48   39   47
   40:    50   62   53   57   53   48   65   52   64   61
   50:    70   61   76   72   79   64   60   77   61   84
   60:    78   83   64   84   77   74   79   68   90   94
   70:    82   92   82   95   91   76   95   77   91   94
   80:    89  100   85   99   99  102   92  111   89   95
   90:   109  114   98   96  105  103  113  119  107  105

Sorted data:
    0:     9   12   13   15   16   16   16   18   26   26
   10:    27   28   28   28   30   32   32   34   34   34
   20:    35   36   36   37   38   38   38   39   40   40
   30:    41   44   44   47   48   48   50   50   52   52
   40:    53   53   55   55   57   57   60   61   61   61
   50:    62   64   64   64   65   68   70   72   74   76
   60:    76   77   77   77   78   79   79   82   82   83
   70:    84   84   85   89   89   90   91   91   92   92
   80:    94   94   95   95   95   96   98   99   99  100
   90:   102  103  105  105  107  109  111  113  114  119



Locality sort pairs benchmark from tests/benchmarklocalitysort.cu

Locality sort pairs demonstration from tests/demo.cu

void DemoLocalitySortPairs(CudaContext& context) {
    printf("\n\nLOCALITY SORT PAIRS DEMONSTRATION:\n\n");
 
    // Generate keys that are roughly sorted but with added noise.
    int N = 100;
    std::vector<int> keysHost(N);
    for(int i = 0; i < N; ++i) 
        keysHost[i] = i + Rand(0, 25);
 
    MGPU_MEM(int) keys = context.Malloc(keysHost);
    MGPU_MEM(int) values = context.FillAscending<int>(N, 0, 1);
 
    printf("Input keys:\n");
    PrintArray(*keys, "%4d", 10);
 
    // Sort by exploiting locality.
    LocalitySortPairs(keys->get(), values->get(), N, context);
 
    printf("\nSorted data:\n");
    PrintArray(*keys, "%4d", 10);
 
    printf("\nSorted indices:\n");
    PrintArray(*values, "%4d", 10);
}

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarklocalitysort.cu


LOCALITY SORT PAIRS DEMONSTRATION:

Input keys:
    0:    19   22   12   17   21   29   24   20   19   26
   10:    10   14   20   38   25   31   23   21   23   20
   20:    41   33   33   43   47   37   36   49   47   45
   30:    40   54   53   33   53   53   45   52   43   41
   40:    60   66   66   48   52   53   63   64   59   73
   50:    71   56   71   77   58   77   78   68   83   71
   60:    73   75   84   84   79   68   70   83   73   94
   70:    80   87   91   84   95   75   96   79   86   92
   80:    93  101   84  102   86   89   89   93  105  100
   90:   102  102   96  110  106   99   99  101   99  101

Sorted data:
    0:    10   12   14   17   19   19   20   20   20   21
   10:    21   22   23   23   24   25   26   29   31   33
   20:    33   33   36   37   38   40   41   41   43   43
   30:    45   45   47   47   48   49   52   52   53   53
   40:    53   53   54   56   58   59   60   63   64   66
   50:    66   68   68   70   71   71   71   73   73   73
   60:    75   75   77   77   78   79   79   80   83   83
   70:    84   84   84   84   86   86   87   89   89   91
   80:    92   93   93   94   95   96   96   99   99   99
   90:   100  101  101  101  102  102  102  105  106  110

Sorted indices:
    0:    10    2   11    3    0    8    7   12   19    4
   10:    17    1   16   18    6   14    9    5   15   21
   20:    22   33   26   25   13   30   20   39   23   38
   30:    29   36   24   28   43   27   37   44   32   34
   40:    35   45   31   51   54   48   40   46   47   41
   50:    42   57   65   66   50   52   59   49   60   68
   60:    61   75   53   55   56   64   77   70   58   67
   70:    62   63   73   82   78   84   71   85   86   72
   80:    79   80   87   69   74   76   92   95   96   98
   90:    89   81   97   99   83   90   91   88   94   93

Host functions

include/mgpuhost.cuh

//////////////////////////////////////////////////////////////////////
//////////
// kernels/segmentedsort.cuh
 
// Mergesort count items in-place in data_global. Keys are compared 
with Comp
// (as they are in MergesortKeys), however keys remain inside the 
segments 
// defined by flags_global. 
 
// flags_global is a bitfield cast to uint*. Each bit in flags_global 
is a 
// segment head flag. Only keys between segment head flags (inclusive 

https://github.com/NVlabs/moderngpu/blob/master/include/mgpuhost.cuh

on the
// left and exclusive on the right) may be exchanged. The first 
element is 
// assumed to start a segment, regardless of the value of bit 0.
 
// Passing verbose=true causes the function to print mergepass 
statistics to the
// console. This may be helpful for developers to understand the 
performance 
// characteristics of the function and how effectively it early-exits 
merge
// operations.
template<typename T, typename Comp>
MGPU_HOST void SegSortKeysFromFlags(T* data_global, int count,
    const uint* flags_global, CudaContext& context, Comp comp,
    bool verbose = false);
 
// SegSortKeysFromFlags specialized with Comp = mgpu::less<T>.
template<typename T>
MGPU_HOST void SegSortKeysFromFlags(T* data_global, int count,
    const uint* flags_global, CudaContext& context, bool verbose = 
false);
 
// Segmented sort using head flags and supporting value exchange.
template<bool Stable, typename KeyType, typename ValType, typename Comp>
MGPU_HOST void SegSortPairsFromFlags(KeyType* keys_global, 
    ValType* values_global, const uint* flags_global, int count,
    CudaContext& context, Comp comp, bool verbose = false);
 
// SegSortPairsFromFlags specialized with Comp = mgpu::less<T>.
template<bool Stable, typename KeyType, typename ValType>
MGPU_HOST void SegSortPairsFromFlags(KeyType* keys_global, 
    ValType* values_global, const uint* flags_global, int count,
    CudaContext& context, bool verbose = false)
 
// Segmented sort using segment indices rather than head flags. 
indices_global
// is a sorted and unique list of indicesCount segment start 
locations. These
// indices correspond to the set bits in the flags_global field. A 
segment
// head index for position 0 may be omitted.
template<typename T, typename Comp>
MGPU_HOST void SegSortKeysFromIndices(T* data_global, int count,
    const int* indices_global, int indicesCount, CudaContext& context,
    Comp comp, bool verbose = false);
 
// SegSortKeysFromIndices specialized with Comp = mgpu::less<T>.
template<typename T>
MGPU_HOST void SegSortKeysFromIndices(T* data_global, int count,
    const int* indices_global, int indicesCount, CudaContext& context,
    bool verbose = false);
 



Algorithm

Segmented sort is the function that I expect to have the most immediate impact on people's applications. It 
addresses one facet of a serious problem facing GPU computing: we can often solve a single big problem but 
find it difficult to process many smaller problems in parallel.

What does the ad-hoc approach for sorting multiple variable-length arrays in parallel look like? We know 
how to sort keys in a thread using sorting networks. We know how to mergesort within a warp or block. 
MGPU Merge can take a coop parameter to merge small lists in parallel, as long as those lists are power-of-
two multiples of the block size.

As a hypothetical, perhaps we could just sort within the segment intervals. Intervals shorter than VT would 
be processed by an intra-thread sorting network kernel. Intervals shorter than 32 * VT would be sorted by a 
warp-sorting kernel. Intervals shorter than NV would be sorted by our blocksort. We could use mergesort or 
radix sort on longer intervals, shifting segment IDs into the most-significant bits of the key, to maintain 
segment stability. After all these launches we'd re-order the data back into their original segment order. 

But this would be a nightmare to write and maintain.

If all we want is simplicity, we could call mergesort on individual segments using CUDA Dynamic 
Parallelism on sm_35 devices. However this would cause severe load-balance issues. It would perform worse 
than a brute-force radix sort with segment identifiers fused to the keys. Performance would degrade with the 
segment size—small segments would be provisioned entire CTAs, wasting huge amounts of compute.

Not knowing how to finesse problems like this is why CUDA has a reputation of being hard to program. It is 
infuriating that the smaller a problem gets, the more desperate the solution becomes.

MGPU Segmented Sort makes a single observation, modifies our mergesort, and delivers an implementation 
that is elegant and fast. 

   0    1    2    3    4    5    6    7    8    9   10   11   12   13   14   15
  41   67   34    0   39   24   78   58   62   64    5   81   45   27   61   91
   ^                        ^                        ^              ^

Consider these 16 random numbers grouped into four irregular segments. Segment heads are marked with 
carets. We're going to launch four tiny 'blocksorts' that sort four inputs each, maintaining segment order:

   0    1    2    3    4    5    6    7    8    9   10   11   12   13   14   15
  41   67   34    0   39   24   78   58   62   64    5   81   45   27   61   91
   ^                        ^                        ^                   ^

Sort into blocks of length 4:
   0    1    2    3 |  4    5    6    7 |  8    9   10   11 | 12   13   14   15
   0   34   41   67 | 39   24   58   78 | 62   64    5   81 | 45   27   61   91
   ^                        ^                        ^              ^

We use induction to explain the segmented blocksort above in terms of the merge operation illustrated below. 
Imagine a blocksort of segmented random data as a sequence of iterative merge operations, beginning with 
lists of length 4.

   0    1    2    3 |  4    5    6    7 |  8    9   10   11 | 12   13   14   15
   0   34   41   67 | 39   24   58   78 | 62   64    5   81 | 45   27   61   91
   ^                        ^                        ^              ^

http://docs.nvidia.com/cuda/pdf/CUDA_Dynamic_Parallelism_Programming_Guide.pdf
http://docs.nvidia.com/cuda/pdf/CUDA_Dynamic_Parallelism_Programming_Guide.pdf


Merge list 0 (block 0) with list 1 (block 1)
Merge list 2 (block 2) with list 3 (block 3):
 ( 0    1    2    3 |  4    5    6    7)|( 8    9   10   11 | 12   13   14   15)
 ( 0   34   39   41 | 67   24   58   78)|(62   64    5   45 | 81   27   61   91)
   ^                        ^                        ^              ^

Merge list 0 (blocks 0 and 1) with list 1 (blocks 2 and 3):

 ( 0    1    2    3 |  4    5    6    7 |  8    9   10   11 | 12   13   14   15)
 ( 0   34   39   41 | 67   24   58   62 | 64   78    5   45 | 81   27   61   91)
   ^                        ^                        ^              ^

In the coop = 2 stage, where two blocks cooperatively merge pairs of lists, the green segment that spans 
the block 0/block 1 boundary and the green segment that spans the block 2/block 3 boundary is modified. In 
the coop = 4 stage, where four blocks cooperatively merge pairs of lists, the black segment that spans the 
block 1/block 2 boundary is modified. This fully sorts the inputs.

The observation that enables efficient segmented sorting is that only segments that span the active interface 
between the two input lists are modified. The active interface is where the list on the left ends and the list on 
the right begins. During the coop = 2 stage, the portion of the first black segment being merged into list 0 
cannot be mixed with data to its left because those terms belong to the green segment; it cannot be mixed 
with the portion of the same segment to the right, because that data is belongs to a different output list (the 
boundary between block 1 and block 2 isn't an active interface until the next merge stage).

During the coop = 2 stage, only the black segment in the middle - the one that spans the interface - is 
modified. The other three segments are simply copied directly from source to destination.

Important: When merging lists of segmented data, only elements in the segments that span the active 
interface between inputs lists is modified. All other segments are locked in place and their terms are copied 
directly from input to output.

Segmented blocksort

include/device/ctasegsort.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctasegsort.cuh

template<int NT, int VT, bool Stable, bool HasValues, typename KeyType,
    typename ValType, typename Comp>
MGPU_DEVICE int2 CTASegsort(KeyType threadKeys[VT], ValType 
threadValues[VT], 
    int tid, int headFlags, KeyType* keys_shared, ValType* 
values_shared,
    int* ranges_shared, Comp comp) {
 
    if(Stable)
        // Odd-even transpose sort.
        OddEvenTransposeSortFlags<VT>(threadKeys, threadValues, 
headFlags,
            comp);
    else
        // Batcher's odd-even mergesort.
        OddEvenMergesortFlags<VT>(threadKeys, threadValues, headFlags, 
comp); 
 
    // Record the first and last occurrence of head flags in this 
segment.
    int blockEnd = 31 - clz(headFlags);
    if(-1 != blockEnd) blockEnd += VT * tid;
 
    int blockStart = ffs(headFlags);
    blockStart = blockStart ? (VT * tid - 1 + blockStart) : (NT * VT);
 
    ranges_shared[tid] = (int)bfi(blockEnd, blockStart, 16, 16);
 
    // Store back to shared mem. The values are in VT-length sorted 
lists.
    // These are merged recursively.
    DeviceThreadToShared<VT>(threadKeys, tid, keys_shared);
 
    int2 activeRange = CTASegsortLoop<NT, VT, HasValues>(threadKeys,
        threadValues, keys_shared, values_shared, ranges_shared, tid, 
        make_int2(blockStart, blockEnd), comp);
    return activeRange;
}

CTASegsort is a reusable segmented blocksort and a generalization of CTAMergesort. The caller 
provides segment head flags packed into the bitfield headFlags. As shown in the section on sorting 
networks, the odd-even transposition network includes logic to support segmentation. It only swaps elements 
i and i+1 if the i+1 flag in the bitfield is cleared, meaning both elements are in the same segment. After the 
sorting network, the keys are in sorted lists with length VT. 

For both segmented blocksort and the global merge passes, we build a binary tree of segment active ranges 
(the left- and right-most segment heads in a list). Only elements falling outside the active range can ever be 
modified by the sort. After the sorting network is run we calculate the active ranges (blockStart and 
blockEnd). We can now discard the head flags: the only necessary segment information for this sort is in 
the binary tree of ranges.

Threads use the CUDA instructions ffs (find first set) and clz (count leading zeros) to find the left-most 

http://nvlabs.github.io/moderngpu/mergesort.html#sortnetworks
http://nvlabs.github.io/moderngpu/mergesort.html#sortnetworks
http://nvlabs.github.io/moderngpu/mergesort.html#blocksort


and right-most segment heads in the list. These are referenced in the coordinate system of the CTA: if thread 
tid has a left-most segment head at 3, blockStart is assigned VT * tid + 3. If the thread doesn't contain 
a segment head, the interval (NV, -1) is used. The ranges are packed into a 32-bit integer with bfi (bitfield 
insert) and stored in the auxiliary array ranges_shared.

include/device/ctasegsort.cuh

template<int NT, int VT, bool HasValues, typename KeyType, typename 
ValType,
    typename Comp>
MGPU_DEVICE int2 CTASegsortLoop(KeyType threadKeys[VT], 
    ValType threadValues[VT], KeyType* keys_shared, ValType* 
values_shared, 
    int* ranges_shared, int tid, int2 activeRange, Comp comp) {
 
    const int NumPasses = sLogPow2<NT>::value;
    #pragma unroll
    for(int pass = 0; pass < NumPasses; ++pass) {
        int indices[VT];
        CTASegsortPass<NT, VT>(keys_shared, ranges_shared, tid, pass,
            threadKeys, indices, activeRange, comp);
 
        if(HasValues) {
            // Exchange values through shared memory.
            DeviceThreadToShared<VT>(threadValues, tid, 
values_shared);
            DeviceGather<NT, VT>(NT * VT, values_shared, indices, tid, 
                threadValues);
        }
 
        // Store results in shared memory in sorted order.
        DeviceThreadToShared<VT>(threadKeys, tid, keys_shared);
    }
    return activeRange;
}

CTASegsortLoop is a copy of CTABlocksortLoop which forwards to CTASegsortPass instead of 
CTABlocksortPass. The segmented sort very closely follows the vanilla mergesort, both for ease-of-
maintenance reasons and to help illustrate how enacting one small improvement can bring huge functionality 
and performance benefits to an old and simple algorithm.

include/device/ctasegsort.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctasegsort.cuh
https://github.com/NVlabs/moderngpu/blob/master/include/device/ctasegsort.cuh

template<int NT, int VT, typename T, typename Comp>
MGPU_DEVICE void CTASegsortPass(T* keys_shared, int* ranges_shared, 
int tid,
    int pass, T results[VT], int indices[VT], int2& activeRange, Comp 
comp) {
 
    // Locate the intervals of the input lists.
    int3 frame = FindMergesortFrame(2<< pass, tid, VT);
    int a0 = frame.x;
    int b0 = frame.y;
    int listLen = frame.z;
    int list = tid>> pass;
    int listParity = 1 & list;
    int diag = VT * tid - frame.x;
 
    // Fetch the active range for the list this thread's list is 
merging with.
    int siblingRange = ranges_shared[1 ^ list];
    int siblingStart = 0x0000ffff & siblingRange;
    int siblingEnd = siblingRange>> 16;
     
    // Create a new active range for the merge.
    int leftEnd = listParity ? siblingEnd : activeRange.y;
    int rightStart = listParity ? activeRange.x : siblingStart;
    activeRange.x = min(activeRange.x, siblingStart);
    activeRange.y = max(activeRange.y, siblingEnd);
 
    int p = SegmentedMergePath(keys_shared, a0, listLen, b0, listLen, 
leftEnd, 
        rightStart, diag, comp);
 
    int a0tid = a0 + p;
    int b0tid = b0 + diag - p;
    SegmentedSerialMerge<VT>(keys_shared, a0tid, b0, b0tid, b0 + 
listLen, 
        results, indices, leftEnd, rightStart, comp);
 
    // Store the ranges to shared memory.
    if(0 == diag)
        ranges_shared[list>> 1] = 
            (int)bfi(activeRange.y, activeRange.x, 16, 16);
}

As in CTABlocksortPass, each thread in CTASegsortPass calculates the range of source values (the 
input lists) and its cross-diagonal within that list. As in the vanilla mergesort, the input list has coordinates 
(a0, a1), (b0, b1), and the output list has coordinates (a0, b1).

In addition, the segmented sort computes its list index, parity, and sibling. The index is the list that the thread 
is mapped into for that particular pass. For pass 0 (coop = 2), the index of thread tid is simply tid. The 
left- and right-most segment head positions are passed into activeRange. 

    tid   (a0, a1)  (b0, b1):  diag   list  parity  sibling



pass 0 (coop = 2):
     0:   ( 0,  7)  ( 7, 14):     0      0     (0)        1
     1:   ( 0,  7)  ( 7, 14):     7      1     (1)        0
     2:   (14, 21)  (21, 28):     0      2     (0)        3
     3:   (14, 21)  (21, 28):     7      3     (1)        2
     4:   (28, 35)  (35, 42):     0      4     (0)        5
     5:   (28, 35)  (35, 42):     7      5     (1)        4
     6:   (42, 49)  (49, 56):     0      6     (0)        7
     7:   (42, 49)  (49, 56):     7      7     (1)        6

pass 1 (coop = 4):
     0:   ( 0, 14)  (14, 28):     0      0     (0)        1
     1:   ( 0, 14)  (14, 28):     7      0     (0)        1
     2:   ( 0, 14)  (14, 28):    14      1     (1)        0
     3:   ( 0, 14)  (14, 28):    21      1     (1)        0
     4:   (28, 42)  (42, 56):     0      2     (0)        3
     5:   (28, 42)  (42, 56):     7      2     (0)        3
     6:   (28, 42)  (42, 56):    14      3     (1)        2
     7:   (28, 42)  (42, 56):    21      3     (1)        2

pass 2 (coop = 8):
     0:   ( 0, 28)  (28, 56):     0      0     (0)        1
     1:   ( 0, 28)  (28, 56):     7      0     (0)        1
     2:   ( 0, 28)  (28, 56):    14      0     (0)        1
     3:   ( 0, 28)  (28, 56):    21      0     (0)        1
     4:   ( 0, 28)  (28, 56):    28      1     (1)        0
     5:   ( 0, 28)  (28, 56):    35      1     (1)        0
     6:   ( 0, 28)  (28, 56):    42      1     (1)        0
     7:   ( 0, 28)  (28, 56):    49      1     (1)        0

The list parity indicates what side of the merge a thread maps to: 0 for left and 1 for right. The other list in 
the merge is called sibling, and it has the opposite parity. Each thread loads the packed range of its sibling 
list from shared memory: this is the list it is merging with. The left and right ranges are merged together to 
find the left- and right-most segment heads of the resulting merged list. After the segmented merge, this is 
stored in shared memory and recursively percolated up, building the binary tree of active ranges. 

In CTASegsortPass, leftEnd is the location of the right-most segment in the left input list (drawn 
across the top of the diagram) and rightStart the location of the left-most segment in the right list 
(drawn along the right side). These define the ends of the active segment. For the area outside of (leftEnd, 



rightStart), the Merge Path follows the x- or y-axis, as it simply copies elements from one of the inputs. The 
cross-diagonals are constrained to the active segment region—queries outside this region return immediately.

include/device/ctasearch.cuh

117
118
119
120
121
122
123
124
125
126
127
128
129
130
131
132
133
134
135
136
137
138
139
140
141
142
143

template<typename InputIt, typename Comp>
MGPU_HOST_DEVICE int SegmentedMergePath(InputIt keys, int aOffset, int 
aCount,
    int bOffset, int bCount, int leftEnd, int rightStart, int diag, Comp 
comp) {
 
    // leftEnd and rightStart are defined from the origin, and diag is 
defined
    // from aOffset.
    // We only need to run a Merge Path search if the diagonal 
intersects the
    // segment that strides the left and right halves (i.e. is between 
leftEnd
    // and rightStart).
    if(aOffset + diag <= leftEnd) return diag;
    if(aOffset + diag >= rightStart) return aCount;
 
    bCount = min(bCount, rightStart - bOffset);
    int begin = max(max(leftEnd - aOffset, 0), diag - bCount);
    int end = min(diag, aCount);
 
    while(begin < end) {
        int mid = (begin + end)>> 1;
        int ai = aOffset + mid;
        int bi = bOffset + diag - 1 - mid;
 
        bool pred = !comp(keys[bi], keys[ai]);
        if(pred) begin = mid + 1;
        else end = mid;
    }
    return begin;
}

SegmentedMergePath is a straight-forward modification of the MergePath binary search that clamps 
the cross-diagonals to the area defined by the active range.

include/device/ctasegsort.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctasearch.cuh


59
60
61
62
63
64
65
66
67
68
69
70
71
72
73
74
75
76
77
78
79
80
81
82
83
84
85
86
87
88
89

template<int VT, typename T, typename Comp>
MGPU_DEVICE void SegmentedSerialMerge(const T* keys_shared, int aBegin,
    int aEnd, int bBegin, int bEnd, T results[VT], int indices[VT],
    int leftEnd, int rightStart, Comp comp, bool sync = true) {
         
    bEnd = min(rightStart, bEnd);
    T aKey = keys_shared[aBegin];
    T bKey = keys_shared[bBegin];
 
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        bool p;
 
        // If A has run out of inputs, emit B.
        if(aBegin >= aEnd)
            p = false;
        else if(bBegin >= bEnd || aBegin < leftEnd)
            // B has hit the end of the middle segment.
            // Emit A if A has inputs remaining in the middle segment.
            p = true;
        else
            // Emit the smaller element in the middle segment.
            p = !comp(bKey, aKey);
         
        results[i] = p ? aKey : bKey;
        indices[i] = p ? aBegin : bBegin;
        if(p) aKey = keys_shared[++aBegin];
        else bKey = keys_shared[++bBegin];
    }
    if(sync) { __syncthreads(); }
}

Four conditionals are evaluated when computing the merge predicate:

1. If aBegin >= aEnd the A pointer is out-of-range so we emit the next element in B.

2. If bBegin >= bEnd the B pointer is out-of-range or has hit the end of the active segment in the B 
list, if one exists. If the former is true, we simply emit A. If the latter is true, A must have at least 
entered the active segment (since A and B enter the active segment together), and A needs to 
complete emitting elements of the active segment before B can proceed to copy members of its 
inactive segments. Because the end of A's part of the active segment is also the end of A's input (see 
the illustration), once A finishes emitting its portion of the active segment, the first predicate, 
aBegin >= aEnd, succeeds, so all remaining trips through the loop evaluate p to false, so B's 
inactive segments are streamed out.

3. If aBegin < leftEnd the A pointer hasn't hit the start of its active segment. To preserve segment 
stability, all of A's segments are emitted before merging elements of B.

4. Otherwise, both pointers are inside the active segment. The keys are compared and the appropriate 
pointer is advanced, as in an ordinary merge.



Early-exit

We recursively merge lists, doubling their size and halving their number with each iteration. Since the 
number of active segments per merge remains fixed at one (that is, the segment on the active interface), the 
proportion of segments that are active vanishes as the sort progresses.

Tiles that map over only non-active segments (that is, segments to the left of the right-most segment head for 
tiles in the left half of the merge; and segments to right of the left-most segment head for tiles in the right half 
of the merge) do not require any merging. These stationary tiles check a copy flag: if the flag is set, the 
destination is up-to-date, and there is a no-op; if the flag is not set, the kernel copies from the source to the 
destination and sets the up-to-date flag. Tiles that map over an active segment clear the up-to-date flag and 
enqueue a merge operation.

Mean segment length = 300
                Merge tiles            Copy tiles 
pass  0:      7098 ( 99.93%)           5 (  0.07%)
pass  1:      4443 ( 62.55%)        2658 ( 37.42%)
pass  2:      2347 ( 33.04%)        3409 ( 47.99%)
pass  3:      1168 ( 16.44%)        2275 ( 32.03%)
pass  4:       572 (  8.05%)        1168 ( 16.44%)
pass  5:       302 (  4.25%)         572 (  8.05%)
pass  6:       163 (  2.29%)         302 (  4.25%)
pass  7:        78 (  1.10%)         163 (  2.29%)
pass  8:        37 (  0.52%)          78 (  1.10%)
pass  9:        18 (  0.25%)          37 (  0.52%)
pass 10:         7 (  0.10%)          18 (  0.25%)
pass 11:         6 (  0.08%)           7 (  0.10%)
pass 12:         2 (  0.03%)           6 (  0.08%)
average:      1249 ( 17.59%)         822 ( 11.59%)
total  :     16241 (228.65%)       10698 (150.61%)

Mean segment length = 10000
                Merge tiles            Copy tiles 
pass  0:      7102 ( 99.99%)           1 (  0.01%)
pass  1:      6927 ( 97.52%)         176 (  2.48%)
pass  2:      6639 ( 93.47%)         372 (  5.24%)
pass  3:      6022 ( 84.78%)         831 ( 11.70%)
pass  4:      5019 ( 70.66%)        1508 ( 21.23%)
pass  5:      3758 ( 52.91%)        2084 ( 29.34%)
pass  6:      2217 ( 31.21%)        2625 ( 36.96%)
pass  7:      1329 ( 18.71%)        1892 ( 26.64%)
pass  8:       619 (  8.71%)        1329 ( 18.71%)
pass  9:       232 (  3.27%)         619 (  8.71%)
pass 10:       110 (  1.55%)         232 (  3.27%)
pass 11:       148 (  2.08%)         110 (  1.55%)
pass 12:        88 (  1.24%)         148 (  2.08%)
average:      3093 ( 43.55%)         917 ( 12.92%)
total  :     40210 (566.10%)       11927 (167.91%)

SegmentedSort launch with verbose = true work reporting

The coarse-grained partitioning kernel used by MGPU Merge and Mergesort is augmented to also enqueue 
merge and copy work-items. The number of merge and copy tasks for a segmented sort grows with the mean 
segment length. In the above figure, the number of enqueued merge and copy operations is printed for each 
pass, for 10,000,000 inputs with average segment lengths of 300 and 10000 elements.

Once the merged list length exceeds the mean segment length, the number of merge operations required each 
pass begins to decrease by powers of two. The case with segment length of 300 exhibits this geometric 



decrease after the first global merge pass; the entire sort only requires the equivalent of only 2.28 global 
passes over the data. For the case with segment length of 10,000, this geometric decrease in workload takes 
longer to manifest, and 5.66 passes are required to sort the data.

Although this same early-exit tactic could be used within the segmented blocksort, we wouldn't expect an 
equivalent increase in performance there. Threads that discover that they're mapped over only non-active 
segments could choose to no-op, but due to SIMD execution rules, a diverged warp runs only as fast as its 
slowest lane. It may be feasible to dynamically reconfigure the segmented blocksort to benefit from early-
exit, but that is beyond the scope of this effort.

During the global merge passes, tiles that map over fully-sorted segments (or any other 'stable' data) can 
early-exit rather than merging. Only CTAs that map over "active segments" (segments straddling the merge 
interface of two sorted input lists) need to be merged. Since the number of active segments decrees 
geometrically (the number of lists is cut in half each iteration as the length of the lists double), a 
geometrically increasing percentage of CTAs may early-exit. We implement a special work-queueing system 
to process active tiles on a fixed-size launch of persistent CTAs—this eliminates the start-up cost for CTAs 
that simply exit out.

Mergesort is an out-of-place operation. After blocksorting we load from one buffer and store to another, 
swap, and repeat until the data is fully sorted. A temporary buffer is allocated with enough capacity to hold a 
flag per tile. If the domain of the tile has the same elements in both of the double buffers, the flag is set. 
Early-exit with tile granularity is enabled by this status flag: if the flag is set and the tile source interval maps 
to the destination interval, the tile operation can be elided.

The work-queueing system supports two different operations on tiles:

1. Merge - If the tile's source interval (as returned by ComputeMergeRange) does not map directly 
into the destination interval, we merge the two source lists and clear the tile's status flag. 

2. Copy - If the tile's source interval maps directly into the destination interval and the status flag is 
cleared (indicating the source and destination tile ranges do not contain the same elements), the 
source interval is copied to the destination and the status flag is set.

Each global merge pass launches two kernels: the first performs global partitioning and schedules merge and 
copy tasks by pushing to two work queues; the second cooperatively executes all queued merge and copy 
tasks.

Filling the work queue

include/kernels/segmentedsort.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/segmentedsort.cuh

template<int NT, bool Segments, typename KeyType, typename Comp>
__global__ void KernelSegSortPartitionBase(const KeyType* keys_global,
    SegSortSupport support, int count, int nv, int numPartitions, Comp 
comp) {
 
    union Shared {
        int partitions[NT];
        typename CTAScan<NT, ScanOpAdd>::Storage scan;
    };
    __shared__ Shared shared;
 
    int tid = threadIdx.x;
    int partition = tid + (NT - 1) * blockIdx.x;
 
    // Compute one extra partition per CTA. If the CTA size is 128 
threads, we
    // compute 128 partitions and 127 blocks. The next CTA then 
starts at
    // partition 127 rather than 128. This relieves us from having to 
launch
    // a second kernel to build the work queues.
    int p0;
    int3 frame;
    if(partition < numPartitions) {
        frame = FindMergesortFrame(2, partition, nv);
        int listLen = frame.z;
        int a0 = frame.x;
        int b0 = min(frame.y, count);
        int diag = nv * partition - a0;
        int aCount = min(listLen, count - a0);
        int bCount = min(listLen, count - b0);
 
        if(Segments) {
            // Segmented merge path calculation. Use the ranges as 
constraints.
            int leftRange = support.ranges_global[~1 & partition];
            int rightRange = support.ranges_global[
                min(numPartitions - 2, 1 | partition)];
         
            // Unpack the left and right ranges. Transform them into 
the global
            // coordinate system by adding a0 or b0.
            int leftStart = 0x0000ffff & leftRange;
            int leftEnd = leftRange>> 16;
            if(nv == leftStart) leftStart = count;
            else leftStart += a0;
            if(-1 != leftEnd) leftEnd += a0;
 
            int rightStart = 0x0000ffff & rightRange;
            int rightEnd = rightRange>> 16;
            if(nv == rightStart) rightStart = count;
            else rightStart += b0;
            if(-1 != rightEnd) rightEnd += b0;



After launching the segmented blocksort, the host launches KernelSegSortPartitionBase for the 
first global merge pass or KernelSegSortPartitionDerived for all subsequent passes. Each thread 
computes one partition, but since a tile needs both a starting and ending partition, a CTA of NT threads can 
enqueue only NT - 1 tiles.

For the base-pass kernel above, the 16-bit active ranges for each blocksort tile are unpacked and transformed 
by addition from tile-local coordinates to global coordinates. Neighboring active ranges (covering the two 
sorted lists that serve as sources for the sort's first merge pass) are then stored as int2 types in 
support.ranges2_global, into which a binary tree of active ranges is constructed; this corresponds 
exactly to the active-range tree constructed by CTASegsortPass in the segmented blocksort.

SegmentedMergePath searches global data for the intersection of the Merge Path and the thread's cross-
diagonal. If the cross-diagonal does not pass through the active segment (like any of the cross-diagonals that 
intersect the A- and B-axes as dotted lines in the segmented Merge Path figure above), the search returns 
immediately. As we progress in the mergesort, the percentage of cross-diagonals intersecting the active list 
vanishes, and we effectively benefit from early-exit of partitioning in addition to merging.

include/kernels/segmentedsort.cuh

template<int NT>
MGPU_DEVICE void DeviceSegSortCreateJob(SegSortSupport support,
    int count, bool active, int3 frame, int tid, int pass, int nv, int 
block,
    int p0, int p1, int* shared) {
 
    typedef CTAScan<NT, ScanOpAdd> S; 
    typename S::Storage* scan = (typename S::Storage*)shared;
         
    // Compute the gid'th work time.
    bool mergeOp = false;
    bool copyOp = false;
    int gid = nv * block;
    int4 mergeRange;
    if(active) {
        int4 range = FindMergesortInterval(frame, 2<< pass, block, nv, 
count, 
            p0, p1);
        int a0 = range.x;
        int a1 = range.y;
        int b0 = range.z;
        int b1 = range.w;
        if(a0 == a1) {
            a0 = b0;
            a1 = b1;
            b0 = b1;
        }
 
        mergeRange = make_int4(a0, a1, b0, block);
        mergeOp = (b1 != b0) || (a0 != gid);
        copyOp = !mergeOp && (!pass || !
support.copyStatus_global[block]);
    }

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/segmentedsort.cuh

    int mergeTotal, copyTotal;
    int mergeScan = S::Scan(tid, mergeOp, *scan, &mergeTotal);
    int copyScan = S::Scan(tid, copyOp, *scan, &copyTotal);
    if(!tid) {
        shared[0] = atomicAdd(&support.queueCounters_global->x, 
mergeTotal);
        shared[1] = atomicAdd(&support.queueCounters_global->y, 
copyTotal);
    }
    __syncthreads();
 
    if(mergeOp) {
        support.copyStatus_global[block] = 0;
        support.mergeList_global[shared[0] + mergeScan] = mergeRange;
    } 
    if(copyOp) {
        support.copyStatus_global[block] = 1;
        support.copyList_global[shared[1] + copyScan] = block;
    }
}

DeviceSegSortCreateJob is called at the end of both partition kernels. This is where we put the early-
exit logic. The function is passed the begin and end Merge Path indices. FindMergesortInterval 
takes the Merge Path indices and computes the intervals that serve as the A and B input lists. If either A or B 
interval is empty, then we're loading from just one input. Since both inputs are sorted (remember we are 
merging pairs of sorted lists), then the tile is already sorted. In this case we elide the merge operation.

If we're on the first pass or copyStatus_global[block] is false, then the destination buffer doesn't 
have the same data as the source buffer over the tile's output interval. We enqueue a copy operation by 
stepping on atomicAdd and storing the tile index to copy to copyList_global. If neither source 
interval is empty we must enqueue a merge by pushing the work interval to mergeList_global and clear 
the status bit.

Important: The early-exit heuristic examines only the mapping of the source lists into each tile. Early-exit 
does not require segmentation. Segmentation is a statement of sortedness in the input, but if keys were 
passed in the same relative order without providing segment indices, the early-exit behavior would be 
identical.

Locality sort is a specialization of segmented sort that does not support (or bear the costs of) segmentation. 
It uses the same partitioning and queueing kernels as segmented sort, but takes a branch that uses the 
standard Merge Path search. Locality sort also uses the standard mergesort blocksort. Use locality sort when 
inputs start near to their output locations. This is a qualititative characterization of a dataset. Programmers are 
encouraged to try both vanilla mergesort and locality sort and use the variant with the higher throughput.

Servicing the work queue

include/kernels/segmentedsort.cuh

template<typename Tuning, bool Segments, bool HasValues, typename 
KeyType,
    typename ValueType, typename Comp>
MGPU_LAUNCH_BOUNDS void KernelSegSortMerge(const KeyType* keys_global, 
    const ValueType* values_global, SegSortSupport support, int count, 
    int pass, KeyType* keysDest_global, ValueType* valsDest_global, 
Comp comp) {
 
    typedef MGPU_LAUNCH_PARAMS Params;
    const int NT = Params::NT;
    const int VT = Params::VT;
    const int NV = NT * VT;
    union Shared {
        KeyType keys[NT * (VT + 1)];
        int indices[NV];
        int4 range;
    };
    __shared__ Shared shared;
 
    int tid = threadIdx.x;
     
    // Check for merge work.
    while(true) {
        if(!tid) {
            int4 range = make_int4(-1, 0, 0, 0);
            int next = atomicAdd(&support.queueCounters_global->x, -1) 
- 1;
            if(next >= 0) range = support.mergeList_global[next];
            shared.range = range;
        }
        __syncthreads();
 
        int4 range = shared.range;
        __syncthreads();
 
        if(range.x < 0) break;
 
        int block = range.w;
        int gid = NV * block;
        int count2 = min(NV, count - gid);
        range.w = count2 - (range.y - range.x) + range.z;
 
        if(Segments)
            // Segmented merge
            DeviceSegSortMerge<NT, VT, HasValues>(keys_global, 
values_global,
                support, tid, block, range, pass, shared.keys, 
shared.indices, 
                keysDest_global, valsDest_global, comp);
        else
            // Unsegmented merge (from device/ctamerge.cuh)
            DeviceMerge<NT, VT, HasValues>(keys_global, values_global, 
                keys_global, values_global, tid, block, range, 



Typically we size a grid launch to the data size. Because early-exit dynamically sizes the tasks, we'd have to 
copy task counts from device to host memory each iteration to change the launch size. This would 
synchronize the device and hurt performance. Instead we launch a fixed number of "persistent CTAs" (a 
small multiple of the number of SMs, queried with CudaContext::NumSMs()) and atomically peel tasks 
off the queues until both queues are empty.

All threads loop until the merge queue is empty, then loop until the copy queue is empty. Thread 0 
decrements the queue counter and stores a code in shared memory. After synchronization, all threads read 
this code out and act accordingly.

Locality sort is supported by settings Segments = false: this directs the kernel to use DeviceMerge 
rather than DeviceSegSortMerge. 

http://nvlabs.github.io/moderngpu/merge.html#algorithm


9. Vectorized Sorted Search
Run many concurrent searches where both the needles and haystack arrays are sorted. This input condition 
lets us recast the function as a sequential process resembling merge, rather than as a traditional binary search. 
Complexity improves from A log B to A + B, and because we touch every input, a search can retrieve not 
just the lower-bound of A into B but simultaneously the upper-bound of B into A, plus flags for all elements 
indicating if matches in the other array exist.

Benchmark and usage

Vectorized sorted search (lower_bound A into B) benchmark from tests/benchmarksortedsearch.cu

Vectorized sorted search demostration from tests/demo.cu

void DemoSortedSearch(CudaContext& context) {
    printf("\n\nSORTED SEARCH DEMONSTRATION:\n\n");
 
    // Use CudaContext::SortRandom to generate a haystack of 200 
random integers
    // between 0 and 999 and an array of 100 needles in the same 
range.
    int HaystackSize = 200;
    int NeedlesSize = 100;
    MGPU_MEM(int) haystack = context.SortRandom<int>(HaystackSize, 0, 
299);
    MGPU_MEM(int) needles = context.SortRandom<int>(NeedlesSize, 0, 

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarksortedsearch.cu

299);
 
    printf("Haystack array:\n");
    PrintArray(*haystack, "%4d", 10);
    printf("\nNeedles array:\n");
    PrintArray(*needles, "%4d", 10);
 
    // Run a vectorized sorted search to find lower bounds.
    SortedSearch<MgpuBoundsLower>(needles->get(), NeedlesSize, 
haystack->get(),
        HaystackSize, needles->get(), context);
 
    printf("\nLower bound array:\n");
    PrintArray(*needles, "%4d", 10);
}

SORTED SEARCH DEMONSTRATION:

Haystack array:
    0:     0    5    5    7    7    7    7    8    9    9
   10:    10   11   12   14   15   15   16   17   19   19
   20:    20   24   25   28   28   29   31   33   36   36
   30:    37   38   40   42   42   43   45   46   49   50
   40:    51   51   51   52   53   55   56   57   60   60
   50:    61   61   62   62   64   66   68   69   73   74
   60:    79   81   82   84   85   88   90   90   95   97
   70:    99  101  105  108  108  111  115  118  118  119
   80:   119  119  119  122  122  123  125  126  126  130
   90:   133  133  135  135  139  140  143  145  145  146
  100:   147  149  149  149  154  158  160  161  165  166
  110:   168  169  170  172  172  174  174  174  175  175
  120:   175  177  179  182  183  184  186  187  188  190
  130:   192  193  194  196  198  199  199  205  205  208
  140:   209  215  217  218  218  218  220  220  221  221
  150:   223  224  225  230  234  234  235  240  240  243
  160:   244  249  250  251  252  253  253  254  255  255
  170:   255  257  258  258  259  262  263  265  267  270
  180:   270  274  278  278  278  279  280  281  284  284
  190:   284  285  285  292  294  295  296  296  296  298

Needles array:
    0:     3    3   12   16   16   17   17   19   20   21
   10:    24   27   27   28   30   31   35   39   40   42
   20:    52   52   53   53   54   55   57   58   62   63
   30:    72   75   83   86   86   89   92   95   98   98
   40:    99   99   99  100  104  105  107  109  110  111
   50:   112  117  118  121  124  126  129  132  133  139
   60:   140  148  156  160  161  167  168  173  179  186
   70:   191  198  202  202  212  212  214  220  223  229
   80:   233  239  245  254  256  256  260  268  269  269
   90:   271  271  272  273  277  285  296  296  299  299

Lower bound array:
    0:     1    1   12   16   16   17   17   18   20   21
   10:    21   23   23   23   26   26   28   32   32   33
   20:    43   43   44   44   45   45   47   48   52   54
   30:    58   60   63   65   65   66   68   68   70   70
   40:    70   70   70   71   72   72   73   75   75   75



   50:    76   77   77   83   86   87   89   90   90   94
   60:    95  101  105  106  107  110  110  115  122  126
   70:   130  134  137  137  141  141  141  146  150  153
   80:   154  157  161  167  171  171  175  179  179  179
   90:   181  181  181  181  182  191  196  196  200  200

Vectorized sorted search (complete search) benchmark from tests/benchmarksortedsearch.cu

Vectorized sorted search demostration (2) from tests/demo.cu

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarksortedsearch.cu

void DemoSortedSearch2(CudaContext& context) {
    printf("\n\nSORTED SEARCH DEMONSTRATION (2):\n\n");
 
    int ACount = 100;
    int BCount = 100;
    MGPU_MEM(int) aData = context.SortRandom<int>(ACount, 0, 299);
    MGPU_MEM(int) bData = context.SortRandom<int>(BCount, 0, 299);
    MGPU_MEM(int) aIndices = context.Malloc<int>(ACount);
    MGPU_MEM(int) bIndices = context.Malloc<int>(BCount);
 
    printf("A array:\n");
    PrintArray(*aData, "%4d", 10);
    printf("\nB array:\n");
    PrintArray(*bData, "%4d", 10);
 
    // Run a vectorized sorted search to find lower bounds.
    SortedSearch<MgpuBoundsLower, MgpuSearchTypeIndexMatch,
        MgpuSearchTypeIndexMatch>(aData->get(), ACount, bData->get(), 
BCount,
        aIndices->get(), bIndices->get(), context);
 
    printf("\nLower bound of A into B (* for match):\n");
    PrintArrayOp(*aIndices, FormatOpMaskBit("%c%3d"), 10);
    printf("\nUpper bound of B into A (* for match):\n");
    PrintArrayOp(*bIndices, FormatOpMaskBit("%c%3d"), 10);
}

SORTED SEARCH DEMONSTRATION (2):

A array:
    0:     0    3    5   13   14   15   16   18   18   21
   10:    24   26   26   30   31   32   38   38   38   40
   20:    60   72   72   74   81   83   86   88   88   89
   30:    89   99   99  101  101  102  114  115  118  118
   40:   119  128  136  139  145  148  149  150  151  151
   50:   157  160  164  165  167  177  181  181  182  182
   60:   189  190  191  192  196  197  199  200  207  212
   70:   213  213  216  218  220  222  223  228  231  233
   80:   233  234  234  234  239  239  240  247  249  264
   90:   265  267  271  271  275  277  282  284  293  298

B array:
    0:     1    2   15   23   24   25   25   25   25   27
   10:    27   29   30   31   33   33   35   39   45   49
   20:    58   59   61   61   62   63   64   67   67   68
   30:    70   71   82   85   87   87   88   91   98   98
   40:   109  110  110  116  116  118  121  121  126  129
   50:   129  134  145  155  159  165  174  174  179  181
   60:   183  186  192  192  196  196  201  202  204  205
   70:   205  208  209  212  216  218  220  222  224  227
   80:   231  233  233  234  235  236  250  251  251  253
   90:   260  263  272  275  276  285  289  291  291  293

Lower bound of A into B (* for match):
    0:     0    2    2    2    2 *  2    3    3    3    3



   10:  *  4    9    9 * 12 * 13   14   17   17   17   18
   20:    22   32   32   32   32   33   34 * 36 * 36   37
   30:    37   40   40   40   40   40   43   43 * 45 * 45
   40:    46   49   52   52 * 52   53   53   53   53   53
   50:    54   55   55 * 55   56   58 * 59 * 59   60   60
   60:    62   62   62 * 62 * 64   66   66   66   71 * 73
   70:    74   74 * 74 * 75 * 76 * 77   78   80 * 80 * 81
   80:  * 81 * 83 * 83 * 83   86   86   86   86   86   92
   90:    92   92   92   92 * 93   95   95   95 * 99  100

Upper bound of B into A (* for match):
    0:     1    1 *  6   10 * 11   11   11   11   11   13
   10:    13   13 * 14 * 15   16   16   16   19   20   20
   20:    20   20   21   21   21   21   21   21   21   21
   30:    21   21   25   26   27   27 * 29   31   31   31
   40:    36   36   36   38   38 * 40   41   41   41   42
   50:    42   42 * 45   50   51 * 54   55   55   56 * 58
   60:    60   60 * 64 * 64 * 65 * 65   68   68   68   68
   70:    68   69   69 * 70 * 73 * 74 * 75 * 76   77   77
   80:  * 79 * 81 * 81 * 84   84   84   89   89   89   89
   90:    89   89   94 * 95   95   98   98   98   98 * 99

Host functions

include/mgpuhost.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/mgpuhost.cuh

/////////////////////////////////////////////////////////////////////
///////////
// kernels/sortedsearch.cuh
 
// Vectorized sorted search. This is the most general form of the 
function.
// Executes two simultaneous linear searches on two sorted inputs.
 
// Bounds:
//      MgpuBoundsLower - 
//          lower-bound search of A into B.
//          upper-bound search of B into A.
//      MgpuBoundsUpper - 
//          upper-bound search of A into B.
//          lower-bound search of B into A.
// Type[A|B]:
//      MgpuSearchTypeNone - no output for this input.
//      MgpuSearchTypeIndex - return search indices as integers.
//      MgpuSearchTypeMatch - return 0 (no match) or 1 (match).
//          For TypeA, returns 1 if there is at least 1 matching 
element in B
//              for element in A.
//          For TypeB, returns 1 if there is at least 1 matching 
element in A
//              for element in B.
//      MgpuSearchTypeIndexMatch - return search indices as integers. 
Most
//          significant bit is match bit.
//  aMatchCount, bMatchCount - If Type is Match or IndexMatch, return 
the total 
//      number of elements in A or B with matches in B or A, if the 
pointer is
//      not null. This generates a synchronous cudaMemcpyDeviceToHost 
call that
//      callers using streams should be aware of.
template<MgpuBounds Bounds, MgpuSearchType TypeA, MgpuSearchType 
TypeB,
    typename InputIt1, typename InputIt2, typename OutputIt1, 
    typename OutputIt2, typename Comp>
MGPU_HOST void SortedSearch(InputIt1 a_global, int aCount, InputIt2 
b_global,
    int bCount, OutputIt1 aIndices_global, OutputIt2 bIndices_global,
    Comp comp, CudaContext& context, int* aMatchCount = 0, 
    int* bMatchCount = 0);
 
// SortedSearch specialized with Comp = mgpu::less<T>.
template<MgpuBounds Bounds, MgpuSearchType TypeA, MgpuSearchType 
TypeB,
    typename InputIt1, typename InputIt2, typename OutputIt1, 
    typename OutputIt2>
MGPU_HOST void SortedSearch(InputIt1 a_global, int aCount, InputIt2 
b_global,
    int bCount, OutputIt1 aIndices_global, OutputIt2 bIndices_global,



Algorithm

Searching data is a critical part of all computing systems. On the GPU, because of the extreme width of the 
processor, we need to be a bit creative to fully utilize the device while executing a search. The Thrust library 
includes vectorized binary searches in which all threads in the grid run their own independent binary search 
on sorted inputs. The user passes multiple "needles" (the keys you search for) and a sorted "haystack" (what 
you are looking in). Thousands of needles are required to fill the width of the machine. 

Even when batching a large array of queries, performance will drag if the needles are unsorted—random 
access to the haystack results in many cache misses. GPU cache lines are 128 bytes, and if querying 4-byte 
data types, sustained throughput will only hit 3% of peak bandwidth.

This project focuses on functions that take one or more sorted inputs and emit a sorted output. Consider 
taking this requirement to vectorized binary searching: rather than accept the divergent memory accesses 
caused by queries on random needles, we could sort needles to keep the L2 cache hot. Instead of using 3% of 
each fetched cache line, we'd expect high re-use of cache lines between neighboring threads. The same 
embarrassingly-parallel vectorized search function achieves much higher throughput with better data 
organization. Searching for sorted needles in a haystack is a well-motivated problem. Consider joining two 
database tables—these are likely both sorted coming off disk, will need to be in sorted order to perform a 
sort-merge join, and want sorted results to return as a rowset.

The work presented in previous MGPU pages inspired a specific algorithmic optimization to this problem of 
vectorized searching. If the needles are sorted, the lower- or upper-bound results must also be sorted. That is, 
if a binary search for key1 returns index1, a search for key2 >= key1 must return index2 >= index1. In the 

sequential implementation, if we're searching for key2 having already computed the result for key1, we can 

improve performance slightly by searching the interval (index1, end) rather than (begin, end), since index2 

cannot appear in (begin, index1).

There is a still stronger optimization: turn the binary search (O(A log B) complexity, where A is the needle 
array and B is the haystack) into a linear search (O(A + B) complexity). Starting with pointers A and B at the 
heads of the needle and haystack arrays, respectively:

• If B (haystack) is less than A (needle), advance B. Because the elements in A are sorted, we aren't 
advancing B past any possible results for yet-to-be-encountered elements in A.

• Otherwise, the B pointer is the result for the A query. Set results[A] = B and advance to the next 
query in A.

Sorted search for CPU from tests/benchmarksortedsearch.cu

https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarksortedsearch.cu
http://thrust.github.io/doc/group__vectorized__binary__search.html

// Return lower-bound of A into B.
template<MgpuBounds Bounds, typename T, typename Comp>
void CPUSortedSearch(const T* a, int aCount, const T* b, int bCount, 
    int* indices, Comp comp) {
 
    int aIndex = 0, bIndex = 0;
    while(aIndex < aCount) {
        bool p;
        if(bIndex >= bCount)
            p = true;
        else
            p = (MgpuBoundsUpper == Bounds)?
                comp(a[aIndex], b[bIndex]) :
                !comp(b[bIndex], a[aIndex]);
 
        if(p) indices[aIndex++] = bIndex;
        else ++bIndex;
    }
}

This code is obviously more efficient than processing each needle as a binary search. But what we've gained 
in work-efficiency we may have lost in parallelism: binary searches are embarassingly parallel while this 
linear search code processes queries sequentially.

It's instructive to compare CPUSortedSearch with SerialMerge, the GPU function that powers 
MGPU's merge and mergesort:

include/device/ctamerge.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctamerge.cuh

template<int VT, bool RangeCheck, typename T, typename Comp>
MGPU_DEVICE void SerialMerge(const T* keys_shared, int aBegin, int aEnd,
    int bBegin, int bEnd, T* results, int* indices, Comp comp) { 
 
    T aKey = keys_shared[aBegin];
    T bKey = keys_shared[bBegin];
 
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        bool p;
        if(RangeCheck) 
            p = (bBegin >= bEnd) || ((aBegin < aEnd) && !comp(bKey, 
aKey));
        else
            p = !comp(bKey, aKey);
 
        results[i] = p ? aKey : bKey;
        indices[i] = p ? aBegin : bBegin;
 
        if(p) aKey = keys_shared[++aBegin];
        else bKey = keys_shared[++bBegin];
    }
    __syncthreads();
}

Although they have some differences, both routines check out-of-range pointers and set the predicate 
appropriately. If both pointers are in-range, keys are compared with the comparator object. For sorted search, 
if A (the needle) is smaller, the output is set and A is incremented; if B (the haystack) is smaller, B is 
incremented. For serial merge, if A is smaller, output is stored and A is incremented; if B is smaller, output is 
stored and B is incremented. The only material difference between these routines is how results are returned: 
the traversals over the input arrays are identical.

We parallelize the vectorized sorted search just like we do the merge. Coarse- and fine-grained scheduling 
and partitioning code is reused from MGPU Merge. However because we stream over A and B data but only 
emit A elements, we need an additional in-CTA pass to compact results.

There is a surprising and welcomed benefit from using a linear search, beyond just the improved work-
efficiency: If we search for the lower-bound of A into B, we can also recover the upper-bound of B into A in 
the same pass. This is possible with linear search because we encounter every element from B; it does not 
hold with binary search.

Sorted search (2) for CPU from tests/benchmarksortedsearch.cu

https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarksortedsearch.cu

// Return lower-bound of A into B and upper-bound of B into A.
template<typename T, typename Comp>
void CPUSortedSearch2(const T* a, int aCount, const T* b, int bCount, 
    int* aIndices, int* bIndices, Comp comp) {
 
    int aIndex = 0, bIndex = 0;
    while(aIndex < aCount || bIndex < bCount) {
        bool p;
        if(bIndex >= bCount) p = true;
        else if(aIndex >= aCount) p = false;
        else p = !comp(b[bIndex], a[aIndex]);
 
        if(p) aIndices[aIndex++] = bIndex;
        else bIndices[bIndex++] = aIndex;
    }
}

The inputs are traversed in the same order as in CPUSortedSearch (and in merge), but there is now an 
output on every iteration. This function looks very much like merge. We even lose the sense of 'needles' and 
'haystack,' as the arrays are treated symmetrically. Scheduling and partioning the parallel version will be 
handled the same as merge. This is further evidence for the argument presented in the introduction and 
reiterated through these pages: we gain flexibility, clarity, and performance by separating 
partitioning/scheduling and work logic.

Sorted search (3) for CPU from tests/benchmarksortedsearch.cu

// Return lower-bound of A into B and set the high bit if A has a match 
in B.
// Return upper-bound of B into A and set the high bit if B has a match 
in A.
template<typename T, typename Comp>
void CPUSortedSearch3(const T* a, int aCount, const T* b, int bCount,
    int* aIndices, int* bIndices, Comp comp) {
 
    int aIndex = 0, bIndex = 0;
    while(aIndex < aCount || bIndex < bCount) {
        bool p;
        if(bIndex >= bCount) p = true;
        else if(aIndex >= aCount) p = false;
        else p = !comp(b[bIndex], a[aIndex]);
 
        if(p) {
            // Compare the current key in A with the current key in B.
            bool match = bIndex < bCount && !comp(a[aIndex], 
b[bIndex]);
            aIndices[aIndex++] = bIndex + ((int)match<< 31);
        } else {
            // Compare the current key in B with the previous key in 
A.
            bool match = aIndex && !comp(a[aIndex - 1], b[bIndex]);
            bIndices[bIndex++] = aIndex + ((int)match<< 31);
        }

https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarksortedsearch.cu


    }
}

The second version of the sorted search is augmented with additional compares. It sets the most significant 
bit of the output index if the corresponding key in A or B has a match in the complementary array. We're 
supporting comparators (C++-style), so to test equality we must verify that both !(A < B) and !(B < A). 
Fortunately the state of the two pointers implies one half of the expression for both match tests.

• Advancing A (lower-bound search): We know that B cannot be less than A (or else we would be 
advancing B!). We can test for equality simply by testing that A is not less than B: !
comp(a[aIndex], b[bIndex]). 

• Advancing B (upper-bound search): The match in A (if there is one) must be the previous element 
in A, a[aIndex - 1]. However because we got to this point, we know that B is not less than the 
previous element in A: if it were, then when examining the previous A we would have advanced B 
rather than A, and we wouldn't be at this state. The expression !comp(a[aIndex - 1], 
b[bIndex]) is all that is needed to test equality.

Vectorized sorted search is a detailed and information-heavy function. That it runs nearly as fast as merge 
encourages its use in many situations. It has a pleasing symmetry with merge, and in fact we can re-
implement merge using vectorized sorted search to produce indices and Bulk Insert to insert data at these 
indices.

Load-balancing search, an amazingly useful intra-CTA utility, is a straight-forward specialization of 
vectorized sorted search. We demonstate pairing vectorized sorted search with load-balancing search to 
implement relational joins in a later page.

Parallel sorted search

include/kernels/sortedsearch.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/sortedsearch.cuh
http://nvlabs.github.io/moderngpu/join.html
http://nvlabs.github.io/moderngpu/loadbalance.html
http://nvlabs.github.io/moderngpu/bulkinsert.html

template<int NT, int VT, MgpuBounds Bounds, bool IndexA, bool MatchA, 
    bool IndexB, bool MatchB, typename InputIt1, typename InputIt2, 
typename T,
    typename Comp>
MGPU_DEVICE int2 DeviceLoadSortedSearch(int4 range, InputIt1 a_global, 
    int aCount, InputIt2 b_global, int bCount, int tid, int block,
    T* keys_shared, int* indices_shared, Comp comp) {
 
    int a0 = range.x;
    int a1 = range.y;
    int b0 = range.z;
    int b1 = range.w;
    int aCount2 = a1 - a0;
    int bCount2 = b1 - b0;
 
    // For matching:
    // If UpperBound
    //      MatchA requires preceding B
    //      MatchB requires trailing A
    // If LowerBound
    //      MatchA requires trailing B
    //      MatchB requires preceding A
    int leftA = MatchB && (MgpuBoundsLower == Bounds) && (a0 > 0);
    int leftB = MatchA && (MgpuBoundsUpper == Bounds) && (b0 > 0);
    int rightA = a1 < aCount;
    int rightB = b1 < bCount;
 
    int aStart = leftA;
    int aEnd = aStart + aCount2 + rightA;
    int bStart = aEnd + leftB;
    int bEnd = bStart + bCount2 + rightB;
 
    // Cooperatively load all the data including halos.
    DeviceLoad2ToShared<NT, VT, VT + 1>(a_global + a0 - leftA, aEnd, 
        b_global + b0 - leftB, bEnd - aEnd, tid, keys_shared);
 
    // Run the serial searches and compact the indices into shared 
memory.
    bool extended = rightA && rightB && (!MatchA || leftB) &&
        (!MatchB || leftA);
    int2 matchCount = CTASortedSearch<NT, VT, Bounds, IndexA, MatchA, 
IndexB,
        MatchB>(keys_shared, aStart, aCount2, aEnd, a0, bStart, 
bCount2, bEnd,
        b0, extended, tid, indices_shared, comp);
 
    return matchCount;
}

One vectorized sorted search kernel supports many modes of operation. The host function that launches 
KernelSortedSearch separates the MgpuSearchType enums into individual flags for IndexA, 
MatchA, IndexB, and MatchB. The kernel calls DeviceLoadSortedSearch which loads the 



intervals from the A and B arrays, runs a MergePath search on each thread, and performs a serial search 
over VT inputs per thread.

Different modes have different requirements:

 Lower-bound Upper-bound

MatchA trailing B preceding B

MatchB preceding A trailing A

Merge Path partitioning divides A and B inputs into distinct, non-overlapping intervals. Loading just these 
intervals into a tile's shared memory is insufficient to support match operations. For a lower-bound search, 
equal elements are consumed from A before B. If an element A[i] has a match in B at B[j], B[j] will appear 
after A[i] in the Merge Path. The thread checking the match of A[i] needs access to B[j], even if B[j] is 
mapped to a subsequent tile.

The search types are decomposed and the interval pointers incremented or decremented to accommodate the 
extra terms requried to verify matches. DeviceLoad2ToShared cooperatively loads intervals from two 
source arrays and stores to shared memory. It incorporates an optimization to handle extended cases like this: 
for a nominal tile (i.e. all tiles except the partial tile at the end) each thread loads VT items—these VT loads 
are included in an unpredicated form, written to encourage maximum outstanding loads and reduce latency. 
At most the kernel loads only four additional items beyond this (one each for the preceding and trailing items 
from A and B), and only this final load is predicated. In other words: DeviceLoad2ToShared<NT, 
VT, VT + 1> generates an optimized path for full tiles, in which the first VT loads are unpredicated and 
the last load is predicated.

If a "halo" element has been loaded after the last A and B inputs, the extended flag is set and we omit 
range checks in the serial search code. CTASortedSearch computes search indices and matches into 
shared memory and returns match counts (of both A into B and B into A) to the caller. The calling function, 
KernelSortedSearch, copies the indices and match flags out of shared memory and into their 
respective output arrays.

CTASortedSearch

include/device/ctasortedsearch.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctasortedsearch.cuh

template<int NT, int VT, MgpuBounds Bounds, bool IndexA, bool MatchA, 
    bool IndexB, bool MatchB, typename T, typename Comp>
MGPU_DEVICE int2 CTASortedSearch(T* keys_shared, int aStart, int 
aCount,
    int aEnd, int a0, int bStart, int bCount, int bEnd, int b0, bool 
extended, 
    int tid, int* indices_shared, Comp comp) {
 
    // Run a merge path to find the start of the serial search for 
each thread.
    int diag = VT * tid;
    int mp = MergePath<Bounds>(keys_shared + aStart, aCount, 
        keys_shared + bStart, bCount, diag, comp);
    int a0tid = mp;
    int b0tid = diag - mp;
 
    // Serial search into register.
    int3 results;
    int indices[VT];
    if(extended)
        results = DeviceSerialSearch<VT, Bounds, false, IndexA, 
MatchA, IndexB,
            MatchB>(keys_shared, a0tid + aStart, aEnd, b0tid + bStart, 
bEnd, 
            a0 - aStart, b0 - bStart, indices, comp);
    else
        results = DeviceSerialSearch<VT, Bounds, true, IndexA, MatchA, 
IndexB, 
            MatchB>(keys_shared, a0tid + aStart, aEnd, b0tid + bStart, 
bEnd, 
            a0 - aStart, b0 - bStart, indices, comp);
    __syncthreads();
 
    // Compact the indices into shared memory. Use the decision bits 
(set is A,
    // cleared is B) to select the destination.
    int decisions = results.x;
    b0tid += aCount;
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        if((1<< i) & decisions) {
            if(IndexA || MatchA) indices_shared[a0tid++] = indices[i];
        } else {
            if(IndexB || MatchB) indices_shared[b0tid++] = indices[i];
        }
    }
    __syncthreads();
 
    // Return the match counts for A and B keys.
    return make_int2(results.y, results.z);
}



CTASortedSearch follows the same recipe that makes MGPU Merge (and Mergesort and Segmented 
Sort) so efficient, but adds a few twists:

1. The host function calls MergePathPartitions to globally partition the input arrays into tile-
sized chunks, as in merge.

2. In the sorted search kernel, data is cooperatively loaded from A and B arrays into shared memory.

3. Each thread runs a MergePath search for every VT * tid cross-diagonal.

4. DeviceSerialSearch is invoked with the offsets from 3. Each thread traverses VT elements in 
an unrolled loop and computes search results. Search indices/matches are returned in the order in 
which they are encountered. The set of decision bits are returned in order in results.x.

5. After synchronization, each thread steps through its VT indices and distributes them to either the A or 
B output arrays.

6. The number of A matches in B and B matches in A are returned to the caller, which may then use 
CTAReduce to find a total within the tile, and atomically increment the global match counters.

include/device/ctasortedsearch.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctasortedsearch.cuh
http://nvlabs.github.io/moderngpu/segsort.html
http://nvlabs.github.io/moderngpu/segsort.html
http://nvlabs.github.io/moderngpu/mergesort.html
http://nvlabs.github.io/moderngpu/merge.html

template<int VT, MgpuBounds Bounds, bool RangeCheck, bool IndexA, bool 
MatchA,
    bool IndexB, bool MatchB, typename T, typename Comp>
MGPU_DEVICE int3 DeviceSerialSearch(const T* keys_shared, int aBegin, 
    int aEnd, int bBegin, int bEnd, int aOffset, int bOffset, int* 
indices,
    Comp comp) {
 
    const int FlagA = IndexA ? 0x80000000 : 1;
    const int FlagB = IndexB ? 0x80000000 : 1;
 
    T aKey = keys_shared[aBegin];
    T bKey = keys_shared[bBegin];
    T aPrev, bPrev;
    if(aBegin > 0) aPrev = keys_shared[aBegin - 1];
    if(bBegin > 0) bPrev = keys_shared[bBegin - 1];
    int decisions = 0;
    int matchCountA = 0;
    int matchCountB = 0;
 
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        bool p;
        if(RangeCheck && aBegin >= aEnd) p = false;
        else if(RangeCheck && bBegin >= bEnd) p = true;
        else p = (MgpuBoundsUpper == Bounds) ?
            comp(aKey, bKey) : 
            !comp(bKey, aKey);
     
        if(p) {
            // aKey is smaller than bKey, so it is inserted before 
bKey. 
            // Save bKey's index (bBegin + first) as the result of the 
search
            // and advance to the next needle in A.
            bool match = false;
            if(MatchA) {
                // Test if there is an element in B that matches aKey.
                if(MgpuBoundsUpper == Bounds) {
                    bool inRange = !RangeCheck || (bBegin > aEnd);
                    match = inRange && !comp(bPrev, aKey);
                } else {
                    bool inRange = !RangeCheck || (bBegin < bEnd);
                    match = inRange && !comp(aKey, bKey);
                }
            }
             
            int index = 0;
            if(IndexA) index = bOffset + bBegin;
            if(match) index |= FlagA;
            if(IndexA || MatchA) indices[i] = index;
            matchCountA += match;
 



DeviceSerialSearch is a massive function that services the many permutations that vectorized sorted 
search supports. Search results are stored directly into the indices register array and are redistributed into 
shared memory by CTASortedSearch. Rather than having to provision space for both the source and 
destination in shared memory, halving occupancy, we provision only enough for the source data—outputs are 
produced into register; the CTA is synchronized; and the results are stored back to the same shared memory 
array.

When run on full tiles, DeviceSerialSearch is specialized with RangeCheck = false 
(corresponding to extended = true in CTASortedSearch), allowing it to elide range-checking logic 
that adds significant latency to execution. This is an optimization that can be made for merge, mergesort, and 
segmented mergesort as well. However, it has been prioritized here because A) there's potentially much more 
logic here to contend with; and B) DeviceLoadSortedSearch already is loading in halo elements to 
support match operations, so adding a specialization to elide range checks took minimal effort.

While reviewing the logic for the four match tests, keep in mind that equality is established with two less-
than checks: !(aKey < bKey) && !(bKey < aKey); or when written with comparators: !
comp(aKey, bKey) && !comp(bKey, aKey).

When computing the lower-bound of A into B (and the upper-bound of B into A);

• To match A (p = true): We're inserting aKey before bKey. If there is a match for aKey it must be 
bKey. Check that bKey is in range and equal to aKey. Half of the equality test has already been 
computed when we made the predicate test !comp(bKey, aKey). Check the other half of the 
equality condition with a second comparison: !comp(aKey, bKey).

• To match B (p = false): The predicate test has confirmed that bKey is smaller than aKey 
(comp(bKey, aKey) is true). If there is a match for bKey it must be aPrev. The previous A-
advancing iteration proved that !comp(bKey, aPrev). Check the other half of the equality 
condition with a second comparison: !comp(aPrev, bKey).

When computing the upper-bound of A into B (and the lower-bound of B into A) we flip the arguments 
around:

• To match A (p = true): We're inserting aKey after bKey. If there is a match for aKey it must be 
bPrev. Check that bPrev is in range and equal to aKey. The previous A-advancing iteration 
proved that !comp(aKey, bPrev); it failed the comp(aKey, bKey) test and advanced B to 
get us to this point. Check the other half of the equality condition with a second comparison: !
comp(bPrev, aKey).

• To match B (p = false): The predicate test has confirmed that aKey is not smaller than bKey. If 
there is a match for bKey it must be aKey. Check the other half of the equality condition with a 
second comparison: !comp(bKey, aKey).

Vectorized sorted search is a powerful merge-like function. It's put to good use when implementing relational 
joins. More importantly, it motivates the useful and elegant load-balancing search, the subject of the next 
page.

SortedEqualityCount

include/kernels/sortedsearch.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/sortedsearch.cuh
http://nvlabs.github.io/moderngpu/loadbalance.html
http://nvlabs.github.io/moderngpu/join.html
http://nvlabs.github.io/moderngpu/join.html

struct SortedEqualityOp {
    MGPU_HOST_DEVICE int operator()(int lb, int ub) const {
        return ub - lb;
    }
};
 
template<typename Tuning, typename InputIt1, typename InputIt2, 
    typename InputIt3, typename OutputIt, typename Comp, typename Op>
MGPU_LAUNCH_BOUNDS void KernelSortedEqualityCount(InputIt1 a_global, 
int aCount,
    InputIt2 b_global, int bCount, const int* mp_global, InputIt3 
lb_global,
    OutputIt counts_global, Comp comp, Op op) {
 
    typedef MGPU_LAUNCH_PARAMS Params;
    const int NT = Params::NT;
    const int VT = Params::VT;
    const int NV = NT * VT;
 
    union Shared {
        int keys[NT * (VT + 1)];
        int indices[NV];
    };
    __shared__ Shared shared;
     
    int tid = threadIdx.x;
    int block = blockIdx.x;
    int4 range = ComputeMergeRange(aCount, bCount, block, 0, NV, 
mp_global);
 
    // Compute the upper bound.
    int2 matchCount = DeviceLoadSortedSearch<NT, VT, MgpuBoundsUpper, 
true,
        false, false, false>(range, a_global, aCount, b_global, 
bCount, tid, 
        block, shared.keys, shared.indices, comp);
    int aCount2 = range.y - range.x;
 
    // Load the lower bounds computed by the previous launch.
    int lb[VT];
    DeviceGlobalToReg<NT, VT>(aCount2, lb_global + range.x, tid, lb);
 
    // Subtract the lower bound from the upper bound and store the 
count.
    counts_global += range.x;
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        int index = NT * i + tid;
        if(index < aCount2) {
            int count = op(lb[i], shared.indices[index]);
            counts_global[index] = count;
        }



    }
}

C++ standard library functions std::equal_range binary searches for a single key in an array and returns the 
pair of (lower-bound, upper-bound) iterators. std::count runs a similar search but returns the count of 
occurrences, which is equal to the difference of the upper- and lower-bounds.

Vectorized sorted search doesn't extend naturally to support equal-range queries in a single pass because 
partitioning and scheduling decisions are made specifically for either lower- or upper-bound duplicate 
semantics. We can, at least, provide a modest optimization for achieving a vectorized count function. Rather 
than running lower- and upper-bound searches independently and launching a third kernel to take differences, 
we've provided a SortedSearch specialization that finds upper-bound indices of A into B, then loads 
lower-bound indices that correspond to each output and computes and stores counts directly.

SortedEqualityCount is specialized over a user-provided difference operator. This adds flexibility to 
the function, allowing it to extract index bits of a match-decorated sorted search result, or to max the 
difference with a constant. (It's this usage that enables our relational left-join function.)

http://nvlabs.github.io/moderngpu/join.html
http://www.cplusplus.com/reference/algorithm/count/
http://www.cplusplus.com/reference/algorithm/equal_range/


10. Load-Balancing Search
Load-balancing search is a specialization of vectorized sorted search. It coordinates output items with the 
input objects that generated them. The CTA load-balancing search is a fundamental tool for partitioning 
irregular problems.

Benchmark and usage



Load-balancing search benchmark from tests/benchmarkloadbalance.cu

Load-balancing search demonstration from tests/demo.cu

void DemoLBS(CudaContext& context) {
    printf("\n\nLOAD-BALANCING SEARCH DEMONSTRATION:\n\n");
     
    // Use CudaContext::GenRandom to generate work counts between 0 
and 5.
    int N = 50;
    MGPU_MEM(int) counts = context.GenRandom<int>(N, 0, 5);
     
    printf("Object counts\n");
    PrintArray(*counts, "%4d", 10);
 
    // Scan the counts.
    int total = Scan(counts->get(), N, context);
    printf("\nScan of object counts:\n");
    PrintArray(*counts, "%4d", 10);
    printf("Total: %4d\n", total);
 
    // Allocate space for the object references and run load-
balancing search.
    MGPU_MEM(int) refsData = context.Malloc<int>(total);
    LoadBalanceSearch(total, counts->get(), N, refsData->get(), 
context);
 
    printf("\nObject references:\n");
    PrintArray(*refsData, "%4d", 10);

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarkloadbalance.cu


}

LOAD-BALANCING SEARCH DEMONSTRATION:

Object counts
    0:     0    3    5    2    1    3    1    5    4    5
   10:     2    5    4    0    2    3    1    4    0    5
   20:     4    3    2    4    2    4    3    3    0    3
   30:     1    4    4    4    4    2    0    3    0    5
   40:     0    0    0    0    2    2    3    0    4    4

Scan of object counts:
    0:     0    0    3    8   10   11   14   15   20   24
   10:    29   31   36   40   40   42   45   46   50   50
   20:    55   59   62   64   68   70   74   77   80   80
   30:    83   84   88   92   96  100  102  102  105  105
   40:   110  110  110  110  110  112  114  117  117  121
Total:  125

Object references:
    0:     1    1    1    2    2    2    2    2    3    3
   10:     4    5    5    5    6    7    7    7    7    7
   20:     8    8    8    8    9    9    9    9    9   10
   30:    10   11   11   11   11   11   12   12   12   12
   40:    14   14   15   15   15   16   17   17   17   17
   50:    19   19   19   19   19   20   20   20   20   21
   60:    21   21   22   22   23   23   23   23   24   24
   70:    25   25   25   25   26   26   26   27   27   27
   80:    29   29   29   30   31   31   31   31   32   32
   90:    32   32   33   33   33   33   34   34   34   34
  100:    35   35   37   37   37   39   39   39   39   39
  110:    44   44   45   45   46   46   46   48   48   48
  120:    48   49   49   49   49

Host function

include/mgpuhost.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/mgpuhost.cuh

//////////////////////////////////////////////////////////////////////
//////////
// kernels/loadbalance.cuh
 
// LoadBalanceSearch is a special vectorized sorted search. Consider 
bCount
// objects that generate a variable number of work items, with aCount 
work
// items in total. The caller computes an exclusive scan of the work-
item counts
// into b_global.
 
// indices_global has aCount outputs. indices_global[i] is the index 
of the 
// object that generated the i'th work item.
// Eg:
// work-item counts:  2,  5,  3,  0,  1.
// scan counts:       0,  2,  7, 10, 10.   aCount = 11.
// 
// LoadBalanceSearch computes the upper-bound of 
counting_iterator<int>(0) with
// the scan of the work-item counts and subtracts 1:
// LBS: 0, 0, 1, 1, 1, 1, 1, 2, 2, 2, 4.
 
// This is equivalent to expanding the index of each object by the 
object's
// work-item count.
 
MGPU_HOST void LoadBalanceSearch(int aCount, const int* b_global, int 
bCount,
    int* indices_global, CudaContext& context);

Algorithm

Consider an array of objects O[i] (i < N) that each generate a non-negative variable number of work-items 
counts[i]. The sum of counts is M:

Work-item counts:
    0:     1    2    4    0    4    4    3    3    2    4
   10:     0    0    1    2    1    1    0    2    2    1
   20:     1    4    2    3    2    2    1    1    3    0
   30:     2    1    1    3    4    2    2    4    0    4

Exc-scan of counts:
    0:     0    1    3    7    7   11   15   18   21   23
   10:    27   27   27   28   30   31   32   32   34   36
   20:    37   38   42   44   47   49   51   52   53   56
   30:    56   58   59   60   63   67   69   71   75   75

Inc-scan of counts:
    0:     1    3    7    7   11   15   18   21   23   27
   10:    27   27   28   30   31   32   32   34   36   37
   20:    38   42   44   47   49   51   52   53   56   56
   30:    58   59   60   63   67   69   71   75   75   79



Total work-items: 79

It is simple to calculate the range of work-items that each object creates. We exclusive scan the work-item 
counts: these are the 'begin' indices for each object's run of outputs. The 'end' indices, if desired, are the 
inclusive scan of the objects' counts, or the exclusive scan plus the counts.

Consider this mapping of object counts into work-items a forward transformation. The corresponding inverse 
transformation, which maps work-items into the objects that generated them, is not as straight-forward.

Lower-bound search of work-items into exc-scan of counts:
    0:     0    1    2    2    3    3    3    3    5    5
   10:     5    5    6    6    6    6    7    7    7    8
   20:     8    8    9    9   10   10   10   10   13   14
   30:    14   15   16   18   18   19   19   20   21   22
   40:    22   22   22   23   23   24   24   24   25   25
   50:    26   26   27   28   29   29   29   31   31   32
   60:    33   34   34   34   35   35   35   35   36   36
   70:    37   37   38   38   38   38   40   40   40
   
Lower-bound search of work-items into inc-scan of counts:
    0:     0    0    1    1    2    2    2    2    4    4
   10:     4    4    5    5    5    5    6    6    6    7
   20:     7    7    8    8    9    9    9    9   12   13
   30:    13   14   15   17   17   18   18   19   20   21
   40:    21   21   21   22   22   23   23   23   24   24
   50:    25   25   26   27   28   28   28   30   30   31
   60:    32   33   33   33   34   34   34   34   35   35
   70:    36   36   37   37   37   37   39   39   39

The 40 objects generated 79 work-items. Running a lower-bound search from each work-item index (i.e. 
keys from 0 to 78) on either the exclusive or inclusive scan of object counts doesn't quite work—the indices 
in red indicate mismatches. What does work is taking the upper-bound of work-item indices with the 
exclusive scan of the counts and subtracting one:

Work-item counts:
    0:     1    2    4    0    4    4    3    3    2    4
   10:     0    0    1    2    1    1    0    2    2    1
   20:     1    4    2    3    2    2    1    1    3    0
   30:     2    1    1    3    4    2    2    4    0    4

Exc-scan of counts:
    0:     0    1    3    7    7   11   15   18   21   23
   10:    27   27   27   28   30   31   32   32   34   36
   20:    37   38   42   44   47   49   51   52   53   56
   30:    56   58   59   60   63   67   69   71   75   75
   
Load-balancing search:
    0:     0    1    1    2    2    2    2    4    4    4
   10:     4    5    5    5    5    6    6    6    7    7
   20:     7    8    8    9    9    9    9   12   13   13
   30:    14   15   17   17   18   18   19   20   21   21
   40:    21   21   22   22   23   23   23   24   24   25
   50:    25   26   27   28   28   28   30   30   31   32
   60:    33   33   33   34   34   34   34   35   35   36
   70:    36   37   37   37   37   39   39   39   39

Work-item rank (i - excscan[LBS[i]]):
    0:     0    0    1    0    1    2    3    0    1    2
   10:     3    0    1    2    3    0    1    2    0    1



   20:     2    0    1    0    1    2    3    0    0    1
   30:     0    0    0    1    0    1    0    0    0    1
   40:     2    3    0    1    0    1    2    0    1    0
   50:     1    0    0    0    1    2    0    1    0    0
   60:     0    1    2    0    1    2    3    0    1    0
   70:     1    0    1    2    3    0    1    2    3

The load-balancing search providhes each work-item with the index of the object that generated it. The object 
index can then be used to find the work-item's rank within the generating object. For example, work-item 10 
in the figure above was generated by object 4 (see element 10 in the load-balancing search). The scan of 
counts at position 4 is 7. The difference between the work-item's index (10) and the object's scan (7) is the 
work-item's rank within the object: 10 - 7 = 3. 

CPULoadBalanceSearch from tests/benchmarkloadbalance.cu

void CPULoadBalanceSearch(int aCount, const int* b, int bCount, int* 
indices) {
    int ai = 0, bi = 0;
    while(ai < aCount || bi < bCount) {
        bool p;
        if(bi >= bCount) p = true;
        else if(ai >= aCount) p = false;
        else p = ai < b[bi]; // aKey < bKey is upper-bound condition.
         
        if(p) indices[ai++] = bi - 1;   // subtract 1 from the upper-
bound.
        else ++bi;
    }
}

The serial implementation for the load-balancing search is very simple. We only support integer types and 
the A array is just the sequence of natural numbers. When written this way it's clear that the load-balancing 
search is immediately parallelizable, and as both input arrays are monotonically non-decreasing, it is in fact a 
special case of the vectorized sorted search from the previous page.

Important: Load-balancing search is kind of scan inverse. It operates on scanned work-item counts and 
returns the index of the object that generated each work-item. It's more accurate to consider the load-
balancing search as an idiom or pattern rather than an algorithm. It's not a step-by-step procedure and it's not 
intended to directly solve problems. Rather, the load-balancing search is a concept that helps the 
programmer better understand scheduling in problems with irregular parallelism.

CTALoadBalance

CTALoadBalance is a very light-weight operator. It can be included at the top of kernels as boilerplate, 
transforming thread IDs (or global output IDs) into the coordinate space of generating objects. The next two 
algorithms covered, Interval Move and relational join, use this embedded form of load-balancing search.

You'll usually need to call MergePathPartitions in the host code immediately prior to launching a 
kernel that uses intra-CTA load-balancing search. This global search runs an upper-bound binary search to 
find the intersection of each CTA's cross-diagonal with the Merge Path curve defined by the set of all work-
item indices (a counting_iterator<int>) and the exclusive scan of work-item counts.

http://nvlabs.github.io/moderngpu/mergesort.html#mergepathpartitions
http://nvlabs.github.io/moderngpu/join.html
http://nvlabs.github.io/moderngpu/intervalmove.html
http://nvlabs.github.io/moderngpu/sortedsearch.html
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarkloadbalance.cu


include/device/ctaloadbalance.cuh

template<int VT, bool RangeCheck>
MGPU_DEVICE void DeviceSerialLoadBalanceSearch(const int* b_shared, int 
aBegin,
    int aEnd, int bFirst, int bBegin, int bEnd, int* a_shared) {
 
    int bKey = b_shared[bBegin];
 
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        bool p;
        if(RangeCheck) 
            p = (aBegin < aEnd) && ((bBegin >= bEnd) || (aBegin < 
bKey));
        else
            p = aBegin < bKey;
 
        if(p)
            // Advance A (the needle).
            a_shared[aBegin++] = bFirst + bBegin;
        else
            // Advance B (the haystack).
            bKey = b_shared[++bBegin];
    }
}

We'll start with the serial loop DeviceSerialLoadBalanceSearch, a GPU treatment of 
CPULoadBalanceSearch. The interval of scan elements available to the thread, b_shared, are passed 
to the function in shared memory. Elements of the A array are output (work-item) indices and are generated 
directly from the interval range. 

Because the A inputs take no space in shared memory, and because we emit one output per A input, we store 
search results directly to shared memory rather than to register array. This is a break from the other routines 
in this library, where we gather sources from shared memory and keep temporary outputs in register, 
synchronize, then store back to shared memory to conserve space. The sequential nature of the A inputs lets 
us store the upper-bound - 1 directly into shared memory, simplifying the routine.

Like vectorized sorted search, full tiles that load a halo element at the end of the CTA's B interval can elide 
range checking. The nominal form of DeviceSerialLoadBalanceSearch makes only a single 
comparison (aBegin < bKey) per iteration, giving us a very lightweight and low-latency function.

include/device/ctaloadbalance.cuh

template<int NT, int VT>
MGPU_DEVICE int4 CTALoadBalance(int destCount, const int* b_global, 
    int sourceCount, int block, int tid, const int* mp_global, 
    int* indices_shared, bool loadPrecedingB) {
             
    int4 range = ComputeMergeRange(destCount, sourceCount, block, 0, 
NT * VT, 
        mp_global);

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctaloadbalance.cuh
https://github.com/NVlabs/moderngpu/blob/master/include/device/ctaloadbalance.cuh

    int a0 = range.x;
    int a1 = range.y;
    int b0 = range.z;
    int b1 = range.w;
 
    if(loadPrecedingB) { 
        if(!b0) loadPrecedingB = false;
        else --b0;
    }
 
    bool extended = a1 < destCount && b1 < sourceCount;
    int aCount = a1 - a0;
    int bCount = b1 - b0;
 
    int* a_shared = indices_shared;
    int* b_shared = indices_shared + aCount;
 
    // Load the b values (scan of work item counts).
    DeviceMemToMemLoop<NT>(bCount + (int)extended, b_global + b0, tid, 
        b_shared);
 
    // Run a merge path to find the start of the serial merge for each 
thread.
    int diag = min(VT * tid, aCount + bCount - (int)loadPrecedingB);
    int mp = 
MergePath<MgpuBoundsUpper>(mgpu::counting_iterator<int>(a0),
        aCount, b_shared + (int)loadPrecedingB, bCount - 
(int)loadPrecedingB,
        diag, mgpu::less<int>());
 
    int a0tid = a0 + mp;
    int b0tid = diag - mp + (int)loadPrecedingB;
     
    // Subtract 1 from b0 because we want to return upper_bound - 1.
    if(extended)
        DeviceSerialLoadBalanceSearch<VT, false>(b_shared, a0tid, a1, 
b0 - 1, 
            b0tid, bCount, a_shared - a0);
    else
        DeviceSerialLoadBalanceSearch<VT, true>(b_shared, a0tid, a1, 
b0 - 1, 
            b0tid, bCount, a_shared - a0);
    __syncthreads();
 
    return make_int4(a0, a1, b0, b1);
}

CTALoadBalance is the standard CTA-entry point for load-balancing search and a function we've 
demonstrated back in the introduction. 

An upper-bound MergePath search divides input arrays A (the natural numbers) and B (the scan of counts) 

http://nvlabs.github.io/moderngpu/intro.html#expand


into distinct, non-overlapping ranges. ComputeMergeRange returns the tuple (a0, a1, b0, b1) of input 
intervals. Scan offsets are loaded into shared memory with DeviceMemToMemLoop, a device function that 
cooperatively loads intervals that are expected to be much smaller than NV elements.

MergePath<MgpuBoundsUpper> is called on counting_iterator<int>(0) to divide the input 
domains into equal-size partitions. DeviceSerialLoadBalance sequentially traverses the inputs and 
stores search indices to the start of shared memory, where the caller expects to see them returned.

A array:
    0:     0    1    2    3    4    5    6    7    8    9
   10:    10   11   12   13   14   15   16   17   18   19
   20:    20   21   22   23   24   25   26   27   28   29
   30:    30   31   32   33   34   35   36   37   38   39
   40:    40   41   42   43   44   45   46   47   48   49
   50:    50   51   52   53   54   55   56   57   58   59
   60:    60   61   62   63   64   65   66   67   68   69
   70:    70   71   72   73   74   75   76   77   78

B array (Exc-scan of counts):
    0:     0    1    3    7    7   11   15   18   21   23
   10:    27   27   27   28   30   31   32   32   34   36
   20:    37   38   42   44   47   49   51   52   53   56
   30:    56   58   59   60   63   67   69   71   75   75

Divide into 4 equal partitions:
Tile 0: A = ( 0, 21)  B = ( 0,  9)
Tile 1: A = (21, 38)  B = ( 9, 22)
Tile 2: A = (38, 58)  B = (22, 31)
Tile 3: A = (58, 79)  B = (31, 40)

Load-balancing search:
    0:     0    1    1    2    2    2    2    4    4    4
   10:     4    5    5    5    5    6    6    6    7    7
   20:     7    8    8    9    9    9    9   12   13   13
   30:    14   15   17   17   18   18   19   20   21   21
   40:    21   21   22   22   23   23   23   24   24   25
   50:    25   26   27   28   28   28   30   30   31   32
   60:    33   33   33   34   34   34   34   35   35   36
   70:    36   37   37   37   37   39   39   39   39

There is a minor complication regarding the ranges of data to load. Consider dividing the sample objects into 
four evenly-sized parts. Tile 0 loads, notionally, (0, 21) from A and (0, 9) from B. Tile 1 loads (21, 38) from 
A and (9, 22) from B; etc. If a CTA only wishes to compute the load-balancing search, adhering to this non-
overlapping coverage is adequete, as we know from dealing with vectorized sorted search.

If, on the other hand, the caller wishes to compute the rank of each work-item within its generating object in 
addition to that object's index, a modification is required. Take, for example, the tile that loads the elements 
in red. Its first work-item (item 21) is generated by object 8 (see index 21 in the load-balancing search). We 
try to compute the rank of item 21 by looking up element 8 of the scan of counts, but that element is mapped 
into a different tile! This is due to the upper-bound Merge Path consuming elements of B (the scan) before 
consuming equal elements of A (the work-item indices).

We rectify this problem by simply loading the preceding element of B, if available. This element consumes 
an extra shared memory slot but doesn't complicate the serial search: each thread still traverses exactly VT 
elements. We simply load the preceding element of B to make it available when computing work-item ranks.

Load-balancing search, when used from inside a kernel, maps a variable number of work-items to each tile. 



CTALoadBalance returns an int4 type with the ranges (a0, a1, b0, b1), where (a0, a1) is the non-
overlapping interval of outputs and (b0, b1) is the range of inputs (potentially overlapping by 1 if 
precedingB is true). This decomposition of work is unusual in GPU programming but is actually very 
helpful when it comes to negotiating storage inside the CTA. IntervalExpand and IntervalMove on the next 
page exploit this irregular division of output to enable some powerful new primitives.

http://nvlabs.github.io/moderngpu/intervalmove.html#intervalmove
http://nvlabs.github.io/moderngpu/intervalmove.html#intervalexpand


11. IntervalExpand and IntervalMove
Schedule multiple variable-length fill, gather, scatter, or move operations. Partitioning is handled by load-
balancing search. Small changes in problem logic enable different behaviors. These functions are coarse-
grained counterparts to Bulk Remove and Bulk Insert.

Benchmark and usage



IntervalExpand benchmark from tests/benchmarkintervalmove.cu

Interval expand demonstration from tests/demo.cu

void DemoIntervalExpand(CudaContext& context) {
    printf("\n\nINTERVAL-EXPAND DEMONSTRATION:\n\n");
 
    const int NumInputs = 20;
    const int Counts[NumInputs] = { 
        2, 5, 7, 16, 0, 1, 0, 0, 14, 10, 
        3, 14, 2, 1, 11, 2, 1, 0, 5, 6 
    };
    const int Inputs[NumInputs] = {
        1, 1, 2, 3, 5, 8, 13, 21, 34, 55,
        89, 144, 233, 377, 610, 987, 1597, 2584, 4181, 6765
    };
    printf("Expand counts:\n");
    PrintArray(Counts, NumInputs, "%4d", 10);
 
    printf("\nExpand values:\n");
    PrintArray(Inputs, NumInputs, "%4d", 10);
 
    MGPU_MEM(int) countsDevice = context.Malloc(Counts, NumInputs);
    int total = Scan(countsDevice->get(), NumInputs, context);
 
    MGPU_MEM(int) fillDevice = context.Malloc(Inputs, NumInputs);
 
    MGPU_MEM(int) dataDevice = context.Malloc<int>(total);
    IntervalExpand(total, countsDevice->get(), fillDevice->get(), 

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarkintervalmove.cu

NumInputs, 
        dataDevice->get(), context);
     
    printf("\nExpanded data:\n");
    PrintArray(*dataDevice, "%4d", 10);
}

INTERVAL-EXPAND DEMONSTRATION:

Expand counts:
    0:     2    5    7   16    0    1    0    0   14   10
   10:     3   14    2    1   11    2    1    0    5    6

Expand values:
    0:     1    1    2    3    5    8   13   21   34   55
   10:    89  144  233  377  610  987 1597 2584 4181 6765

Expanded data:
    0:     1    1    1    1    1    1    1    2    2    2
   10:     2    2    2    2    3    3    3    3    3    3
   20:     3    3    3    3    3    3    3    3    3    3
   30:     8   34   34   34   34   34   34   34   34   34
   40:    34   34   34   34   34   55   55   55   55   55
   50:    55   55   55   55   55   89   89   89  144  144
   60:   144  144  144  144  144  144  144  144  144  144
   70:   144  144  233  233  377  610  610  610  610  610
   80:   610  610  610  610  610  610  987  987 1597 4181
   90:  4181 4181 4181 4181 6765 6765 6765 6765 6765 6765



IntervalMove benchmark from tests/benchmarkintervalmove.cu

Interval move demonstration from tests/demo.cu

void DemoIntervalMove(CudaContext& context) {
    printf("\n\nINTERVAL-MOVE DEMONSTRATION:\n\n");
 
    const int NumInputs = 20;
    const int Counts[NumInputs] = {
        3, 9, 1, 9, 8, 5, 10, 2, 5, 2,
        8, 6, 5, 2, 4, 0, 8, 2, 5, 6
    };
    const int Gather[NumInputs] = {
        75, 86, 17, 2, 67, 24, 37, 11, 95, 35,
        52, 18, 47, 0, 13, 75, 78, 60, 62, 29
    };
    const int Scatter[NumInputs] = {
        10, 80, 99, 27, 41, 71, 15, 0, 36, 13,
        89, 49, 66, 97, 76, 76, 2, 25, 61, 55
    };
 
    printf("Interval counts:\n");
    PrintArray(Counts, NumInputs, "%4d", 10);
 
    printf("\nInterval gather:\n");
    PrintArray(Gather, NumInputs, "%4d", 10);
 
    printf("\nInterval scatter:\n");
    PrintArray(Scatter, NumInputs, "%4d", 10);

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarkintervalmove.cu

    MGPU_MEM(int) countsDevice = context.Malloc(Counts, NumInputs);
    MGPU_MEM(int) gatherDevice = context.Malloc(Gather, NumInputs);
    MGPU_MEM(int) scatterDevice = context.Malloc(Scatter, NumInputs);
    int total = Scan(countsDevice->get(), NumInputs, context);
 
    MGPU_MEM(int) dataDevice = context.Malloc<int>(total);
 
    IntervalMove(total, gatherDevice->get(), scatterDevice->get(), 
        countsDevice->get(), NumInputs, 
mgpu::counting_iterator<int>(0),
        dataDevice->get(), context);
 
    printf("\nMoved data:\n");
    PrintArray(*dataDevice, "%4d", 10);
}

INTERVAL-MOVE DEMONSTRATION:

Interval counts:
    0:     3    9    1    9    8    5   10    2    5    2
   10:     8    6    5    2    4    0    8    2    5    6

Interval gather:
    0:    75   86   17    2   67   24   37   11   95   35
   10:    52   18   47    0   13   75   78   60   62   29

Interval scatter:
    0:    10   80   99   27   41   71   15    0   36   13
   10:    89   49   66   97   76   76    2   25   61   55

Moved data:
    0:    11   12   78   79   80   81   82   83   84   85
   10:    75   76   77   35   36   37   38   39   40   41
   20:    42   43   44   45   46   60   61    2    3    4
   30:     5    6    7    8    9   10   95   96   97   98
   40:    99   67   68   69   70   71   72   73   74   18
   50:    19   20   21   22   23   29   30   31   32   33
   60:    34   62   63   64   65   66   47   48   49   50
   70:    51   24   25   26   27   28   13   14   15   16
   80:    86   87   88   89   90   91   92   93   94   52
   90:    53   54   55   56   57   58   59    0    1   17

Host functions

include/mgpuhost.cuh

/////////////////////////////////////////////////////////////////////
///////////
// kernels/intervalmove.cuh
 
// IntervalExpand duplicates intervalCount items in values_global.
// indices_global is an intervalCount-sized array filled with the 

https://github.com/NVlabs/moderngpu/blob/master/include/mgpuhost.cuh

scan of item
// expand counts. moveCount is the total number of outputs (sum of 
expand 
// counts).
 
// Eg:
//      values  =  0,  1,  2,  3,  4,  5,  6,  7,  8
//      counts  =  1,  2,  1,  0,  4,  2,  3,  0,  2
//      indices =  0,  1,  3,  4,  4,  8, 10, 13, 13 (moveCount = 
15).
// Expand values[i] by counts[i]:
// output  =  0, 1, 1, 2, 4, 4, 4, 4, 5, 5, 6, 6, 6, 8, 8 
template<typename IndicesIt, typename ValuesIt, typename OutputIt>
MGPU_HOST void IntervalExpand(int moveCount, IndicesIt indices_global, 
    ValuesIt values_global, int intervalCount, OutputIt output_global,
    CudaContext& context);
 
// IntervalMove is a load-balanced and vectorized device memcpy.
// It copies intervalCount variable-length intervals from user-
defined sources
// to user-defined destinations. If destination intervals overlap, 
results are
// undefined.
 
// Eg:
// Interval counts:
//    0:     3    9    1    9    8    5   10    2    5    2
//   10:     8    6    5    2    4    0    8    2    5    6
// Scan of interval counts (indices_global):
//    0:     0    3   12   13   22   30   35   45   47   52
//   10:    54   62   68   73   75   79   79   87   89   94  
(moveCount = 100).
// Interval gather (gather_global):
//    0:    75   86   17    2   67   24   37   11   95   35
//   10:    52   18   47    0   13   75   78   60   62   29
// Interval scatter (scatter_global):
//    0:    10   80   99   27   41   71   15    0   36   13
//   10:    89   49   66   97   76   76    2   25   61   55
 
// This vectorizes into 20 independent memcpy operations which are 
load-balanced
// across CTAs:
// move 0: (75, 78)->(10, 13)       move 10: (52, 60)->(10, 18)
// move 1: (86, 95)->(80, 89)       move 11: (18, 24)->(49, 55)
// move 2: (17, 18)->(99,100)       move 12: (47, 52)->(66, 71)
// move 3: ( 2, 11)->(27, 36)       move 13: ( 0,  2)->(97, 99)
// move 4: (67, 75)->(41, 49)       move 14: (13, 17)->(76, 80)
// move 5: (24, 29)->(71, 76)       move 15: (75, 75)->(76, 76)
// move 6: (37, 47)->(15, 25)       move 16: (78, 86)->( 2, 10)
// move 7: (11, 13)->( 0,  3)       move 17: (60, 62)->(25, 27)
// move 8: (95,100)->(36, 41)       move 18: (62, 67)->(61, 66)
// move 9: (35, 37)->(13, 15)       move 19: (29, 35)->(55, 61)
template<typename GatherIt, typename ScatterIt, typename IndicesIt, 



IntervalExpand

IntervalExpand was discussed in the introduction and we revisit it here, now with a solid understanding of 
the load-balancing search pattern at the implementation's heart.

include/kernels/intervalmove.cuh

template<typename Tuning, typename IndicesIt, typename ValuesIt,
    typename OutputIt>
MGPU_LAUNCH_BOUNDS void KernelIntervalExpand(int destCount, 
    IndicesIt indices_global, ValuesIt values_global, int sourceCount, 
    const int* mp_global, OutputIt output_global) {
 
    typedef MGPU_LAUNCH_PARAMS Tuning;
    const int NT = Tuning::NT;
    const int VT = Tuning::VT;
    typedef typename std::iterator_traits<ValuesIt>::value_type T;
 
    union Shared {
        int indices[NT * (VT + 1)];
        T values[NT * VT];
    };
    __shared__ Shared shared;
    int tid = threadIdx.x;
    int block = blockIdx.x;
 
    // Compute the input and output intervals this CTA processes.
    int4 range = CTALoadBalance<NT, VT>(destCount, indices_global, 
sourceCount,
        block, tid, mp_global, shared.indices, true);
 
    // The interval indices are in the left part of shared memory 
(moveCount).
    // The scan of interval counts are in the right part 
(intervalCount).
    destCount = range.y - range.x;
    sourceCount = range.w - range.z;
 
    // Copy the source indices into register.
    int sources[VT];
    DeviceSharedToReg<NT, VT>(NT * VT, shared.indices, tid, sources);
 
    // Load the source fill values into shared memory. Each value is 
fetched
    // only once to reduce latency and L2 traffic.
    DeviceMemToMemLoop<NT>(sourceCount, values_global + range.z, tid,
        shared.values);
     
    // Gather the values from shared memory into register. This uses a 
shared
    // memory broadcast - one instance of a value serves all the 
threads that

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/intervalmove.cuh
http://nvlabs.github.io/moderngpu/intro.html#expandloadbalance


93
94

    // comprise its fill operation.
    T values[VT];
    DeviceGather<NT, VT>(destCount, shared.values - range.z, sources, 
tid,
        values, false);
 
    // Store the values to global memory.
    DeviceRegToGlobal<NT, VT>(destCount, values, tid, output_global + 
range.x);
}

The load-balancing search maps output elements (range.x, range.y) and input elements (range.z, 
range.w) into each tile. All earlier routines process items in thread order—thread tid loads elements VT * 
tid + i (0 ≤ i < VT) from shared memory into register and processes them in register in an unrolled loop. This 
pattern is used to implement the load-balancing search called at the top of KernelIntervalExpand as 
boilerplate, but it's not used to implement the specific behavior of IntervalExpand. 

The load-balancing search returns enough context in shared memory to allow the kernel to cooperatively 
process elements in strided order rather than the customary thread order—thread tid processes elements 
index = NT * i + tid, where index < destCount. The change from strided to thread order means we can use the 
device functions in device/loadstore.cuh to solve this problem.

After CTALoadBalance is run we cooperatively load source indices from shared memory into register. 
These indices identify the "generating objects" for each output; that is, the search locates the index of the fill 
value for each of the outputs.

Expand counts:
    0:     2    5    7   16    0    1    0    0   14   10
   10:     3   14    2    1   11    2    1    0    5    6

Scan of expand counts:
    0:     0    2    7   14   30   30   31   31   31   45
   10:    55   58   72   74   75   86   88   89   89   94

Load-balancing search:
    0:     0    0    1    1    1    1    1    2    2    2
   10:     2    2    2    2    3    3    3    3    3    3
   20:     3    3    3    3    3    3    3    3    3    3
   30:     5    8    8    8    8    8    8    8    8    8
   40:     8    8    8    8    8    9    9    9    9    9
   50:     9    9    9    9    9   10   10   10   11   11
   60:    11   11   11   11   11   11   11   11   11   11
   70:    11   11   12   12   13   14   14   14   14   14
   80:    14   14   14   14   14   14   15   15   16   18
   90:    18   18   18   18   19   19   19   19   19   19

Consider the example of IntervalExpand in demo.cu. We have 20 random expand counts that add up to 100. 
The client performs an exclusive scan over the counts and calls IntervalExpand. Our kernel runs 
CTALoadBalance to pair each of the 100 outputs with one of the 20 fill values. 
DeviceMemToMemLoop loads the interval of fill values (each associated with a "generating object") 
referenced by the tile into shared memory. Because the load-balancing search maps in a constant number of 
output plus input items to each tile, there's no risk of not having enough shared memory capacity in the CTA 
to accommodate this load: a run of thousands of 0 counts would result in a CTA that is mapped to a full-tile 
of source objects (NV + 1, keeping in mind the precedingB index) and no output objects. Although this may 
seem like an inefficiency, this division of source and destination items lets the kernel handle any distribution 

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu

of expand counts without any special-case code.

Expand values:
    0:     1    1    2    3    5    8   13   21   34   55
   10:    89  144  233  377  610  987 1597 2584 4181 6765

Expanded data:
    0:     1    1    1    1    1    1    1    2    2    2
   10:     2    2    2    2    3    3    3    3    3    3
   20:     3    3    3    3    3    3    3    3    3    3
   30:     8   34   34   34   34   34   34   34   34   34
   40:    34   34   34   34   34   55   55   55   55   55
   50:    55   55   55   55   55   89   89   89  144  144
   60:   144  144  144  144  144  144  144  144  144  144
   70:   144  144  233  233  377  610  610  610  610  610
   80:   610  610  610  610  610  610  987  987 1597 4181
   90:  4181 4181 4181 4181 6765 6765 6765 6765 6765 6765

The fill values, the first 20 numbers of the Fibonacci sequence, are cooperatively loaded into shared memory. 
DeviceGather cooperatively gathers the fill values for all destCount outputs using the source indices 
computed by CTALoadBalance and pulled from shared memory earlier. Because we process data in 
strided order rather than thread order, we can store directly to global memory without first having to 
transpose through shared memory: DeviceRegToGlobal cooperatively stores outputs to 
dest_global.

Important: The techniques illustrated on this page are about exposing parallelism in irregular problems. 
The naive approach for IntervalExpand would be to assign one thread to each source value: each thread reads 
its source value and copies it a variable-number of times to the output. This is a miserably unsatsifying 
solution, though. Huge penalties are taken due to control and memory divergence, the L2 cache is thrashed, 
and depending on the distribution of counts in the problem, there may not even be enough parallelism to even 
keep the device busy. 

To address load imbalance, the developer could try to build heuristics that examine the expand counts and 
assign different widths of execution to each source value. Entire warps could be assigned to sources that fill 
more than 128 outputs, and whole CTAs to sources that fill more than 2048 elements, for example. But now 
we are innovating scheduling strategy rather than simply solving the interval expand problem. 
CTALoadBalance incurs only a modest cost to expose parallelism and nimbly load-balance any data 
distribution for this very common class of problems. Instead of thinking about scheduling you can focus on 
solving your problem.

IntervalMove

IntervalMove is a vectorized cudaMemcpy. The caller enqueues transfers with (source offset, dest offset, 
item count) tuples. As with IntervalExpand, the counts are scanned prior to launch. The ability to load 
balance many cudaMemcpys over a single launch is crucial to performance—CUDA synchronizes at every 
cudaMemcpy, so calling that API directly will not deliver high throughput for many small requests. Host 
code may enqueue any number of transfers of any size and expect reasonable performance from 
IntervalMove.

IntervalMove and its special-case siblings—IntervalGather and IntervalScatter—are important primitives for 
GPU data structures. You can imagine "shaggy" binned data structures that resemble priority queues:

• IntervalGather pulls items from the front bins.



• IntervalScatter distributes sorted elements into the ends of all the bins.

• IntervalMove, segmented sort, merge, and vectorized sorted search cooperate in joining and splitting 
bins to rebalance the data structure.

Operations could be scheduled on the CPU and executed en masse with MGPU's vectorized functions. It is 
hoped that the availability of these functions encourages users to experiment with parallel data structures, an 
area of computing that has gone almost totally unexamined.

include/kernels/intervalmove.cuh

template<typename Tuning, bool Gather, bool Scatter, typename GatherIt,
    typename ScatterIt, typename IndicesIt, typename InputIt, typename 
OutputIt>
MGPU_LAUNCH_BOUNDS void KernelIntervalMove(int moveCount,
    GatherIt gather_global, ScatterIt scatter_global, IndicesIt 
indices_global, 
    int intervalCount, InputIt input_global, const int* mp_global, 
    OutputIt output_global) {
     
    typedef MGPU_LAUNCH_PARAMS Params;
    const int NT = Params::NT;
    const int VT = Params::VT;
 
    __shared__ int indices_shared[NT * (VT + 1)];
    int tid = threadIdx.x;
    int block = blockIdx.x;
 
    // Load balance the move IDs (counting_iterator) over the scan of 
the
    // interval sizes.
    int4 range = CTALoadBalance<NT, VT>(moveCount, indices_global, 
        intervalCount, block, tid, mp_global, indices_shared, true);
 
    // The interval indices are in the left part of shared memory 
(moveCount).
    // The scan of interval counts are in the right part 
(intervalCount).
    moveCount = range.y - range.x;
    intervalCount = range.w - range.z;
    int* move_shared = indices_shared;
    int* intervals_shared = indices_shared + moveCount;
    int* intervals_shared2 = intervals_shared - range.z;
 
    // Read out the interval indices and scan offsets.
    int interval[VT], rank[VT];
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        int index = NT * i + tid;
        int gid = range.x + index;
        interval[i] = range.z;
        if(index < moveCount) {

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/intervalmove.cuh
http://nvlabs.github.io/moderngpu/sortedsearch.html
http://nvlabs.github.io/moderngpu/merge.html
http://nvlabs.github.io/moderngpu/segsort.html


            interval[i] = move_shared[index];
            rank[i] = gid - intervals_shared2[interval[i]];
        }
    }
    __syncthreads();

The IntervalMove host function runs an upper-bound MergePathPartitions in preparation for the 
load-balancing search. KernelIntervalMove calls CTALoadBalance which computes source indices 
into shared memory. The interval index (i.e. the index of the request that generated the output) and rank (of 
the element within the interval) are cooperatively pulled from shared memory in strided order. Recall that the 
rank is the difference between the output index and the exclusive scan of the generating object—both of these 
terms are returned by CTALoadBalance.

KernelIntervalMove (continued) from include/kernels/intervalmove.cuh

    // Load and distribute the gather and scatter indices.
    int gather[VT], scatter[VT];
    if(Gather) {
        // Load the gather pointers into intervals_shared.
        DeviceMemToMemLoop<NT>(intervalCount, gather_global + range.z, 
tid,
            intervals_shared);
 
        // Make a second pass through shared memory. Grab the start 
indices of
        // the interval for each item and add the scan into it for the 
gather
        // index.
        #pragma unroll
        for(int i = 0; i < VT; ++i)
            gather[i] = intervals_shared2[interval[i]] + rank[i];
        __syncthreads();
    } 
    if(Scatter) {
        // Load the scatter pointers into intervals_shared.
        DeviceMemToMemLoop<NT>(intervalCount, scatter_global + 
range.z, tid,
            intervals_shared);
 
        // Make a second pass through shared memory. Grab the start 
indices of
        // the interval for each item and add the scan into it for the 
scatter
        // index.
        #pragma unroll
        for(int i = 0; i < VT; ++i)
            scatter[i] = intervals_shared2[interval[i]] + rank[i];
        __syncthreads();
    }
 
    // Gather the data into register.
    typedef typename std::iterator_traits<InputIt>::value_type T;

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/intervalmove.cuh

    T data[VT];
    if(Gather)
        DeviceGather<NT, VT>(moveCount, input_global, gather, tid, 
data, false);
    else
        DeviceGlobalToReg<NT, VT>(moveCount, input_global + range.x, 
tid, data);
 
    // Scatter the data into global.
    if(Scatter)
        DeviceScatter<NT, VT>(moveCount, data, tid, scatter, 
output_global,
            false);
    else
        DeviceRegToGlobal<NT, VT>(moveCount, data, tid, 
            output_global + range.x);
}

If gather indices are needed (for IntervalGather and IntervalMove), they are cooperatively loaded into shared 
memory. This is just one load per interval. After synchronization the gather indices (the source index for 
each copy request) are loaded from shared memory into register. We add the rank of the element into the 
gather index to produce a load index. For the i'th output, gather is the position to load the i'th input.

Scatter indices are treated symmetrically. If scatter indices are needed (for IntervalScatter and IntervalMove), 
they are cooperatively loaded into shared memory. This is just one load per interval. Scatter indices are 
loaded from shared memory into register. The rank is added to produce a store index. scatter is the 
position to store the i'th output.

As with IntervalExpand, the intra-CTA load-balancing search provides enough context so that each element 
can be processed independently. Rather than processing elements in thread order, where each thread 
processes elements VT * tid + i (0 ≤ i < VT), we cooperatively copy elements in strided order. The 
loadstore.cuh support functions complete the vectorized cudaMemcpys.

https://github.com/NVlabs/moderngpu/blob/master/include/device/loadstore.cuh




12. Relational Joins
Sort-merge joins supporting inner, left, right, and outer variants. Uses vectorized sorted search to match keys 
between input arrays and load-balancing search to manage Cartesian products.

Benchmark and usage



Relational joins benchmark from tests/benchmarkjoin.cu

Relational joins demonstration from tests/demo.cu

void DemoJoin(CudaContext& context) {
    printf("RELATIONAL JOINS DEMONSTRATION\n\n");
 
    int ACount = 30;
    int BCount = 30;
 
    MGPU_MEM(int) aKeysDevice = context.SortRandom<int>(ACount, 100, 
130);
    MGPU_MEM(int) bKeysDevice = context.SortRandom<int>(BCount, 100, 
130);
    std::vector<int> aKeysHost, bKeysHost;
    aKeysDevice->ToHost(aKeysHost);
    bKeysDevice->ToHost(bKeysHost);
 
    printf("A keys:\n");
    PrintArray(*aKeysDevice, "%4d", 10);
 
    printf("\nB keys:\n");
    PrintArray(*bKeysDevice, "%4d", 10);
 
    MGPU_MEM(int) aIndices, bIndices;
    int innerCount = RelationalJoin<MgpuJoinKindInner>(aKeysDevice-
>get(),
        ACount, bKeysDevice->get(), BCount, &aIndices, &bIndices, 
context);

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarkjoin.cu

    std::vector<int> aHost, bHost;
    aIndices->ToHost(aHost);
    bIndices->ToHost(bHost);
 
    printf("\nInner-join (%d items):\n", innerCount);
    printf("output   (aIndex, bIndex) : (aKey, bKey)\n");
    printf("----------------------------------------\n");
    for(int i = 0; i < innerCount; ++i)
        printf("%3d      (%6d, %6d) : (%4d, %4d)\n", i, aHost[i], 
bHost[i],
            aKeysHost[aHost[i]], bKeysHost[bHost[i]]);
 
    int outerCount = RelationalJoin<MgpuJoinKindOuter>(aKeysDevice-
>get(),
        ACount, bKeysDevice->get(), BCount, &aIndices, &bIndices, 
context);
 
    aIndices->ToHost(aHost);
    bIndices->ToHost(bHost);
    printf("\nOuter-join (%d items):\n", outerCount);
    printf("output   (aIndex, bIndex) : (aKey, bKey)\n");
    printf("----------------------------------------\n");
    for(int i = 0; i < outerCount; ++i) {
        char aKey[5], bKey[5];
        if(-1 != aHost[i]) itoa(aKeysHost[aHost[i]], aKey, 10);
        if(-1 != bHost[i]) itoa(bKeysHost[bHost[i]], bKey, 10);
        printf("%3d      (%6d, %6d) : (%4s, %4s)\n", i, aHost[i], 
bHost[i],
            (-1 != aHost[i]) ? aKey : "---", (-1 != bHost[i]) ? 
bKey : "---");
    }
}

RELATIONAL JOINS DEMONSTRATION

A keys:
    0:   100  102  103  103  103  103  103  104  104  105
   10:   106  106  106  107  108  109  109  110  111  113
   20:   114  114  114  116  116  116  118  119  121  127

B keys:
    0:   100  101  102  102  105  105  105  105  106  107
   10:   109  112  116  117  117  118  119  121  125  125
   20:   126  126  126  126  128  128  128  129  130  130

Inner-join (19 items):
output   (aIndex, bIndex) : (aKey, bKey)
----------------------------------------
  0      (     0,      0) : ( 100,  100)
  1      (     1,      2) : ( 102,  102)
  2      (     1,      3) : ( 102,  102)
  3      (     9,      4) : ( 105,  105)
  4      (     9,      5) : ( 105,  105)



  5      (     9,      6) : ( 105,  105)
  6      (     9,      7) : ( 105,  105)
  7      (    10,      8) : ( 106,  106)
  8      (    11,      8) : ( 106,  106)
  9      (    12,      8) : ( 106,  106)
 10      (    13,      9) : ( 107,  107)
 11      (    15,     10) : ( 109,  109)
 12      (    16,     10) : ( 109,  109)
 13      (    23,     12) : ( 116,  116)
 14      (    24,     12) : ( 116,  116)
 15      (    25,     12) : ( 116,  116)
 16      (    26,     15) : ( 118,  118)
 17      (    27,     16) : ( 119,  119)
 18      (    28,     17) : ( 121,  121)

Outer-join (50 items):
output   (aIndex, bIndex) : (aKey, bKey)
----------------------------------------
  0      (     0,      0) : ( 100,  100)
  1      (     1,      2) : ( 102,  102)
  2      (     1,      3) : ( 102,  102)
  3      (     2,     -1) : ( 103,  ---)
  4      (     3,     -1) : ( 103,  ---)
  5      (     4,     -1) : ( 103,  ---)
  6      (     5,     -1) : ( 103,  ---)
  7      (     6,     -1) : ( 103,  ---)
  8      (     7,     -1) : ( 104,  ---)
  9      (     8,     -1) : ( 104,  ---)
 10      (     9,      4) : ( 105,  105)
 11      (     9,      5) : ( 105,  105)
 12      (     9,      6) : ( 105,  105)
 13      (     9,      7) : ( 105,  105)
 14      (    10,      8) : ( 106,  106)
 15      (    11,      8) : ( 106,  106)
 16      (    12,      8) : ( 106,  106)
 17      (    13,      9) : ( 107,  107)
 18      (    14,     -1) : ( 108,  ---)
 19      (    15,     10) : ( 109,  109)
 20      (    16,     10) : ( 109,  109)
 21      (    17,     -1) : ( 110,  ---)
 22      (    18,     -1) : ( 111,  ---)
 23      (    19,     -1) : ( 113,  ---)
 24      (    20,     -1) : ( 114,  ---)
 25      (    21,     -1) : ( 114,  ---)
 26      (    22,     -1) : ( 114,  ---)
 27      (    23,     12) : ( 116,  116)
 28      (    24,     12) : ( 116,  116)
 29      (    25,     12) : ( 116,  116)
 30      (    26,     15) : ( 118,  118)
 31      (    27,     16) : ( 119,  119)
 32      (    28,     17) : ( 121,  121)
 33      (    29,     -1) : ( 127,  ---)
 34      (    -1,      1) : ( ---,  101)
 35      (    -1,     11) : ( ---,  112)
 36      (    -1,     13) : ( ---,  117)
 37      (    -1,     14) : ( ---,  117)
 38      (    -1,     18) : ( ---,  125)
 39      (    -1,     19) : ( ---,  125)
 40      (    -1,     20) : ( ---,  126)
 41      (    -1,     21) : ( ---,  126)
 42      (    -1,     22) : ( ---,  126)
 43      (    -1,     23) : ( ---,  126)
 44      (    -1,     24) : ( ---,  128)



 45      (    -1,     25) : ( ---,  128)
 46      (    -1,     26) : ( ---,  128)
 47      (    -1,     27) : ( ---,  129)
 48      (    -1,     28) : ( ---,  130)
 49      (    -1,     29) : ( ---,  130)

Host functions

include/mgpuhost.cuh

//////////////////////////////////////////////////////////////////////
//////////
// kernels/join.cuh
 
// RelationalJoin is a sort-merge join that returns indices into one 
of the four
// relational joins:
//      MgpuJoinKindInner
//      MgpuJoinKindLeft
//      MgpuJoinKindRight
//      MgpuJoinKindOuter.
 
// A =  100, 101, 103, 103
// B =  100, 100, 102, 103
// Outer join:
//     ai, bi   a[ai], b[bi]
// 0:  (0, 0) -  (100, 100)    // cross-product expansion for key 100
// 1:  (0, 1) -  (100, 100)
// 2:  (1, -) -  (101, ---)    // left-join for key 101
// 3:  (-, 2) -  (---, 102)    // right-join for key 102
// 4:  (3, 3) -  (103, 103)    // cross-product expansion for key 103
 
// MgpuJoinKindLeft drops the right-join on line 3.
// MgpuJoinKindRight drops the left-join on line 2.
// MgpuJoinKindInner drops both the left- and right-joins.
 
// The caller passes MGPU_MEM(int) pointers to hold indices. Memory is 
allocated
// by the join function using the allocator associated with the 
context. It 
// returns the number of outputs.
 
// RelationalJoin performs one cudaMemcpyDeviceToHost to retrieve the 
size of
// the output array. This is a synchronous operation and may prevent 
queueing
// for callers using streams.
template<MgpuJoinKind Kind, typename InputIt1, typename InputIt2,
    typename Comp>
MGPU_HOST int RelationalJoin(InputIt1 a_global, int aCount, InputIt2 
b_global,
    int bCount, MGPU_MEM(int)* ppAJoinIndices, MGPU_MEM(int)* 

https://github.com/NVlabs/moderngpu/blob/master/include/mgpuhost.cuh

ppBJoinIndices, 
    Comp comp, CudaContext& context);
 
// Specialization of RelationJoil with Comp = mgpu::less<T>.
template<MgpuJoinKind Kind, typename InputIt1, typename InputIt2>
MGPU_HOST int RelationalJoin(InputIt1 a_global, int aCount, InputIt2 
b_global,
    int bCount, MGPU_MEM(int)* ppAJoinIndices, MGPU_MEM(int)* 
ppBJoinIndices, 
    CudaContext& context);

Algorithm

Join is a foundational operation in relational algebra and relational databases. Joins take two tables and return 
a new table. A column from each table serves as a key and the join operator produces the Cartesian product 
of all rows with matching keys. MGPU Join is a merge-join that supports duplicate keys and left-, right-, and 
outer-join semantics.

Row: 0 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15

Table A: A0 A1 B0 E0 E1 E2 E3 F0 F1 G0 H0 H1 J0 J1 M0 M1

Table B: A0 A1 B0 B1 B2 C0 C1 F0 G0 G1 H0 I0 L0 L1

Table A has 16 rows and B has 14. The key fields for the join are displayed above. The keys are sorted 
within each table (a requirement for merge-join) and the ranks of the keys are inferred. We join over 
matching letters and generate a Cartesian product for all ranks.

Row A index A key B key B index Join type

0 0 A0 A0 0 inner

1 0 A0 A1 1 inner

2 1 A1 A0 0 inner

3 1 A1 A1 1 inner

4 2 B0 B0 2 inner

5 2 B0 B1 3 inner

6 2 B0 B2 4 inner

7 3 E0 --- -1 left

http://en.wikipedia.org/wiki/Sort-merge_join
http://en.wikipedia.org/wiki/Cartesian_product
https://en.wikipedia.org/wiki/Relational_database
http://en.wikipedia.org/wiki/Relational_algebra#Joins_and_join-like_operators
http://en.wikipedia.org/wiki/Join_(SQL)


8 4 E1 --- -1 left

9 5 E2 --- -1 left

10 6 E3 --- -1 left

11 7 F0 F0 7 inner

12 8 F1 F0 7 inner

13 9 G0 G0 8 inner

14 9 G0 G1 9 inner

15 10 H0 H0 10 inner

16 11 H1 H0 10 inner

17 12 J0 --- -1 left

18 13 J1 --- -1 left

19 14 M0 --- -1 left

20 15 M1 --- -1 left

21 -1 --- C0 5 right

22 -1 --- C1 6 right

23 -1 --- I0 11 right

24 -1 --- L0 12 right

25 -1 --- L1 13 right

Merge-join takes sorted inputs and returns an output table that is sorted first by A, and within matching A 
keys, by B keys. 

• Inner-join produces the Cartesian product of matching keys over their ranks. In this case both keys 
are non-null and indices are defined.

• Left-join adds to inner-join the set of rows in A with keys that are unmatched in B. With left-join, all 
rows in the A input are included in the output. The B component of left-join tuples is a null key. 
MGPU Join uses the index -1 for null keys; this key is lexicographically larger than all other keys.



• Right-join adds to inner-join the set of rows in B with keys that are unmatched in A. With right-join, 
all rows in the B input are included in the output. The A component of right-join tuples is a null key. 
Because outputs are sorted first by A key and then by B key, rows generated by right-join are 
appended to the end of the output table and are sorted by B index.

• Full outer-join is the union of inner-, left-, and outer-join rows. All input rows are returned by an 
outer-join operation.

MGPU Join supports all four join types. It returns a dynamically-allocated set of A index/B index pairs. The 
caller can retrieve the joined keys with a simple gather.

The function is implemented by leaning heavily on vectorized sorted search and load-balancing search. It is 
decomposed into a few simple steps:

1. Use vectorized sorted search to find the lower-bound of A into B. For right/outer-join, also return the 
set of matches and the match count of B into A.

2. Use SortedEqualityCount to find the number of matches in B for each element in A. For 
left/outer-join, use the LeftJoinEqualityOp operator to always return a count of at least 1.

3. Scan the Cartesian-product counts in 2 and save the reduction of the counts as leftJoinTotal, 
which is the number of outputs contributed by the left/inner-join parts.

4. Add the leftJoinTotal in 3 with the right-join total in 1 (the right-join contribution is the 
number of elements in B that have no matches in A) and save as joinTotal. Allocate device 
memory to hold this many join outputs.

5. Run an upper-bound MergePath search as part of the load-balancing search that enables the 
left/inner-join implementation.

6. Launch KernelLeftJoin:

a. Run the CTALoadBalance boilerplate: rows in the A table are considered "generating 
objects" and outputs are work-items.

b. Threads locate the rank of each output within its generating object—that is, the superscript on 
the B key of the output. In the table above, output row 2 (A1, A0) is rank 0, because it is the 
first element of the Cartesian product with key A1 on the left. Output row 3 (A1, A1) is rank 1, 
because it is the second element of the Cartesian product with key A1 on the left. Left-join 
outputs always are rank 0.

c. Cooperatively load the lower-bound of A into B (computed in 1) into shared memory for each 
row of table A that is referenced inside the CTA.

d. Store the left/inner-join indices to global memory. The A index is the generating object's index 
as computed by the load-balancing search. For an inner-join, the B index is the lower-bound of 
A into B plus the rank of the output. For a left-join, the B index is -1, representing the null 
key.

7. For a right/outer-join, compact the indices of the rows in B that were not matched in 1 to the end of 
the output array. cudaMemset -1s to the corresponding A indices.

http://nvlabs.github.io/moderngpu/loadbalance.html
http://nvlabs.github.io/moderngpu/sortedsearch.html

template<MgpuJoinKind Kind, typename InputIt1, typename InputIt2,
    typename Comp>
MGPU_HOST int RelationalJoin(InputIt1 a_global, int aCount, InputIt2 
b_global,
    int bCount, MGPU_MEM(int)* ppAJoinIndices, MGPU_MEM(int)* 
ppBJoinIndices, 
    Comp comp, CudaContext& context) {
 
    typedef typename std::iterator_traits<InputIt1>::value_type T;
    const bool SupportLeft = MgpuJoinKindLeft == Kind || 
        MgpuJoinKindOuter == Kind;
    const bool SupportRight = MgpuJoinKindRight == Kind ||
        MgpuJoinKindOuter == Kind;
 
    const MgpuSearchType LeftType = SupportLeft ? 
        MgpuSearchTypeIndexMatch : MgpuSearchTypeIndex;
 
    MGPU_MEM(int) aLowerBound = context.Malloc<int>(aCount);
    MGPU_MEM(byte) bMatches;
 
    // Find the lower bound of A into B. If we are right joining also 
return the
    // set of matches of B into A.
    int rightJoinTotal = 0;
    if(SupportRight) {
        // Support a right or outer join. Count the number of B 
elements that
        // have matches in A. These matched values are included in the 
inner
        // join part. The other values (bCount - bMatchCount) are 
copied to the
        // end for the right join part.
        bMatches = context.Malloc<byte>(bCount);
        int bMatchCount;
        SortedSearch<MgpuBoundsLower, LeftType, 
MgpuSearchTypeMatch>(a_global,
            aCount, b_global, bCount, aLowerBound->get(), bMatches-
>get(), comp,
            context, 0, &bMatchCount);
        rightJoinTotal = bCount - bMatchCount;
    } else
        SortedSearch<MgpuBoundsLower, LeftType, 
MgpuSearchTypeNone>(a_global,
            aCount, b_global, bCount, aLowerBound->get(), (int*)0, 
comp,
            context, 0, 0);

The host function RelationalJoin starts by calling SortedSearch to find the lower-bound of A into 
B. The function is specialized over one of four possible parameterizations, depending on join type. 

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/join.cuh


When supporting left-join, matches of A into B are computed in addition to the lower-bound indices—
matches are indicated by setting the high bit of the indices. 

When supporting right-join, matches of B into A are returned in bytes. All we really need are bits, but those 
aren't directly addressable. The total number of matches is returned in the last SortedSearch argument; it 
is subtracted from the size of the B array: this is the number of right-join rows to append to the end of the 
output.

RelationalJoin (continued) from include/kernels/join.cuh

    // Use the lower bounds to compute the counts for each element.
    MGPU_MEM(int) aCounts = context.Malloc<int>(aCount);
    if(SupportLeft) 
        SortedEqualityCount(a_global, aCount, b_global, bCount,
            aLowerBound->get(), aCounts->get(), comp, 
LeftJoinEqualityOp(), 
            context);
    else
        SortedEqualityCount(a_global, aCount, b_global, bCount,
            aLowerBound->get(), aCounts->get(), comp, 
SortedEqualityOp(), 
            context);
 
    // Scan the product counts. This is part of the load-balancing 
search.
    int leftJoinTotal = Scan(aCounts->get(), aCount, context);

The second section calls SortedEqualityCount: an upper-bound of A into B is run and its different 
from the lower-bound in returned as a count. This is the count of B values created for each A value—the 
Cartesian product is implemented by generating a variable number of outputs for each individual element of 
A. To support left-join, we specialize with the LeftJoinEqualityOp; this returns a 1 count when there 
are no elements in B matching a key in A. Because the join kernel uses load-balancing search we scan the 
counts in-place. This creates a sorted array that can be pushed through the upper-bound 
MergePathPartitions.

RelationalJoin (continued) from include/kernels/join.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/join.cuh
http://nvlabs.github.io/moderngpu/sortedsearch.html#equalitycount
https://github.com/NVlabs/moderngpu/blob/master/include/kernels/join.cuh


243
244
245
246
247
248
249
250
251
252
253
254
255
256
257
258
259
260
261
262
263
264
265

    // Allocate space for the join indices from the sum of left and 
right join
    // sizes.
    int joinTotal = leftJoinTotal + rightJoinTotal;
    MGPU_MEM(int) aIndicesDevice = context.Malloc<int>(joinTotal);
    MGPU_MEM(int) bIndicesDevice = context.Malloc<int>(joinTotal);
 
    // Launch the inner/left join kernel. Run an upper-bounds 
partitioning 
    // to load-balance the data.
    const int NT = 128;
    const int VT = 7;
    typedef LaunchBoxVT<NT, VT> Tuning;
    int2 launch = Tuning::GetLaunchParams(context);
    int NV = launch.x * launch.y;
 
    MGPU_MEM(int) partitionsDevice = 
MergePathPartitions<MgpuBoundsUpper>(
        mgpu::counting_iterator<int>(0), leftJoinTotal, aCounts-
>get(),
        aCount, NV, 0, mgpu::less<int>(), context);
 
    int numBlocks = MGPU_DIV_UP(leftJoinTotal + aCount, NV);
    KernelLeftJoin<Tuning, SupportLeft>
        <<<numBlocks, launch.x, 0, context.Stream()>>>(leftJoinTotal, 
        aLowerBound->get(), aCounts->get(), aCount, partitionsDevice-
>get(),
        aIndicesDevice->get(), bIndicesDevice->get());

A LaunchBox is created to support device-specific parameterizations. Although we launch a number of 
routines from RelationalJoin, we only control the tuning parameters for KernelLeftJoin—the 
other kernels are pre-packaged in host functions that define their own launch parameters. 

Index pairs are allocated, MergePathPartitions is called to prepare the load-balancing search, and 
KernelLeftJoin is launched. This kernel performs both the left- and inner-join parts. Right-join is a 
comparatively trivial operation involving a simple index compaction to the end of the index arrays. It is 
saved for the end.

include/kernels/join.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/join.cuh
http://nvlabs.github.io/moderngpu/performance.html#launchbox

template<typename Tuning, bool LeftJoin>
MGPU_LAUNCH_BOUNDS void KernelLeftJoin(int total, const int* 
aLowerBound_global,
    const int* aCountsScan_global, int aCount, const int* mp_global,
    int* aIndices_global, int* bIndices_global) { 
 
    typedef MGPU_LAUNCH_PARAMS Params;
    const int NT = Params::NT;
    const int VT = Params::VT;
 
    __shared__ int indices_shared[NT * (VT + 1)];
    int tid = threadIdx.x;
    int block = blockIdx.x;
 
    int4 range = CTALoadBalance<NT, VT>(total, aCountsScan_global, 
aCount,
        block, tid, mp_global, indices_shared, true);
    int outputCount = range.y - range.x;
    int inputCount = range.w - range.z;
    int* output_shared = indices_shared;
    int* input_shared = indices_shared + outputCount;
 
    int aIndex[VT], rank[VT];
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        int index = NT * i + tid;
        if(index < outputCount) {
            int gid = range.x + index;
            aIndex[i] = output_shared[index];
            rank[i] = gid - input_shared[aIndex[i] - range.z];
            aIndices_global[gid] = aIndex[i];
        }
    }
    __syncthreads();
 
    // Load the lower bound of A into B for each element of A.
    DeviceMemToMemLoop<NT>(inputCount, aLowerBound_global + range.z, 
tid,
        input_shared);
 
    // Store the lower bound of A into B back for every output.
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        int index = NT * i + tid;
        if(index < outputCount) {
            int gid = range.x + index;
            int lb = input_shared[aIndex[i] - range.z];
            int bIndex;
            if(LeftJoin)
                bIndex = (0x80000000 & lb) ? 
                    ((0x7fffffff & lb) + rank[i]) :
                    -1;
            else



The left-join kernel closely resembles Interval Move. We load-balance outputs and inputs (each input is one 
row of A) in shared memory. Ranks for each output are computed. 

Recall the figure from the top: the A-rank of an output row is equal to the superscript of the B key. Row 6, 
for example, match keys "B". It is the third occurrence of B0 in A, or rank 2 (we count ranks in zero-based 
indexing). Therefore it must be paired with B2 in B.

Load-balancing search provides the rank of each key occurrence in A. The rank is used to infer the index of 
the corresponding row in B. The lower-bound of A into B, computed earlier in RelationalJoin, 
provides the index of the first key-match in B. We add the rank of the output into this lower-bound for B's 
index in the output: 

bIndex = lb + rank[i]; Infer the B index from the lower-bound of A into B and the A-rank of the 
output row.

If the user has requested a left/outer-join we check the match bit of the lower-bound (the most significant 
bit), and emit the null index -1 to form a left-join output: 

bIndex = (0x80000000 & lb) ? ((0x7fffffff & lb) + rank[i]) : -1; Return the B 
index only if this is an inner-type output, as indicated by a set match bit on the lower-bound term.

RelationalJoin (continued) from include/kernels/join.cuh

    // Launch the right join kernel. Compact the non-matches from B 
into A.
    if(SupportRight) {
        const int NT = 128;
        int numBlocks = MGPU_DIV_UP(bCount, 8 * NT);
 
        MGPU_MEM(int) totals = context.Malloc<int>(numBlocks);
        KernelRightJoinUpsweep<NT><<<numBlocks, NT>>>(
            (const uint64*)bMatches->get(), bCount, totals->get());
         
        Scan<MgpuScanTypeExc>(totals->get(), numBlocks, totals->get(),
            ScanOpAdd(), (int*)0, false, context);
 
        KernelRightJoinDownsweep<NT><<<numBlocks, NT>>>(
            (const uint64*)bMatches->get(), bCount, totals->get(), 
            bIndicesDevice->get() + leftJoinTotal);
 
        cudaMemset(aIndicesDevice->get() + leftJoinTotal, -1, 
            sizeof(int) * rightJoinTotal);
    }
 
    *ppAJoinIndices = aIndicesDevice;
    *ppBJoinIndices = bIndicesDevice;
    return joinTotal;
}

The right-join code performs a simple index compaction into the end of the output arrays. 
KernelRightJoinUpsweep counts the number of elements in B that do not have matches in A. (Recall 
that we already computed the match terms into a byte array with a vectorized sorted search specialization.) 

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/join.cuh
http://nvlabs.github.io/moderngpu/join.html#algorithm
http://nvlabs.github.io/moderngpu/intervalmove.html#intervalmove


The partials are scanned to find the offset within the output array for each tile to stream its indices. 
KernelRightJoinDownsweep revisits the match flags and streams the B indices. We finalize the 
relational join by setting the A indices to -1, indicating a null key and a right-join output.



13. Multisets
Replace Merge Path partitioning with the sophisticated Balanced Path to search for key-rank matches. The 
new partitioning strategy is combined with four different serial set operations to support CUDA analogs 
of std::set_intersection, set_union, set_difference, and set_symmetric_difference.

Benchmark and usage



Multisets-keys benchmark from tests/benchmarksets.cu

Multisets-keys demonstration from tests/demo.cu

void DemoSets(CudaContext& context) {
    printf("\nMULTISET-KEYS DEMONSTRATION:\n\n");
 
    // Use CudaContext::SortRandom to generate 100 random sorted 
integers 
    // between 0 and 99.
    int N = 100;
    MGPU_MEM(int) aData = context.SortRandom<int>(N, 0, 99);
    MGPU_MEM(int) bData = context.SortRandom<int>(N, 0, 99);
 
    printf("A:\n");
    PrintArray(*aData, "%4d", 10);
    printf("\nB:\n\n");
    PrintArray(*bData, "%4d", 10);
     
    MGPU_MEM(int) intersectionDevice;
    SetOpKeys<MgpuSetOpIntersection, true>(aData->get(), N, bData-
>get(), N,
        &intersectionDevice, context, false);
 
    printf("\nIntersection:\n");
    PrintArray(*intersectionDevice, "%4d", 10);
 
    MGPU_MEM(int) symDiffDevice;
    SetOpKeys<MgpuSetOpSymDiff, true>(aData->get(), N, bData->get(), 

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarksets.cu

N,
        &symDiffDevice, context, false);
 
    printf("\nSymmetric difference:\n");
    PrintArray(*symDiffDevice, "%4d", 10);
}

MULTISET-KEYS DEMONSTRATION:

A:
    0:     1    1    3    5    7    7    8    9   10   10
   10:    10   11   12   13   14   15   16   16   16   16
   20:    17   18   19   20   21   21   25   25   28   29
   30:    29   29   31   31   31   31   32   33   33   35
   40:    36   38   39   40   40   42   44   45   46   47
   50:    47   51   51   53   53   53   55   55   56   57
   60:    58   59   59   59   60   61   62   62   63   63
   70:    64   68   68   70   70   72   73   73   75   78
   80:    79   82   82   83   84   85   85   85   86   87
   90:    89   91   91   91   92   95   97   98   98   98

B:

    0:     1    2    2    3    5    6    6    9    9   10
   10:    10   10   11   12   12   12   13   13   15   16
   20:    16   17   17   18   21   21   22   24   25   25
   30:    29   29   31   32   32   32   33   35   35   37
   40:    39   39   40   41   41   42   42   44   45   46
   50:    46   47   48   49   50   50   51   52   52   53
   60:    54   54   54   55   56   57   59   60   65   65
   70:    66   66   66   67   68   68   70   72   74   74
   80:    74   74   74   75   76   76   80   82   89   89
   90:    90   92   92   93   93   95   95   96   97   98

Intersection:
    0:     1    3    5    9   10   10   10   11   12   13
   10:    15   16   16   17   18   21   21   25   25   29
   20:    29   31   32   33   35   39   40   42   44   45
   30:    46   47   51   53   55   56   57   59   60   68
   40:    68   70   72   75   82   89   92   95   97   98

Symmetric Difference:
    0:     1    2    2    6    6    7    7    8    9   12
   10:    12   13   14   16   16   17   19   20   22   24
   20:    28   29   31   31   31   32   32   33   35   36
   30:    37   38   39   40   41   41   42   46   47   48
   40:    49   50   50   51   52   52   53   53   54   54
   50:    54   55   58   59   59   61   62   62   63   63
   60:    64   65   65   66   66   66   67   70   73   73
   70:    74   74   74   74   74   76   76   78   79   80
   80:    82   83   84   85   85   85   86   87   89   90
   90:    91   91   91   92   93   93   95   96   98   98



Multisets-pairs benchmark from tests/benchmarksets.cu

Multisets-pairs demonstration from tests/demo.cu

https://github.com/NVlabs/moderngpu/blob/master/tests/demo.cu
https://github.com/NVlabs/moderngpu/blob/master/tests/benchmarksets.cu

void DemoSetsPairs(CudaContext& context) {
    printf("\nMULTISET-PAIRS DEMONSTRATION:\n\n");
 
    // Use CudaContext::SortRandom to generate 100 random sorted 
integers 
    // between 0 and 99.
    int N = 100;
    MGPU_MEM(int) aData = context.SortRandom<int>(N, 0, 99);
    MGPU_MEM(int) bData = context.SortRandom<int>(N, 0, 99);
 
    printf("A:\n");
    PrintArray(*aData, "%4d", 10);
    printf("\nB:\n\n");
    PrintArray(*bData, "%4d", 10);
     
    MGPU_MEM(int) intersectionDevice, intersectionValues;
    SetOpPairs<MgpuSetOpIntersection, true>(aData->get(), 
        mgpu::counting_iterator<int>(0), N, bData->get(), 
        mgpu::counting_iterator<int>(N), N, &intersectionDevice,
        &intersectionValues, context);
 
    printf("\nIntersection keys:\n");
    PrintArray(*intersectionDevice, "%4d", 10);
 
    printf("\nIntersection indices:\n");
    PrintArray(*intersectionValues, "%4d", 10);
 
    MGPU_MEM(int) symDiffDevice, symDiffValues;
    SetOpPairs<MgpuSetOpSymDiff, true>(aData->get(), 
        mgpu::counting_iterator<int>(0), N, bData->get(), 
        mgpu::counting_iterator<int>(N), N, &symDiffDevice, 
&symDiffValues, 
        context);
 
    printf("\nSymmetric difference keys:\n");
    PrintArray(*symDiffDevice, "%4d", 10);
 
    printf("\nSymmetric difference indices:\n");
    PrintArray(*symDiffValues, "%4d", 10);
}

MULTISET-PAIRS DEMONSTRATION:

A:
    0:     0    1    1    2    3    6    6    8   11   11
   10:    14   17   18   18   20   22   22   22   24   25
   20:    26   27   27   31   31   31   32   33   33   34
   30:    35   35   37   37   38   39   39   40   41   41
   40:    42   43   44   44   44   47   50   52   56   56
   50:    57   57   57   60   62   63   63   63   64   64
   60:    64   65   66   67   67   68   71   72   73   75
   70:    76   76   77   78   79   81   81   82   84   85
   80:    85   86   86   88   89   90   91   91   91   92



   90:    92   92   93   95   95   95   98   99   99   99

B:

    0:     0    1    2    2    4    4    4    4    5    6
   10:     6    8    8   10   10   12   13   14   18   21
   20:    21   22   22   22   24   26   26   27   28   28
   30:    30   32   33   34   35   38   38   38   39   40
   40:    40   41   41   42   43   44   45   45   48   51
   50:    53   53   53   53   54   55   57   61   61   61
   60:    62   62   64   64   66   66   67   68   70   70
   70:    72   74   76   78   78   79   80   80   80   80
   80:    81   81   87   88   88   89   91   91   92   93
   90:    93   93   94   96   97   98   98   98   98   99

Intersection keys:
    0:     0    1    2    6    6    8   14   18   22   22
   10:    22   24   26   27   32   33   34   35   38   39
   20:    40   41   41   42   43   44   57   62   64   64
   30:    66   67   68   72   76   78   79   81   81   88
   40:    89   91   91   92   93   98   99

Intersection indices:
    0:     0    1    3    5    6    7   10   12   15   16
   10:    17   18   20   21   26   27   29   30   34   35
   20:    37   38   39   40   41   42   50   54   58   59
   30:    62   63   65   67   70   73   74   75   76   83
   40:    84   86   87   89   92   96   97

Symmetric difference keys:
    0:     1    2    3    4    4    4    4    5    8   10
   10:    10   11   11   12   13   17   18   20   21   21
   20:    25   26   27   28   28   30   31   31   31   33
   30:    35   37   37   38   38   39   40   44   44   45
   40:    45   47   48   50   51   52   53   53   53   53
   50:    54   55   56   56   57   57   60   61   61   61
   60:    62   63   63   63   64   65   66   67   70   70
   70:    71   73   74   75   76   77   78   80   80   80
   80:    80   82   84   85   85   86   86   87   88   90
   90:    91   92   92   93   93   94   95   95   95   96
  100:    97   98   98   98   99   99

Symmetric difference indices:
    0:     2  103    4  104  105  106  107  108  112  113
   10:   114    8    9  115  116   11   13   14  119  120
   20:    19  126   22  128  129  130   23   24   25   28
   30:    31   32   33  136  137   36  140   43   44  146
   40:   147   45  148   46  149   47  150  151  152  153
   50:   154  155   48   49   51   52   53  157  158  159
   60:   161   55   56   57   60   61  165   64  168  169
   70:    66   68  171   69   71   72  174  176  177  178
   80:   179   77   78   79   80   81   82  182  184   85
   90:    88   90   91  190  191  192   93   94   95  193
  100:   194  196  197  198   98   99

Host functions

include/mgpuhost.cuh

522 /////////////////////////////////////////////////////////////////////

https://github.com/NVlabs/moderngpu/blob/master/include/mgpuhost.cuh

///////////
// kernels/sets.cuh
 
// SetOpKeys implements multiset operations with C++ set_* semantics.
// MgpuSetOp may be:
//      MgpuSetOpIntersection -     like std::set_intersection
//      MgpuSetOpUnion -            like std::set_union
//      MgpuSetOpDiff -             like std::set_difference
//      MgpuSetOpSymDiff -          like 
std::set_symmetric_difference
 
// Setting Duplicates to false increases performance for inputs with 
no 
// duplicate keys in either array.
 
// The caller passes MGPU_MEM(T) pointers to hold outputs. Memory is 
allocated
// by the multiset function using the allocator associated with the 
context. It 
// returns the number of outputs.
 
// SetOpKeys performs one cudaMemcpyDeviceToHost to retrieve the size 
of
// the output array. This is a synchronous operation and may prevent 
queueing
// for callers using streams.
 
// If compact = true, SetOpKeys pre-allocates an output buffer is 
large as the
// sum of the input arrays. Partials results are computed into this 
temporary
// array before being moved into the final array. It consumes more 
space but
// results in higher performance.
template<MgpuSetOp Op, bool Duplicates, typename It1, typename It2,
    typename T, typename Comp>
MGPU_HOST int SetOpKeys(It1 a_global, int aCount, It2 b_global, int 
bCount,
    MGPU_MEM(T)* ppKeys_global, Comp comp, CudaContext& context, 
    bool compact = true);
 
// Specialization of SetOpKeys with Comp = mgpu::less<T>.
template<MgpuSetOp Op, bool Duplicates, typename It1, typename It2, 
typename T>
MGPU_HOST int SetOpKeys(It1 a_global, int aCount, It2 b_global, int 
bCount,
    MGPU_MEM(T)* ppKeys_global, CudaContext& context, bool compact = 
true);
 
// SetOpPairs runs multiset operations by key and supports value 
exchange.
template<MgpuSetOp Op, bool Duplicates, typename KeysIt1, typename 



KeysIt2,
    typename ValsIt1, typename ValsIt2, typename KeyType, typename 
ValType,
    typename Comp>
MGPU_HOST int SetOpPairs(KeysIt1 aKeys_global, ValsIt1 aVals_global, 
int aCount,
    KeysIt2 bKeys_global, ValsIt2 bVals_global, int bCount,
    MGPU_MEM(KeyType)* ppKeys_global, MGPU_MEM(ValType)* 
ppVals_global, 
    Comp comp, CudaContext& context);
 
// Specialization of SetOpPairs with Comp = mgpu::less<T>.
template<MgpuSetOp Op, bool Duplicates, typename KeysIt1, typename 
KeysIt2,
    typename ValsIt1, typename ValsIt2, typename KeyType, typename 
ValType>
MGPU_HOST int SetOpPairs(KeysIt1 aKeys_global, ValsIt1 aVals_global, 
int aCount,
    KeysIt2 bKeys_global, ValsIt2 bVals_global, int bCount,
    MGPU_MEM(KeyType)* ppKeys_global, MGPU_MEM(ValType)* 
ppVals_global, 
    CudaContext& context);

The four multiset operations

The C++ standard library includes four multiset operations: std::set_intersection, 
std::set_union, std::set_difference, and std::set_symmetric_difference. These 
functions find key-rank matches over two sorted input arrays.

Consider inputs A and B:

A: 1 1 2 3 3 3 5 6 6 6 6 7 7 8 8 9

B: 1 2 2 3 3 3 3 6 6 6 6 8

We rank the keys in each array by the order of appearance in their respective arrays:

A: 10 11 20 30 31 32 50 60 61 62 63 70 71 80 81 90

B: 10 20 21 30 31 32 33 60 61 62 63 80

Elements are placed in slots according to key-rank. Elements from both inputs that match are placed in the 
same slot:

A: 10 11 20  30 31 32  50 60 61 62 63 70 71 80 81 90

B: 10  20 21 30 31 32 33  60 61 62 63   80   

Once the inputs are partitioned and paired, the set operations can be described and implemented by 
examining key-rank slots in isolation. For all operations, results are emitted from left-to-right, and each key-



rank slot may generate zero or one outputs.

C++ std::set_intersection reference implementation

1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16

template <class InputIterator1, class InputIterator2, class 
OutputIterator>
  OutputIterator set_intersection (InputIterator1 first1, 
InputIterator1 last1,
                                   InputIterator2 first2, 
InputIterator2 last2,
                                   OutputIterator result)
{
  while (first1!=last1 && first2!=last2)
  {
    if (*first1<*first2) ++first1;
    else if (*first2<*first1) ++first2;
    else {
      *result = *first1; first2;
      ++result; ++first1; ++first2;
    }
  }
  return result;
}

A: 10 11 20  30 31 32  50 60 61 62 63 70 71 80 81 90

B: 10  20 21 30 31 32 33  60 61 62 63   80   

Set-intersection selects elements in A (in green) that have no key-rank match in B. Only elements in A are 
returned to the caller.

C++ std::set_union reference implementation

template <class InputIterator1, class InputIterator2, class 
OutputIterator>
  OutputIterator set_union (InputIterator1 first1, InputIterator1 
last1,
                            InputIterator2 first2, InputIterator2 
last2,
                            OutputIterator result)
{
  while (true)
  {
    if (first1==last1) return std::copy(first2,last2,result);
    if (first2==last2) return std::copy(first1,last1,result);
 
    if (*first1<*first2) { *result = *first1; ++first1; }
    else if (*first2<*first1) { *result = *first2; ++first2; }
    else { *result = *first1; ++first1; ++first2; }
    ++result;

http://www.cplusplus.com/reference/algorithm/set_union/
http://www.cplusplus.com/reference/algorithm/set_intersection/


  }
}

A: 10 11 20  30 31 32  50 60 61 62 63 70 71 80 81 90

B: 10  20 21 30 31 32 33  60 61 62 63   80   

Set-union selects all elements in A plus any elements in B that have no key-rank match in A.

C++ std::set_difference reference implementation

7
8
9
10
11
12
13

template <class InputIterator1, class InputIterator2, class 
OutputIterator>
  OutputIterator set_difference (InputIterator1 first1, InputIterator1 
last1,
                                 InputIterator2 first2, InputIterator2 
last2,
                                 OutputIterator result)
{
  while (first1!=last1 && first2!=last2)
  {
    if (*first1<*first2) { *result = *first1; ++result; ++first1; }
    else if (*first2<*first1) ++first2;
    else { ++first1; ++first2; }
  }
  return std::copy(first1,last1,result);
}

A: 10 11 20  30 31 32  50 60 61 62 63 70 71 80 81 90

B: 10  20 21 30 31 32 33  60 61 62 63   80   

Set-difference selects only elements in A that don't have key-rank matches in B. Set-difference returns the 
elements in A that aren't returned by set-intersection.

C++ std::set_symmetric_difference reference implementation

http://www.cplusplus.com/reference/algorithm/set_symmetric_difference/
http://www.cplusplus.com/reference/algorithm/set_difference/

template <class InputIterator1, class InputIterator2, class 
OutputIterator>
  OutputIterator set_symmetric_difference (
                            InputIterator1 first1, InputIterator1 
last1,
                            InputIterator2 first2, InputIterator2 
last2,
                            OutputIterator result)
{
  while (true)
  {
    if (first1==last1) return std::copy(first2,last2,result);
    if (first2==last2) return std::copy(first1,last1,result);
 
    if (*first1<*first2) { *result=*first1; ++result; ++first1; }
    else if (*first2<*first1) { *result = *first2; ++result; ++first2; }
    else { ++first1; ++first2; }
  }
}

A: 10 11 20  30 31 32  50 60 61 62 63 70 71 80 81 90

B: 10  20 21 30 31 32 33  60 61 62 63   80   

Set-symmetric-difference selects elements in A that have no key-rank match in B and elements in B that have 
no key-rank match in A.

Balanced Path

The four C++ reference codes are not obviously parallelizable. Although they appear merge-like, parallel 
Merge Path partitioning is inadequete. Both components of a key-rank matches must be made available to the 
same thread and therefore always must appear on the same side of any cross-diagonal. In the sample arrays 
above, Merge Path follows all four 6 keys from A before pursuing any of the 6 keys from B. 

The merge order is

6a
0 6a

1 6a
2 6a

3 6b
0 6b

1 6b
1 6b

3.

We want a search that finds the intersection of cross-diagonals with a path which interleaves key-rank 
matches:

6a
0 6b

0 6a
1 6b

1 6a
2 6b

2 6a
3 6b

3.

Isolating the discovering of key-rank matches to partitioning code helps us cleanly separate scheduling logic 
from multiset logic (code resembling the C++ reference implementations from above). This has been a theme 
of all MGPU algorithms.



The green line in this figure is the now-familiar Merge Path. It greedily consumes all duplicates from A (the 
long right-directed segments) and then consumes all duplicates from B (the long down-directed segments). 
The red line is our answer for multiset-like problems: Balanced Path. This curve pairs all inputs by key-rank 
match.

To satisfy our partitioning requirements, finding the intersection of the Balanced Path with cross-diagonals is 
not sufficient. We must also introduce the concept of starred diagonals. 

Consider diag6 in this figure. It cuts the Balanced Path putting key-rank pair (6a
1, 6b

1) on the left and key-

rank pair (6a
2, 6b

2) on the right. This is proper behavior, as the partition to the left is assigned both parts of 

the key-rank pair 61 and the partition on the right is assigned both parts of key-rank pair 62.

diag7, on the other hand, intersects the Balanced Path between elements 6a
3 and 6b

3. This violates multisets' 

partitioning conditions: the operation-specific logic is constructed with the requirement that each CTA or 
thread is given both parts of all key-rank matches. To satisfy this we star the offending cross-diagonal, 
causing the partition on the left to steal the next element in B (on the right side of the cross-diagonal) if it's a 
key-rank match for the last element in A (on the left side of the cross-diagonal). Starring diag7 assigns four 

elements to the partition (diag6, diag7) and only two elements to the partition (diag7, diag8). Balanced Path 

partitions aren't precisely equal in size, but they deviate from the target grain size by only ±1.

Important: Like Merge Path, this new Balanced Path curve is constructed sequentially. We are interested in 
finding the intersection of the Balanced Path with a cross-diagonal without actually constructing the 
Balanced Path. We start by finding the intersection of the cross-diagonal with the Merge Path for the same 
data. Establishing this, additional searches locate where the Merge Path and Balanced Path most recently 
diverged. We then forward project from this point onto the cross-diagonal to complete the partitioning 
search.

This is an intricate task, but fortunately there is a simple geometrically-motivated algorithm:

1. Find the intersection of the cross-diagonal with the Merge Path (the curve in green).



2. Read the key in the B array at the point of intersection. If the cross-diagonal intersects a run of 
duplicates, it will be a run of elements with this key, because Merge Path consumes all duplicates 
from A before any in B.

3. Binary search to find the first occurrence of this key in arrays A and B—this is the position where the 
Balanced Path diverges from the Merge Path. In the figure, diag3 intersects the cross-diagonal where 

the A cursor points to 5a
0 and the B cursor to 3b

0. The B key is 3. Binary searching for the lower-

bound of 3 into both arrays reveals the point of divergence at (3, 3), or where diag2 intersects the 

Merge Path.

4. Use the distance between the intersection computed in 1 and the divergence point in 3 to establish the 
duplicate run length. Project the Balanced Path (the curve in red) from the point of divergence along 
this run length, creating a stair-step pattern by dividing the run length evenly over A and B intervals.

5. We can't advance the A and B cursors evenly if the run length is odd. If this projection would separate 
a key-rank match, putting the A match on the left and the B match on the right of the cross-diagonal, 
we star the cross-diagonal, instructing the left partition to steal the B match and the right partition to 
cede it.

include/device/ctasearch.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctasearch.cuh

template<bool Duplicates, typename IntT, typename InputIt1, typename 
InputIt2, 
    typename Comp>
MGPU_HOST_DEVICE int2 BalancedPath(InputIt1 a, int aCount, InputIt2 b,
    int bCount, int diag, int levels, Comp comp) {
 
    typedef typename std::iterator_traits<InputIt1>::value_type T;
 
    int p = MergePath<MgpuBoundsLower>(a, aCount, b, bCount, diag, 
comp);
    int aIndex = p;
    int bIndex = diag - p;
 
    bool star = false;
    if(bIndex < bCount) {
        if(Duplicates) {
            T x = b[bIndex];
 
            // Search for the beginning of the duplicate run in both A 
and B.
            // Because 
            int aStart = BiasedBinarySearch<MgpuBoundsLower, IntT>(a, 
aIndex, x,
                levels, comp);
            int bStart = BiasedBinarySearch<MgpuBoundsLower, IntT>(b, 
bIndex, x,
                levels, comp);
 
            // The distance between the merge path and the lower_bound 
is the 
            // 'run'. We add up the a- and b- runs and evenly 
distribute them to
            // get a stairstep path.
            int aRun = aIndex - aStart;
            int bRun = bIndex - bStart;
            int xCount = aRun + bRun;
 
            // Attempt to advance b and regress a.
            int bAdvance = max(xCount>> 1, xCount - aRun);
            int bEnd = min(bCount, bStart + bAdvance + 1);
            int bRunEnd = BinarySearch<MgpuBoundsUpper>(b + bIndex, 
                bEnd - bIndex, x, comp) + bIndex;
            bRun = bRunEnd - bStart;
 
            bAdvance = min(bAdvance, bRun);
            int aAdvance = xCount - bAdvance;
 
            bool roundUp = (aAdvance == bAdvance + 1) && (bAdvance < 
bRun);
            aIndex = aStart + aAdvance;
 
            if(roundUp) star = true;
        } else {



BalancedPath returns the intersection of the cross-diagonal diag and the Balanced Path for input arrays 
a and b in .x, and the star status of the intersection in .y. When Duplicates is true, we closely follow 
the five steps already listed. (aIndex, bIndex) is the coordinate of the cross-diagonal intersection with the 
Merge Path. Binary searches on A and B returns the point of divergence with the Balanced Path at (aStart, 
bStart). 

The function attemps to evenly distribute the total run length xCount over both inputs. However it can only 
distribute within duplicate runs of the sought-for key x. It runs a third binary search, an upper-bound, to find 
the last occurrence of the key in A. If the number of duplicates of x in A is less than half the distance from 
the divergence point to the cross-diagonal's intersection with the Merge Path, we project the Balanced Path 
only to the end of A's duplicate run and distribute the remainder of A's half to B. Graphically this keeps the 
red stair-step Balanced Path bounded on the top and to the right by the green Merge Path. In the figure, if we 
were to blindly project a stair-step path from the intersection of diag2 with the Merge Path, we'd violate key-

rank ordering where diag3 happens to intersect the Merge Path: 5a
0 would be consumed prior to 3b

3.

FindSetPartition, the multisets counterpart to MergePathPartitions, moves the starred flag into the 
most-significant bit of the index when executing a global partitioning pass.

include/device/ctasearch.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/ctasearch.cuh
http://nvlabs.github.io/moderngpu/mergesort.html#mergepathpartitions

template<MgpuBounds Bounds, typename IntT, typename It, typename T, 
    typename Comp>
MGPU_HOST_DEVICE void BinarySearchIt(It data, int& begin, int& end, T 
key, 
    int shift, Comp comp) {
 
    IntT scale = (1<< shift) - 1;
    int mid = (int)((begin + scale * end)>> shift);
 
    T key2 = data[mid];
    bool pred = (MgpuBoundsUpper == Bounds) ? 
        !comp(key, key2) : 
        comp(key2, key);
    if(pred) begin = mid + 1;
    else end = mid;
}
 
template<MgpuBounds Bounds, typename IntT, typename T, typename It,
    typename Comp>
MGPU_HOST_DEVICE int BiasedBinarySearch(It data, int count, T key, int 
levels,
    Comp comp) {
 
    int begin = 0;
    int end = count;
 
    if(levels >= 4 && begin < end)
        BinarySearchIt<Bounds, IntT>(data, begin, end, key, 9, comp);
    if(levels >= 3 && begin < end)
        BinarySearchIt<Bounds, IntT>(data, begin, end, key, 7, comp);
    if(levels >= 2 && begin < end)
        BinarySearchIt<Bounds, IntT>(data, begin, end, key, 5, comp);
    if(levels >= 1 && begin < end)
        BinarySearchIt<Bounds, IntT>(data, begin, end, key, 4, comp);
 
    while(begin < end)
        BinarySearchIt<Bounds, int>(data, begin, end, key, 1, comp);
    return begin;
}

The cost of launching two conventional binary searches in addition to the Merge Path inside 
BalancedPath would be very high. If both input arrays have 10 million elements, and the intersection of 
the center cross-diagonal at 10 million with the Merge Path splits both arrays in half at 5 million, it hardly 
makes sense to run a conventional binary search over (0, 5000000) on both arrays to find the first occurrence 
of key x. Each search has 23 levels of depth, a heavy price to pay for key ranking. These multset functions 
assume that the number of duplicates in a run is much smaller than the length of the inputs; i.e., the first 
occurrence is, on average, close to the intersection of the cross-diagonal with the Merge Path, no matter 
where that intersection is. (It's unlikely that the user would want to run multiset operations on arrays with 
many thousands of duplicates of each key.)

We start with the interval (0, 5000000), but rather than splitting at the middle, we split 511/512ths of the way 
to the right, at 4990234. If there are fewer than 9766 duplicates of x in the array (highly likely), the begin 



iterator is advanced to 49909235—this gamble saved us 8 levels of binary searching. Otherwise we search 
again at (0, 4990234).

On the next level, we split the interval (0, 4990234) 127/128ths of the way to the right, at 4999923. If the 
first occurence of x is to the right of this (meaning there are fewer than 77 duplicates), we set begin to 
4999924 and have saved another 6 levels of searching.

This strategy is called BiasedBinarySearch. When executed on data in a CTA's shared memory, two 
biased iterations are run before the dynamic loop of symmetric iterations, with weights 31/32 and 15/16. For 
global memory searches, we run four searches: 511/512, 127/128, 31/32, and 15/16 before entering the 
dynamic loop of symmetric searches.

To avoid undesirable division, we multiply the end iterator by 511 and add it to the left iterator, then shift 9 
bits to simulate division. The multiplication may cause an overflow during searches into global memory, 
where the input arrays are large compared to the capacity of 32-bit ints. Biased search uses 64-bit integers to 
accommodate the need for more bits during midpoint calculation, but only during global search, when 
overflow is acutally possible. When called from the global partitioning kernel, BalancedPath is 
specialized with IntT = int64 and passed levels = 4, to use wide multiplication and more 
aggressive biasing. When called from the intra-CTA multisets function, BalancedPath is specialized with 
IntT = int and passed levels = 2.

Serial multiset operations

BalancedPath partitions global data into NV±1-sized tiles. These intervals are loaded into CTA shared 
memory and further partitioned into VT±1-sized chunks. Just as we have SerialMerge to merge short 
intervals from shared memory into register, we have four serial set functions to read key-rank pairs from 
shared memory and produce results into register.

CUDA serial set-intersection from include/device/serialsets.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/serialsets.cuh

template<int VT, bool RangeCheck, typename T, typename Comp>
MGPU_DEVICE int SerialSetIntersection(const T* data, int aBegin, int 
aEnd,
    int bBegin, int bEnd, int end, T* results, int* indices, Comp comp) 
{
 
    const int MinIterations = VT / 2;
    int commit = 0;
     
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        bool test = RangeCheck ? 
            ((aBegin + bBegin < end) && (aBegin < aEnd) && (bBegin < 
bEnd)) :
            (i < MinIterations || (aBegin + bBegin < end));
 
        if(test) {
            T aKey = data[aBegin];
            T bKey = data[bBegin];
 
            bool pA = comp(aKey, bKey);
            bool pB = comp(bKey, aKey);
             
            // The outputs must come from A by definition of set 
interection.
            results[i] = aKey;
            indices[i] = aBegin;
 
            if(!pB) ++aBegin;
            if(!pA) ++bBegin;
            if(pA == pB) commit |= 1<< i;
        }
    }
    return commit;
}

A: 10 11 20  30 31 32  50 60 61 62 63 70 71 80 81 90

B: 10  20 21 30 31 32 33  60 61 62 63   80   

For a full tile, the number of inputs equals VT±1. When key-rank elements are arranged in slots, as above, 
each iteration of the serial set operation advances exactly one slot, and increments one or both input pointers. 
Because we don't know a priori how many outputs a thread generates, each iteration stores the result (there 
can be no more than one result per slot) to results[i] and sets the i'th bit in the commit bitfield. After 
the loop has ended, each thread will count its outputs, cooperatively scan them, and compact into shared 
memory. This technique lets us load keys from shared memory (we need dynamic indexing, so that much is a 
requirement) and store to register (indexing into the output arrays is static thanks to loop unrolling by the 
template argument VT).

If we're processing a full tile and the caller is able to load one additional element from both A and B arrays, 



the four serial set ops are specialized with RangeCheck = false. In this case the function knows it 
won't be dealing with the end of the array and can elide range-checking tests. This reduces latency in the 
function, and because the kernel is intentionally underoccupied (we jack VT up to amortize the expensive 
Balanced Path intersection searches), it significantly boosts performance.

SerialSetIntersection takes indices (aBegin, bBegin) to the start of the thread's partition in 
shared memory. (aEnd, bEnd) are indices to the end of the tile's A and B intervals in shared memory. end 
is passed as aBegin + bBegin + partition size. This lets us check if we've consumed all the inputs for the 
entire partition with just a single comparison, no matter the state that the set operation takes us to.

Each thread makes exactly VT iterations through the inner loop, although as few as VT / 2 actually perform 
key comparisons. If the cross-diagonal on the left is starred, and the cross-diagonal on the right isn't, the 
thread has VT - 1 inputs. If VT = 11, there are minimally 10 elements per thread (for a full tile), and if all 
inputs are paired, there are only 5 active slots. So the first 5 iterations, 0-4, are executed unconditionally, and 
the next six, 5-10, execute only if aBegin + bBegin < end is true.

The actual logic for multiset operations is very simple:

1. Load the next keys from A and B into aKey and bKey.

2. Evaluate aKey < bKey and bKey < aKey. 

3. Speculatively store A into the results array—all results in set-intersection come from A, and an 
iteration's bit in the commit bitfield must be set for the result to be compacted.

4. Advance the indices to the next key-rank slot. If aKey <= bKey (i.e. !comp(bKey, aKey)), 
increment aBegin. If bKey <= aKey (i.e. !comp(aKey, bKey)), increment bBegin.

5. If the set condition was satisfied, set bit i in commit to commit the result. In the case of set-
intersection, both elements must make a key-rank match. We've already compared both keys against 
each other, and want to emit if they're equal. Predicates pA and pB can both be false (indicating 
equality), but they can't both be true (indicating A < B and B < A). For set-intersection we set the 
commit bit if pA == pB.

CUDA serial set-union from include/device/serialsets.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/serialsets.cuh

template<int VT, bool RangeCheck, typename T, typename Comp>
MGPU_DEVICE int SerialSetUnion(const T* data, int aBegin, int aEnd,
    int bBegin, int bEnd, int end, T* results, int* indices, Comp comp) 
{
 
    const int MinIterations = VT / 2;
    int commit = 0;
     
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        bool test = RangeCheck ? 
            (aBegin + bBegin < end) :
            (i < MinIterations || (aBegin + bBegin < end));
 
        if(test) {
            T aKey = data[aBegin];
            T bKey = data[bBegin];
 
            bool pA = false, pB = false;
            if(RangeCheck && aBegin >= aEnd) 
                pB = true;
            else if(RangeCheck && bBegin >= bEnd) 
                pA = true;
            else {
                // Both are in range.
                pA = comp(aKey, bKey);
                pB = comp(bKey, aKey);
            }
 
            // Output A in case of a tie, so check if b < a.
            results[i] = pB ? bKey : aKey;
            indices[i] = pB ? bBegin : aBegin;
            if(!pB) ++aBegin;
            if(!pA) ++bBegin;
            commit |= 1<< i;
        }
    }
    return commit;
}

A: 10 11 20  30 31 32  50 60 61 62 63 70 71 80 81 90

B: 10  20 21 30 31 32 33  60 61 62 63   80   

The range-checking logic is basically the same for all four serial set op functions. The material difference is 
how results are selected and committed. If both keys are in-range, SerialSetUnion compares them and 
sets the result to A, or B if B is smaller. One value from each key-rank slot is always emitted, so the commit 
flag is set unconditionally.

CUDA serial set-difference from include/device/serialsets.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/serialsets.cuh

template<int VT, bool RangeCheck, typename T, typename Comp>
MGPU_DEVICE int SerialSetDifference(const T* data, int aBegin, int aEnd, 
    int bBegin, int bEnd, int end, T* results, int* indices, Comp comp) 
{ 
 
    const int MinIterations = VT / 2;
    int commit = 0;
     
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        bool test = RangeCheck ? 
            (aBegin + bBegin < end) :
            (i < MinIterations || (aBegin + bBegin < end));
        if(test) {
            T aKey = data[aBegin];
            T bKey = data[bBegin];
 
            bool pA = false, pB = false;
            if(RangeCheck && aBegin >= aEnd)
                pB = true;
            else if(RangeCheck && bBegin >= bEnd)
                pA = true;
            else {
                pA = comp(aKey, bKey);
                pB = comp(bKey, aKey);
            }
 
            // The outputs must come from A by definition of set 
difference.
            results[i] = aKey;
            indices[i] = aBegin;
            if(!pB) ++aBegin;
            if(!pA) ++bBegin;
            if(pA) commit |= 1<< i;
        }
    }
    return commit;
}

A: 10 11 20  30 31 32  50 60 61 62 63 70 71 80 81 90

B: 10  20 21 30 31 32 33  60 61 62 63   80   

Set-difference is the complement of set-intersection. The result is unconditionally set to the A element and 
committed if aKey < bKey.

CUDA serial set-symmetric difference from include/device/serialsets.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/device/serialsets.cuh

template<int VT, bool RangeCheck, typename T, typename Comp>
MGPU_DEVICE int SerialSetSymDiff(const T* data, int aBegin, int aEnd, 
    int bBegin, int bEnd, int end, T* results, int* indices, Comp comp) 
{ 
 
    const int MinIterations = VT / 2;
    int commit = 0;
     
    #pragma unroll
    for(int i = 0; i < VT; ++i) {
        bool test = RangeCheck ? 
            (aBegin + bBegin < end) :
            (i < MinIterations || (aBegin + bBegin < end));
        if(test) {
            T aKey = data[aBegin];
            T bKey = data[bBegin];
 
            bool pA = false, pB = false;
            if(RangeCheck && (bBegin >= bEnd))
                pA = true;
            else if(RangeCheck && (aBegin >= aEnd))
                pB = true;
            else {
                pA = comp(aKey, bKey);
                pB = comp(bKey, aKey);
            }
 
            results[i] = pA ? aKey : bKey;
            indices[i] = pA ? aBegin : bBegin;
            if(!pA) ++bBegin;
            if(!pB) ++aBegin;
            if(pA != pB) commit |= 1<< i;
        }
    }
    return commit;
}

A: 10 11 20  30 31 32  50 60 61 62 63 70 71 80 81 90

B: 10  20 21 30 31 32 33  60 61 62 63   80   

Set-symmetric difference uses the identical range-checking expressions as set-difference. However instead of 
conditionally emitting A, it emits A or B, whichever is smaller. If the keys are equal, the function moves to 
the next key-rank frame.

Multisets kernel

include/kernels/sets.cuh

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/sets.cuh

template<int NT, int VT, MgpuSetOp Op, bool Duplicates, typename 
InputIt1,
    typename InputIt2, typename T, typename Comp>
MGPU_DEVICE int DeviceComputeSetAvailability(InputIt1 a_global, int 
aCount,
    InputIt2 b_global, int bCount, const int* bp_global, Comp comp,
    int tid, int block, T* results, int* indices, int4& range, 
    bool& extended, T* keys_shared) {
          
    const int NV = NT * VT;
    int gid = NV * block;
    int bp0 = bp_global[block];
    int bp1 = bp_global[block + 1];
 
    // Compute the intervals into the two source arrays.
    int a0 = 0x7fffffff & bp0;
    int a1 = 0x7fffffff & bp1;
    int b0 = gid - a0;
    int b1 = min(aCount + bCount, gid + NV) - a1;
 
    // If the most sig bit flag is set, we're dealing with a 'starred' 
diagonal
    // that shifts the point of intersection up.
    int bit0 = (0x80000000 & bp0) ? 1 : 0;
    int bit1 = (0x80000000 & bp1) ? 1 : 0;
    b0 += bit0;
    b1 += bit1;
 
    // Attempt to load an 'extended' frame by grabbing an extra value 
from each
    // array.
    int aCount2 = a1 - a0;
    int bCount2 = b1 - b0;
    extended = (a1 < aCount) && (b1 < bCount);
    int bStart = aCount2 + (int)extended;
 
    DeviceLoad2ToShared<NT, VT, VT + 1>(a_global + a0, aCount2 + 
(int)extended,
        b_global + b0, bCount2 + (int)extended, tid, keys_shared);
    int count = aCount2 + bCount2;
     
    // Run a Balanced Path search for each thread's starting point.
    int diag = min(VT * tid - bit0, count);
    int2 bp = BalancedPath<Duplicates, int>(keys_shared, aCount2, 
        keys_shared + bStart, bCount2, diag, 2, comp);
 
    int a0tid = bp.x;
    int b0tid = VT * tid + bp.y - bp.x - bit0;
 
    int commit;
    if(extended)
        commit = SerialSetOp<VT, false, Op>(keys_shared, a0tid, 

aCount2, 
            bStart + b0tid, bStart + bCount2, bp.y, results, indices, 
comp);
    else
        commit = SerialSetOp<VT, true, Op>(keys_shared, a0tid, 
aCount2, 
            bStart + b0tid, bStart + bCount2, bp.y, results, indices, 
comp);
     
    range = make_int4(a0, a1, b0, b1);
    return commit;
}

Threads load the tile's Balanced Path intersections and extract the source list ranges (a0, a1) and (b0, b1). 
Note that a star flag causes an increment to the B component of each intersection.

Due to the relative complexity of the serial set operations, we attempt to load an extra element in A and B. If 
this succeeds we can elide range-checking logic because we are guaranteed of not running off the end of the 
arrays in shared memory. This optimization is also used in the vectorized sorted search. The MGPU kernels 
tend to run underoccupied on current generation hardware, and reducing predicate latency often outweighs 
the costs of the additional global loads.

The remaining code resembles DeviceMergeKeysIndices of MGPU Merge. BalancedPath finds 
the starting positions for serial set ops for each thread. As with Merge the results and indices are computed 
into register. However because multiset operations generate a data-dependent number of outputs, we return a 
bitfield commit that encodes the validity of each result. Results are compacted over this bitfield by the 
kernel.

The MGPU Multisets function SetOpKeys has two modes of operation:

• If compact = false:

1. KernelSetOp is launched with Stage = 0. This fully processes the input arrays, counts 
the outputs for each tile, and discards the results. Each tile stores its output count to global 
memory.

2. The caller scans the output counts and uses the reduction to allocate exact space for the 
globally-compact results.

3. KernelSetOp is launched with Stage = 1. This makes a second pass over the input 
arrays. The results are now compacted within the tile and cooperatively stored to the 
destination buffer.

• If compact = true:

1. The host function allocates a temporary buffer large enough to hold all the inputs.

2. KernelSetOp is launched with Stage = 2. This compacts multiset results within tiles 
and stores the results to the temporary buffer at tile offsets. Each tile stores its output count to 
global memory.

3. The caller scans the output counts and uses the reduction to allocate exact space for the 

http://nvlabs.github.io/moderngpu/merge.html#algorithm
http://nvlabs.github.io/moderngpu/sortedsearch.html#parallel


globally-compacted results.

4. The host launches KernelSetCompact to compact tiles of results from the temporary 
buffer into the destination buffer.

Compaction behavior is more efficient because the input arrays are processed only once, but it requires a lot 
of temporary storage. This is the default mode for StreamOpKeys (the keys-only multiset function). The 
compaction mode is not available for StreamOpPairs: we'd need two temporary buffers (for keys and 
values), and would have to copy values twice, wasting both space and bandwidth. By using the count-scan-
stream pattern, the first KernelSetOp launch only touches keys; the second KernelSetOp launch 
compacts keys and indices inside the CTA; it stores the keys then cooperatively gathers and stores values.

include/kernels/sets.cuh

template<typename Tuning, MgpuSetOp Op, bool Duplicates, int Stage,
    bool HasValues, typename KeysIt1, typename KeysIt2, typename 
KeysIt3,
    typename ValsIt1, typename ValsIt2, typename ValsIt3, typename Comp>
MGPU_LAUNCH_BOUNDS void KernelSetOp(KeysIt1 aKeys_global, ValsIt1 
aVals_global,
    int aCount, KeysIt2 bKeys_global, ValsIt2 bVals_global, int bCount,
    int* counts_global, const int* bp_global, KeysIt3 keys_global, 
    ValsIt3 values_global, Comp comp) {
 
    typedef typename std::iterator_traits<KeysIt1>::value_type KeyType;
    typedef typename std::iterator_traits<ValsIt1>::value_type ValType;
    typedef MGPU_LAUNCH_PARAMS Params;
    const int NT = Params::NT;
    const int VT = Params::VT;
    const int NV = NT * VT;
    typedef CTAReduce<NT, ScanOpAdd> R;
    typedef CTAScan<NT, ScanOpAdd> S;
 
    union Shared {
        KeyType keys[NT * (VT + 1)];
        int indices[NV];
        typename R::Storage reduce;
        typename S::Storage scan;
    };
    __shared__ Shared shared;
 
    int tid = threadIdx.x;
    int block = blockIdx.x;
     
    // Run the set operation. Return a bitfield for the selected keys.
    KeyType results[VT];
    int indices[VT];
    int4 range;
    bool extended;
    int commit = DeviceComputeSetAvailability<NT, VT, Op, Duplicates>(
        aKeys_global, aCount, bKeys_global, bCount, bp_global, comp, 
tid, block,

https://github.com/NVlabs/moderngpu/blob/master/include/kernels/sets.cuh

        results, indices, range, extended, shared.keys);
    aCount = range.y - range.x;
    bCount = range.w - range.z;
 
    // scan or reduce over the number of emitted keys per thread.
    int outputCount = popc(commit);
    int outputTotal;
    if(0 == Stage) {
        // Stage 0 - count the outputs.
        outputTotal = R::Reduce(tid, outputCount, shared.reduce);
    } else {
        int globalStart = (1 == Stage) ? counts_global[block] : (NV * 
block);
 
        // Stage 1 or 2 - stream the keys.
        int scan = S::Scan(tid, outputCount, shared.scan, 
&outputTotal);
 
        // Write the commit results to shared memory.
        int start = scan;
        #pragma unroll
        for(int i = 0; i < VT; ++i)
            if((1<< i) & commit)
                shared.keys[start++] = results[i];
        __syncthreads();
 
        // Store keys to global memory.
        DeviceSharedToGlobal<NT, VT>(outputTotal, shared.keys, tid, 
            keys_global + globalStart);
 
        if(HasValues) {
            // indices[] has gather indices in thread order. Compact 
and store
            // these to shared memory for a transpose to strided 
order.     
            start = scan;
            #pragma unroll
            for(int i = 0; i < VT; ++i)
                if((1<< i) & commit)
                    shared.indices[start++] = indices[i];
            __syncthreads();
 
            aVals_global += range.x;
            bVals_global += range.z;
            values_global += globalStart;
            if(MgpuSetOpIntersection == Op || MgpuSetOpDiff == Op)
                DeviceGatherGlobalToGlobal<NT, VT>(outputTotal, 
aVals_global,
                    shared.indices, tid, values_global, false);
            else
                DeviceTransferMergeValues<NT, VT>(outputTotal, 
aVals_global,



                    bVals_global, aCount + (int)extended, 
shared.indices, tid, 
                    values_global, false);
        }
    }
 
    if(1 != Stage && !tid)
        counts_global[block] = outputTotal;
}

Judicious factoring allows one implementation of KernelSetOp to support all three multiset launches 
described above. Although the function feels more like a merge, the implementation has more in common 
with vectorized sorted search, in that valid outputs are compacted with a loop over a commit bitfield. To 
copy values, we make a second loop over the set bits in commit, compact the indices of valid outputs to 
shared memory, and cooperatively gather and store data from values_global using these indices. 

Set-union and set-symmetric difference return elements from both arrays, and for these merge-like operations 
we tap DeviceTransferMergeValues to facilitate the gather and store. Set-intersection and set-
difference return only elements from the A input, presenting an opportunity for optimization: we call 
DeviceGatherGlobalToGlobal; it's similar to DeviceTransferMergeValues but drops 
predication by only supports a single input array.

http://nvlabs.github.io/moderngpu/sortedsearch.html

